




DIRECTION DE L'ENERGIE NUCLEAIRE

DEPARTEMENT MODELISATION DE SYSTEMES ET STRUCTURES

SERVICE FLUIDES NUMERIQUES, MODELISATION ET ETUDES



RAPPORT DM2S

SFME/LTMF/RT/03-035/A

Tutorial CAST3M pour la mécanique des fluides

F. Dabbene



RAPPORT DM2S

REFERENCES : SFME/LTMF/RT/03-035/A

TITRE : Tutorial CAST3M pour la mécanique des fluides

AUTEURS	SIGNATURES	AUTEURS	SIGNATURES
F. Dabbene			

RESUME : Ce rapport est le support papier de la partie pratique d'un cours d'initiation à la simulation numérique en mécanique des fluides à l'aide de CAST3M. Dans ce document, seuls sont abordés des exemples simples. Nous espérons qu'ils sont suffisants pour permettre aux utilisateurs de CAST3M de construire des modèles numériques plus complexes ou de comprendre des jeux de données fournis par ailleurs. A chaque section correspond un cas test. Certaines redites permettent de lire chaque section du rapport de façon indépendante. Dans le cas d'une lecture en continu, cela permet d'identifier les points importants.

MOTS CLES : CAST3M, TUTORIAL
AFFAIRE : DSOE/SIMULATION Projet : MECANIQUE EOTP : A-MECAN-02-01
Titre de l'action : MECANIQUE DES FLUIDES

			Visa			
A	15/09/03	65	Nom Date	J.-P. Magnaud		H. Paillère
Indice	Date	Pages		Vérificateur	Autre visa	Approbateur

Les informations contenues dans ce document sont réservées aux destinataires nommément désignés et ne peuvent recevoir aucune diffusion sans l'autorisation expresse du DM2S



DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 3/65

Tutorial CAST3M pour la mécanique des fluides

Liste des modifications

Indice	Date	Motif et description de la modification
A	15/09/03	Document initial



DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 4/65

Tutorial CAST3M pour la mécanique des fluides

Table des matières

Introduction	5
1 Réalisation d'un maillage	6
1.1 Objectif	6
1.2 Consignes	6
1.3 Correction	7
1.4 Complément	10
1.5 Fichier Gibiane	10
2 Objet modèle	14
2.1 Objectif	14
2.2 Consignes	14
2.3 Correction	14
2.4 Complément	19
2.5 Fichier Gibiane	19
3 Equation de la chaleur	23
3.1 Objectif et rappels	23
3.1.1 Equation de la chaleur	23
3.1.2 Solutions particulières de l'équation de la chaleur	24
3.1.3 Recherche d'une solution stationnaire	24
3.2 Consignes	26
3.3 Correction	27
3.3.1 Influence du maillage	27
3.3.2 Estimation de l'ordre de convergence	28
3.3.3 Description du jeu de données	28
3.4 Complément	28
3.5 Fichiers Gibiane	29
4 Transport par diffusion-convection d'un scalaire passif	36
4.1 Objectif	36
4.2 Consignes	36
4.3 Correction	37



DEN

DM2S

Tutorial CAST3M pour la mécanique des fluides

4.3.1	Effet du Pe	37
4.3.2	Décentrement des termes convectifs	37
4.3.3	Description du jeu de données	37
4.4	Complément	39
4.5	Fichier Gibiane	40
5	Marche descendante	44
5.1	Objectif	44
5.2	Consignes	44
5.3	Correction	45
5.3.1	Structure de l'écoulement	45
5.3.2	Description du jeu de données	45
5.4	Complément	48
5.5	Fichiers Gibiane	48
5.5.1	Schéma semi explicite	48
5.5.2	Méthode de projection	50
5.5.3	Schéma implicite	52
6	Approximation de Boussinesq des équations de Navier-Stokes	55
6.1	Objectif	55
6.2	Consignes	55
6.3	Correction	56
6.3.1	Effet de Gr^* sur l'écoulement	56
6.3.2	Description du jeu de données	57
6.4	Complément	59
6.5	Fichier Gibiane	60
	Conclusion	65



DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 6/65

Tutorial CAST3M pour la mécanique des fluides

Introduction

Ce rapport est le support papier de la partie pratique d'un cours d'initiation à la simulation numérique en mécanique des fluides dispensé en troisième année à l'École Nationale Supérieure de Techniques Avancées (E.N.S.T.A.) depuis 1997. La partie théorique aborde sur le plan numérique différentes méthodes de discrétisation tant en espace (différences finies, éléments finis et volumes finis) qu'en temps (schémas explicites et implicites) ; sur le plan physique les modélisations associées à différents régimes d'écoulement (incompressible, compressible, visqueux et non visqueux, *etc.*).

CAST3M, logiciel développé au Commissariat à l'Énergie Atomique (C.E.A.) à Saclay par le Département Modélisation de Systèmes et Structures (DM2S), est le support informatique de ce cours. Certaines parties du cours sont aussi utilisées dans le cadre de formations internes au C.E.A. depuis 2003. Ces formations sont assurées par le «Laboratoire d'études des Transferts et de Mécanique des Fluides» (LTMF), acteur principal du développement de la mécanique des fluides dans CAST3M. Pour mémoire, le pilotage de CAST3M et le développement de la partie calcul de structures, domaine original du code, sont réalisés par le «Laboratoire de Mécanique Systèmes et Simulation» (LM2S). À l'adresse <http://www-cast3m.cea.fr> le lecteur intéressé trouvera un ensemble d'informations relatives au code.

Dans ce document, seuls sont abordés des exemples simples. Nous espérons qu'ils seront suffisants pour permettre aux utilisateurs de CAST3M de construire des modèles numériques plus complexes ou de comprendre des jeux de données fournis par ailleurs. En particulier, le «Recueil d'exemples commentées» de H. Paillère et F. Dabbene (disponible en téléchargement sur le site CAST3M) est un complément possible à ce rapport.

À chaque section est associé un exemple. Les parties «Objectif» et «Consignes» présentent le cas test et les exercices à réaliser. Dans la partie «Correction», on indique les éléments de syntaxe permettant le codage du cas test et les éléments physiques mis en évidence par la/les simulation(s) numérique(s). Le jeu de données est alors fourni. Parfois, une partie «Complément» contient quelques questions et indications subsidiaires ainsi que des références bibliographiques. Certaines redites permettent de lire chaque section du rapport de façon indépendante. Dans le cas d'une lecture en continu, cela permet d'identifier les points importants.

Les deux premières sections permettent d'utiliser le langage de commande de CAST3M et les fonctions nécessaires à la réalisation d'un jeu de données. Ainsi, la section 1 permet de construire un maillage et d'y associer des champs. Dans la section 2, on crée un objet modèle NAVIER_STOKES, porte d'entrée de la mécanique des fluides dans CAST3M.

Les sections suivantes abordent le transport d'un champ scalaire. Dans la section 3, on s'intéresse uniquement au transfert diffusif dont le problème modèle est l'équation de la chaleur stationnaire. Dans la section 4, le cas test de Smith et Hutton proposé dans le cadre d'un workshop permet de traiter du transport par diffusion-convection d'un champ scalaire passif. On couvre ainsi une gamme de *Peclet* allant de 0 à 10^6 , de la diffusion pure à la convection pure.

Enfin, nous traitons de la résolution des équations de Navier-Stokes pour un écoulement incompressible et pour un fluide dilatable sous l'effet de la température — sous l'hypothèse de Boussinesq. Dans la section 5, on traite le problème de la marche descendante (écoulement laminaire en milieu ouvert). Dans la section 6, on s'intéresse à l'écoulement dans un cylindre contenant de l'oxyde d'uranium (géométrie axisymétrique en milieu fermé). Le fluide étant radioactif, il chauffe de sorte qu'on doit tenir compte des effets de convection naturelle (couplage avec la température).

Bonne route.



DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 7/65

Tutorial CAST3M pour la mécanique des fluides

1 Réalisation d'un maillage

1.1 Objectif

L'objectif de ce TP est

- ▷ de comprendre le fonctionnement de CAST3M (langage de commande GIBIANE, exécution, *etc.*);
- ▷ de manipuler les opérateurs permettant de créer un maillage en appliquant le principe d'analyse descendante / réalisation ascendante;
- ▷ de créer et tracer un CHAMPOIN fonction bilinéaire des coordonnées.

Le corrigé indique aussi les erreurs les plus répandues et les moyens d'y remédier.

1.2 Consignes

Il vous faut

1. réaliser et tracer le maillage d'une boîte rectangulaire alignée avec les axes;
2. construire et tracer un CHAMPOIN prenant les valeurs d'une fonction linéaire $f(x, y)$ aux sommets du maillage ($f(x, y) = ax + by + c$ où a , b et c sont des réels, x et y les coordonnées des sommets du maillage);
3. construire et tracer l'évolution du CHAMPOIN créé au cours de l'étape ci-dessus le long de l'axe situé à mi-hauteur du domaine maillé (i.e. à la cote $y = (y_{max} - y_{min})/2$).

Pour cela, vous aurez à utiliser les opérateurs suivants : INFO ; AIDE ; FIN ; OPTI ; '+' ; '-' ; '*' ; '/' ; DROI ; INVE ; ET ; DALL ; SURF ; COOR ; CONT ; TRAC ; EVOL ; DESS.

Pour travailler vous devez créer un fichier d'extension *.dgibi* à l'aide d'un éditeur de texte (comme vi, emacs, nedit, *etc.*). Ce fichier servira d'interface entre vous et le code CAST3M. Lorsque vous lancerez CAST3M depuis une fenêtre X par la commande ad hoc — par exemple *castem monfichier.dgibi* — les commandes seront interprétées par GIBIANE et CAST3M obéira — ou non — à vos injonctions.

Rappelons qu'une ligne de commande se termine par un point virgule (;). En supprimer quelques uns après l'obtention d'un premier résultat vous permettra de voir les messages d'erreurs générés en cas d'oublis involontaires.

Les tabulations et les caractères de contrôle autres que EOL et EOF — fin de ligne et fin de fichier — sont interdits : ils provoquent un comportement bizarre de GIBIANE. Si vous êtes au bord de la crise de nerf parce que le code ne réagit pas à votre convenance, supprimez la ligne qui pose problème et retapez la ! Si un caractère de contrôle s'y est glissé, c'est le plus sûr moyen de le supprimer. Il faut également signaler que certains éditeurs « oublient » dans certaines circonstances le caractère de fin de fichier. Laissez une dernière ligne blanche afin de circonvenir à ce problème.

Pour connaître la syntaxe d'un opérateur, utilisez l'opérateur INFO. Après avoir lancé CAST3M sans nom de fichier, tapez "INFO INFO ;" pour avoir l'affichage de la notice de l'opérateur INFO. Il existe sans doute sur votre machine une version HTML des notices : repérez le directory où CAST3M est stocké — *which castem* devrait vous fournir un début de piste — et cherchez. Nous rappelons qu'une notice débute à la première ligne de la notice et se termine à la dernière ligne. L'utilisateur pressé a en effet souvent tendance à ne lire qu'une partie des notices ... et à ne pas y trouver la réponse à sa question qui pourtant y figure.

L'opérateur AIDE permet de scanner l'ensemble des notices du code et d'y rechercher le mot clef indiqué en paramètre de l'opérateur.

Le choix des noms de variables (à gauche du signe =) est libre dans la limite de 8 caractères. Profitez-en pour mettre des noms significatifs.



DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 8/65

Tutorial CAST3M pour la mécanique des fluides

Les tracés du maillage, du CHAMPOIN et de son évolution le long de l'axe médian vont permettre de contrôler la justesse de chacune des étapes. Ils devraient ressembler aux figures présentées en fin de correction.

1.3 Correction

Après avoir lu l'intégralité des consignes, appliquons l'analyse descendante : on doit tracer l'évolution d'un CHAMPOIN le long d'un axe horizontal. La prise en compte de cet axe découpe naturellement la surface rectangulaire initiale en deux surfaces rectangulaires, de part et d'autre de l'axe médian horizontal. Les cotés de chaque rectangle sont entièrement décrits dès que les points situés aux extrémités des segments et le découpage du coté sont indiqués. Le croquis réalisé sur papier servira de support à la réalisation du jeu de données GIBIANE.

Paramètres généraux : La première des choses à faire dans le jeu de données GIBIANE est d'indiquer à CAST3M à l'aide de l'opérateur OPTI — comme OPTION — un certain nombre de variables d'environnement : la dimension de l'espace de travail, le type d'éléments servant à la réalisation du maillage, *etc.*. Dans le cas présent, nous sommes en dimension 2 et souhaitons utiliser de préférence des quadrangles à quatre noeuds :

```
OPTI DIME 2 ELEM QUA4 ;
```

Variables globales : L'analyse sur papier a permis d'identifier un certain nombre de variables globales qu'on initialise :

```
*  
* Variables globales  
*  
* LXO : Dimension suivant x  
* LYO : Dimension suivant y  
* XMIN,YMIN : Coordonnées du coin inférieur  
* NXO : Nombre de points suivant x  
* NYO : Nombre de points suivant y  
*  
LXO = 2.DO ; LYO = 1.DO ;  
XMIN = 0.DO ; YMIN = 0.DO ;  
NXO = 3 ; NYO = 5 ;  
XMAX = XMIN + LXO ; YMAX = YMIN + LYO ;  
YMID = YMAX + YMIN / 2.DO ;
```

Comme GIBIANE est un langage de commande, on peut tout paramétrer : longueur, nombre de points, coordonnées, *etc.*. L'intérêt est évident : pour la mise au point du jeu de données, inutile de générer de gros maillages. C'est pourquoi dans la programmation ci-dessus, NXO et NYO sont des entiers raisonnables. Ils sont différents et premier entre eux de façon à détecter d'éventuelles erreurs de codage. Après la mise au point, il suffira de modifier ces valeurs pour générer des maillages plus fin. Les flottants sont initialisés en double précision en utilisant la même syntaxe que le Fortran. On peut très bien se passer des DO et coder "LXO=2." mais autant garder de bonnes (vieilles?) habitudes.

Notez que la différence entre un entier et un flottant est l'absence ou la présence d'un point après les valeurs numériques. Faites-y attention surtout si vous effectuez des divisions : la division entre deux entiers — ou division entière — peut provoquer quelques surprises :

```
* Re : Nombre de Reynolds  
Re = 1000 ;  
IRe = 1 / Re ;  
LIST IRe ;
```




DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 9/65

Tutorial CAST3M pour la mécanique des fluides

La division entière de 1 par 1000 donne 0 au lieu de 10^{-3} qui est manifestement le résultat attendu. On s'étonnera alors à l'issue du calcul d'un manque de viscosité dans l'écoulement !

Enfin, la variable YMID contient bien la mi-hauteur : une phrase GIBIANE est décodée de gauche à droite sans règle de priorité. Les deux expressions suivantes sont donc équivalentes :

```
YMID = (YMAX + YMIN) / 2.DO ;
YMID = YMAX + YMIN / 2.DO ;
```

Utilisez à bon escient des parenthèses. Pour calculer le rayon les expressions suivantes ne donnent pas le même résultat. Seules les deux premières sont justes :

```
ROVRAI = X*X + (Y*Y) ** 0.5 ;
R1VRAI = ((X*X) + (Y*Y)) ** 0.5 ;
ROFAUX = X*X + Y*Y ** 0.5 ;
R1FAUX = (X*X + Y*Y) ** 0.5 ;
```

Points, droites et surfaces : L'analyse descendante ayant permis d'identifier tous les éléments du maillage, nous allons successivement créer les points, les segments et les surfaces.

Pour créer un point, il suffit d'indiquer ses coordonnées.

```
*
* Création des points
P1 = XMIN YMIN ;
P2 = XMAX YMIN ;
P3 = XMAX YMID ;
P4 = XMAX YMAX ;
P5 = XMIN YMAX ;
P6 = XMIN YMID ;
```

Pour créer les segments de droite, on utilise l'opérateur DROI. Cet opérateur permet de découper le segment de droite entre deux points en différentes parties de même longueur ou non. Si on transmet en argument de l'opérateur un point dit point initial, le nombre de segment *ns* et le point dit final, le segment de droite entre les deux points précisés sera découpé en *ns* segments d'égale longueur. Des options — faire "INFO DROI ;" pour consulter en ligne la notice complète de l'opérateur — permettent d'avoir un découpage non régulier du segment — options DINI et DFIN.

```
*
* Création des droites
P1P2 = P1 DROI NX0 P2 ;
P2P3 = P2 DROI NY0 P3 ;
P3P4 = P3 DROI NY0 P4 ;
P4P5 = P4 DROI NX0 P5 ;
P5P6 = P5 DROI NY0 P6 ;
P6P1 = P6 DROI NY0 P1 ;
P3P6 = P3 DROI NX0 P6 ;
P6P3 = INVE P3P6 ;
```

On choisit de décrire les frontières dans le sens trigonométrique. La ligne P3P6 commune aux deux sous-surfaces est créée avec DROI. La description de cette ligne dans le sens inverse est créée avec INVE. Si on utilisait à nouveau



DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 10/65

Tutorial CAST3M pour la mécanique des fluides

l'opérateur DR01 pour créer P6P3, les points intermédiaires situés entre les points P3 et P6 auraient été dupliqués. Il aurait alors fallu les fusionner avec ceux de la ligne P3P6 via l'opérateur ELIM. L'utilisation de l'opérateur ELIM est délicate dans certaines situations et donc fortement déconseillée dans un premier temps. Une application rigoureuse de la méthode de construction de maillage par analyse descendante / réalisation ascendante permet de s'affranchir de son usage, au moins en 2D, à condition d'utiliser l'opérateur INVE.

Le domaine étant rectangulaire, nous pouvons utiliser un pavage régulier du domaine par des éléments quadrangulaires. Le dallage de chaque partie est réalisé par l'opérateur DALL. On fournit à cet opérateur les quatre cotés du domaine à daller. Les cotés opposés doivent être découpés de la même façon, c'est à dire avec le même nombre de points. Même si dans certains cas les opérateurs de maillage rattrapent quelques bévues, il est conseillé de décrire les droites dans le même sens.

*

* Création des surfaces

SURF1 = DALL P1P2 P2P3 P3P6 P6P1 ;

SURF2 = DALL P6P3 P3P4 P4P5 P5P6 ;

SURFO = SURF1 ET SURF2 ;

Pour réunir les deux sous-domaines on utilise l'opérateur ET. Cette opérateur permet de fusionner les objets de même type lorsque cela a un sens.

On pourrait mailler, par exemple, la surface SURF2 en utilisant des triangles. Pour cela, il suffit de changer le type d'élément par défaut après avoir créé la première surface SURF1 :

SURF1 = DALL P1P2 P2P3 P3P6 P6P1 ;

OPTI ELEM TRI3 ;

SURF2 = DALL P6P3 P3P4 P4P5 P5P6 ;

SURFO = SURF1 ET SURF2 ;

D'autres opérateurs permettent de mailler une surface. Ainsi, l'opérateur SURF maille l'intérieur d'un contour fermé. Le maillage de SURF1 à l'aide de l'opérateur SURF est obtenu par :

CONT1 = P1P2 ET P2P3 ET P3P6 ET P6P1 ;

SURF1 = SURF CONT1 ;

L'inconvénient des mailleurs automatiques, comme SURF, est qu'on contrôle difficilement la forme des mailles et leurs répartitions. En particulier, les mailles peuvent être déformées ; le maillage généré ne pas respecter certaines contraintes tant géométriques (symétrie, ...) que physiques (alignement avec l'écoulement, ...).

Pour tracer le maillage obtenu, il suffit d'utiliser l'opérateur TRAC. Un autre opérateur de maillage intéressant est l'opérateur CONT qui permet d'obtenir le contour d'un maillage.

TRAC SURFO ;

CONTO = CONT SURFO ;

TRAC CONTO ;

Fonction des coordonnées : Le maillage étant créé, nous allons initialiser un CHAMPOIN contenant une fonction bilinéaire dépendant des coordonnées de chacun des points du maillage.

L'opérateur COOR permet de récupérer les coordonnées des sommets — dans des objets de type CHAMPOIN — du maillage indiqué. Les opérateurs algébriques manipulent les CHAMPOINs comme des vecteurs : les opérations y sont



DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 11/65

Tutorial CAST3M pour la mécanique des fluides

faites selon les règles indiquées dans les notices, point par point, sans intervention particulière de l'utilisateur. En particulier, connaître le numéro d'un point du maillage ne sert à rien.

*

* Création d'un CHAMPOIN contenant la fonction $z = ax + by + c$

A1 = 2.DO ; B1 = 3.DO ; C0 = 5.DO ;

X0 Y0 = COOR SURFO ;

Z0 = A1*X0 + (B1*Y0) + C0 ;

TRAC Z0 SURFO CONTO ;

Là encore, les parenthèses sont importantes afin que l'évaluation de la fonction soit correcte du fait des règles de décodage d'une phrase GIBIANE — de gauche à droite et sans règle de priorité en l'absence de parenthèses. Le tracé de la fonction obtenue est réalisé avec TRAC, les isovaleurs étant automatiquement calculées. De nombreuses options permettent de modifier le rendu du tracé. Voir pour cela la notice des opérateurs TRAC, sans oublier celle d'OPTI.

On peut noter la présence du contour CONTO parmi les arguments de TRAC. La présence d'un deuxième maillage permet de supprimer le tracé du maillage au-dessus des isovaleurs du champ représenté. Essayez sans le contour CONTO pour apprécier la différence !

Evolution le long de la ligne médiane : L'opérateur EVOL permet d'associer les valeurs d'un CHAMPOIN aux abscisses curvilignes des points d'une ligne brisée passant par certains des points où le CHAMPOIN est défini. L'objet créé est un objet EVOLUTION qui peut être tracé à l'aide de la directive de dessin DESS.

*

* Evolution de $z = ax + by + c$ le long de l'axe médian

ZP6P3 = EVOL CHPO Z0 P6P3 ;

DESS ZP6P3 ;

De nombreuses options permettent de modifier le rendu du tracé. Voir pour cela la notice de DESS.

1.4 Complément

▷ P.Joly, Mise en oeuvre de la méthode des éléments finis, Coll. Mathématiques & Applications, SMAI vol2, Ellipses Ed., 1990.

1.5 Fichier Gibiane

OPTI DIME 2 ELEM QUA4 ;

*

* Variables globales

*

* LX0 : Dimension suivant x

* LY0 : Dimension suivant y

* XMIN,YMIN : Coordonnées du coin inférieur

* NX0 : Nombre de points suivant x

* NY0 : Nombre de points suivant y

*

LX0 = 2.DO ; LY0 = 1.DO ;



DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 12/65

Tutorial CAST3M pour la mécanique des fluides

```
XMIN = 0.DO ; YMIN = 0.DO ;
NX0 = 3 ; NY0 = 5 ;
XMAX = XMIN + LX0 ; YMAX = YMIN + LY0 ;
YMID = YMAX + YMIN / 2.DO ;
*
* Création des points
P1 = XMIN YMIN ;
P2 = XMAX YMIN ;
P3 = XMAX YMID ;
P4 = XMAX YMAX ;
P5 = XMIN YMAX ;
P6 = XMIN YMID ;
*
* Création des droites
P1P2 = P1 DROI NX0 P2 ;
P2P3 = P2 DROI NY0 P3 ;
P3P4 = P3 DROI NY0 P4 ;
P4P5 = P4 DROI NX0 P5 ;
P5P6 = P5 DROI NY0 P6 ;
P6P1 = P6 DROI NY0 P1 ;
P3P6 = P3 DROI NX0 P6 ;
P6P3 = INVE P3P6 ;
*
* Création des surfaces
SURF1 = DALL P1P2 P2P3 P3P6 P6P1 ;
'OPTION' 'ELEM' TRI3 ;
SURF2 = DALL P6P3 P3P4 P4P5 P5P6 ;
SURFO = SURF1 ET SURF2 ;
CONTO = CONT SURFO ;
*
* Création d'un CHAMPOIN contenant la fonction  $z = ax + by + c$ 
A1 = 2.DO ; B1 = 3.DO ; C0 = 5.DO ;
X0 Y0 = COOR SURFO ;
Z0 = A1*X0 + (B1*Y0) + C0 ;
*
* Evolution de  $z = ax + by + c$  le long de l'axe médian
ZP6P3 = EVOL CHPO Z0 P6P3 ;
*
* Tracé
*
TRAC SURFO ;
TRAC Z0 SURFO CONTO ;
DESS ZP6P3 ;
*
* ARRET
FIN ;
```



DEN

DM2S

Tutorial CAST3M pour la mécanique des fluides




FIG. 1 – Maillage




FIG. 2 – CHAMPOIN bilinéaire $f(x,y)=2x+3y+5$



DEN

DM2S

Tutorial CAST3M pour la mécanique des fluides




FIG. 3 – Evolution du CHAMPOIN $f(x,y)=2x+3y+5$ le long de l'axe $y=0.5$



DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 15/65

Tutorial CAST3M pour la mécanique des fluides

2 Objet modèle

2.1 Objectif

L'objectif de ce TP est

- ▷ de mailler une boîte rectangulaire présentant sur une de ses faces une partie semi-circulaire;
- ▷ de créer un modèle NAVIER_STOKES;
- ▷ de créer un vecteur et de le tracer;
- ▷ de transformer des CHAMPOINS des centres aux sommets et inversement;
- ▷ de sauver et restituer des données.

2.2 Consignes

Vous devez

1. réaliser et tracer le maillage d'un rectangle de dimension $6R * 5R$, où R désigne le rayon d'une protubérance semi-circulaire présente au centre de la face inférieure de la boîte. Le centre du cercle est l'origine des coordonnées;
2. définir un objet modèle NAVIER_STOKES sur le domaine maillé;
3. utiliser l'opérateur DOMA qui permet de nommer tous les maillages nécessaires à la résolution d'un problème fluide;
4. manipuler des vecteurs ainsi qu'un certain nombre d'opérateurs de projection. En particulier, initialisez le vecteur vitesse de composante $(-y,x)$;
5. sauvegarder des calculs et les récupérer.

Pour cela, vous aurez à utiliser les opérateurs de la fois dernière — INFO; AIDE; FIN; OPTI; '+'; '-' ; '*' ; '/' ; DROI; INVE; ET; DALL; SURF; COOR; CONT; TRAC; EVOL; DESS. — ainsi que CERC ou CER3; CHAN; DOMA; NOMC et/ou EXCO; KCHT; VECT; ELNO; NOEL; SAUV; REST.

2.3 Correction

Maillage : Par rapport à la fois dernière, la principale difficulté provient de la présence d'une partie semi-circulaire. Deux opérateurs permettent de réaliser un arc de cercle : ce sont les opérateurs CERC et CER3.

La lecture de la notice indique qu'ils nécessitent soit trois points de l'arc de cercle (CER3), soit deux points de l'arc de cercle et le centre du cercle (CERC). Dans ce corrigé, nous utiliserons CER3.

La partie inférieure du maillage sera construite en trois parties, la partie centrale étant réalisée à l'aide de l'opérateur CER3.

```
OPTI DIME 2 ELEM QUA4 ;
```

```
*
```

```
* Variables globales
```

```
*
```

```
* LXO : Dimension suivant x
```

```
* LYO : Dimension suivant y
```

```
* XMIN,YMIN : Coordonnées du coin inférieur gauche
```



DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 16/65

Tutorial CAST3M pour la mécanique des fluides

```
* NX0 : Nombre de points suivant x
* NY0 : Nombre de points suivant y
*
R0 = 1.D0 ; LX0 = 6.D0 * R0 ; LY0 = 5.D0 * R0 ;
XMIN = LX0 / 2.D0 * -1.D0 ; YMIN = 0.D0 ;
XMAX = LX0 / 2.D0 ; YMAX = LY0 ;
NX1 = 4 ; NX2 = 6 ; NX3 = NX1 + NX1 + NX2 ; NY0 = 5 ;
*
* Création des points
P1 = XMIN YMIN ;
P2 = (R0 * -1.) YMIN ;
P3 = 0.D0 R0 ;
P4 = R0 YMIN ;
P5 = XMAX YMIN ;
P6 = XMAX YMAX ;
P7 = XMIN YMAX ;
*
* Création des droites
P1P2 = P1 DROI NX1 P2 ;
P2P4 = CER3 NX2 P2 P3 P4 ;
P4P5 = P4 DROI NX1 P5 ;
P5P6 = P5 DROI NY0 P6 ;
P6P7 = P6 DROI NX3 P7 ;
P7P1 = P7 DROI NY0 P1 ;
*
* Création du maillage
SURFO = DALL (P1P2 ET P2P4 ET P4P5) P5P6 P6P7 P7P1 ;
CONTO = CONT SURFO ;
```

Modèle NAVIER_STOKES : L'objet modèle, créé par l'opérateur MODE, permet de transmettre aux opérateurs de CAST3M un ensemble d'informations nécessaires au contrôle des données. En mécanique des fluides, le maillage à transmettre à l'opérateur MODE est un maillage constitué d'éléments dits QUAF — comme QUAdratique pour les Fluides. En pratique, le maillage de QUAF est obtenu à partir du maillage constitué des éléments classiques de la mécanique — TRI3, QUA4, TRI6 et/ou QUA8 en 2D — à l'aide de l'opérateur CHANger. Il faut donc systématiquement coder les phrases GIBIANE suivantes après avoir construit le maillage :

```
*
* Création du modèle NAVIER-STOKES
DISCR = 'LINE' ;
SURF1 = CHAN SURFO QUAF ;
MOD1 = MODE SURF1 'NAVIER_STOKES' DISCR ;
```

La variable DISCR identifie le type d'éléments souhaité pour discrétiser les différentes inconnues des équations de Navier-Stokes. Avec LINE, on choisit un élément linéaire en vitesse ; MACRO aurait permis, dans le cadre de la résolution des équations de Navier-Stokes pour un écoulement incompressible, d'éviter l'apparition de modes de pression parasites tout en conservant une discrétisation linéaire de la vitesse ; QUAF indiquerait un élément quadratique pour l'inconnue de vitesse. Il est à noter que dans le cas d'une discrétisation linéaire LINE, certains points du maillage SURF1 transmis à l'opérateur ne sont plus utilisés par la suite.



DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 17/65

Tutorial CAST3M pour la mécanique des fluides

Pour l'instant, la discrétisation de la pression n'est pas indiquée dans le modèle. Nous verrons, qu'à nouveau, on dispose de trois possibilités : **CENTRE** et **CENTREP1** conduisent respectivement à une représentation constante et linéaire de la pression sur chaque élément du maillage, la pression étant discontinue à la face commune de deux éléments adjacents ; **MSOMMET** enfin permet une discrétisation linéaire et continue de la pression sur le maillage.

Nous verrons dans un prochain TP que tous les couples en vitesse/pression ne sont pas possibles, le plus souvent pour des questions de stabilité/compatibilité ou de précision.

Pour en finir avec la discrétisation des champs, signalons que tous les champs autre que la pression — en particulier la température — sont discrétisés comme la vitesse. Lorsqu'on spécifie l'espace de discrétisation de la vitesse, on spécifie donc aussi la plupart des espaces discrets (tous sauf la pression).

Les maillages pour la mécanique des fluides : En créant l'objet modèle Navier-Stokes, des maillages ont été créés. Ils sont accessibles par l'opérateur **DOMA** dont nous allons ici préciser les options les plus utilisées. La syntaxe générale de cet opérateur est :

```
RES1 = DOMA MOD1 MOT1 ;
```

où **MOT1** désigne le nom de l'option retenue.

Si **MOT1**=**'MAILLAGE'**, l'objet **RES1** contient le maillage. Dans le cas où **DISCR**=**'LINE'**, le maillage **RES1** a la même finesse que le maillage initial **SURFO**. Nous verrons que lorsque **DISCR**=**'MACRO'**, le maillage est deux fois plus fin dans chacune des directions suite à un redécoupage du maillage initial **SURFO**. C'est pourquoi il est prudent de considérer que le maillage au sens classique est celui obtenu via l'option **'MAILLAGE'** de **DOMA** et non pas celui créé initialement par l'utilisateur. Plus généralement, il faut écraser les maillages initiaux — ceux créés avant la construction du modèle — par ceux obtenus via **DOMA**.

Les maillages de classe de points d'un élément — c'est à dire les points situés au centre de gravité de l'élément, au centre de gravité des faces ou aux sommets de l'élément — peuvent être nommés en utilisant les options **'CENTRE'**, **'FACE'** et **'SOMMET'** de **DOMA**. Ces maillages sont constitués d'éléments de type **POI1** — nom donné dans **CAST3M** à l'élément constitué d'un point. Pour s'en convaincre, il suffit de les tracer. Par exemple :

```
RES1 = DOMA MOD1 'CENTRE' ;  
TRAC RES1 ;
```

Création et manipulation d'un vecteur : Nous avons créé durant le précédent TP un **CHAMPOIN** qui contenait pour chaque point du maillage une fonction bilinéaire de ses coordonnées. Pour stocker les vecteurs, par exemple la vitesse du fluide, nous allons créer un **CHAMPOIN** à plusieurs composantes.

Pour cela, nous allons créer un **CHAMPOIN** pour chacune des composantes de la vitesse. L'opérateur **ET** permettra de fusionner les deux **CHAMPOINs** en un seul. Durant cette fusion, nous ne voulons pas que les valeurs de chacune des composantes en un point soient additionnées. Afin d'éviter cette sommation, il existe dans les **CHAMPOINs** un identifiant appelé « nom de la composante ». Par défaut, le nom de la composante d'un **CHAMPOIN** est **SCAL**. Par convention, le nom de la première composante d'un vecteur (resp. deuxième et troisième) est **UX** (resp. **UY** et **UZ**). L'opérateur qui permet de modifier le nom de la composante d'un **CHAMPOIN** est **NOMC** (voir aussi **EXC0**).

Dans l'exemple suivant, nous allons créer un champ de vitesse dont la composante suivant l'axe x (resp. y) est $-y$ (resp. x) :

```
MAIL0 = DOMA MOD1 'MAILLAGE' ;  
X0 Y0 = COOR MAIL0 ;  
UX0 = NOMC 'UX' (-1. * Y0) ;
```



DEN

DM2S

SFME/LTME/R1/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 18/65

Tutorial CAST3M pour la mécanique des fluides

UYO = NOMC 'UY' XO ;
VO = UXO ET UYO ;

Pour tracer VO, il faut transformer le CHAMPOIN à deux composantes en un objet VECTEUR :

```
VECT1 = VECT 0.5 VO 'UX' 'UY' ;
TRAC VECT1 MAIL0 'TITR' 'Vitesse' ;
```

Les paramètres de l'opérateur VECT sont un coefficient de normalisation qui permet d'ajuster la longueur des flèches au moment du tracé, le CHAMPOIN à transformer en VECTEUR et la liste des noms de composantes du CHAMPOIN.

Dans le cas d'un CHAMPOIN constant par zone, l'opérateur KCHT permet de limiter le nombre de phrases GIBIANE. Le vecteur constant (0.2 ; 0.5) défini aux sommets du maillage est créé par

```
VSO = KCHT MOD1 'VECT' 'SOMMET' (0.2 0.5) ;
```

L'opérateur KCHT permet aussi d'initialiser des CHAMPOINs définis sur les classes de points 'CENTRE' et 'FACE'. Ces CHAMPOINs peuvent aussi être des CHAMPOINs scalaires à une composante. Dans ce cas on remplacera le mot clef 'VECT' par 'SCAL'. Ainsi, un champ constant sur le domaine et valant 5. sur chaque élément est créé par :

```
SCO = KCHT MOD1 'SCAL' 'CENTRE' 5. ;
```

Cet opérateur a deux autres grandes vertus :

- ▷ Il permet, sur le plan informatique, de gagner en rapidité en imposant le pointeur du maillage sous-tendant le modèle NAVIER_STOKES. Ainsi les tests internes aux opérateurs sont simplifiés. A l'usage, vous rencontrerez des messages d'erreur concernant des problèmes de pointeur. Pour un CHAMPOIN scalaire défini aux sommets du maillage sur lequel le modèle MOD1 a été construit on impose le pointeur de maillage par :

```
SSO = KCHT MOD1 'SCAL' 'SOMMET' SSO ;
```

- ▷ Il permet de surcharger les valeurs d'un CHAMPOIN par celles contenues dans un deuxième CHAMPOIN. Ce dernier peut n'être défini que sur une partie du maillage. C'est ainsi qu'on pourra très simplement modifier les conditions aux limites lorsqu'elles varient au cours du temps sur une partie de la frontière du domaine.

```
CL1 = KCHT MOD1 'SCAL' 'SOMMET' 0. CHP1 CHP2 ;
```

avec en plus une initialisation à 0. de CL1 sur l'ensemble des points, CHPi étant les CHAMPOINs préalablement construits par l'utilisateur.

Projection de CHAMPOIN : supports CENTRE et SOMMET. Nous verrons en cours que le support d'un CHAMPOIN caractérise l'espace vectoriel discret auquel appartient le champ discrétisé. Passer du support CENTRE au support SOMMET — et inversement — c'est faire une projection d'un espace dans un autre.

Un CHAMPOIN CENTRE permet de «stocker» un champ par l'intermédiaire de ses valeurs moyennes en chaque élément du maillage. Un CHAMPOIN SOMMET permet d'évaluer un champ en n'importe quel point d'un élément en interpolant les valeurs du champ exprimées aux sommets de l'élément tout en assurant la continuité de l'approximation à la traversée de la frontière entre deux éléments. Pour les éléments associés au modèle NAVIER_STOKES, les fonctions d'interpolation sont linéaires — discrétisations LINE et MACRO — ou quadratiques — discrétisation QUAF.

Il peut être intéressant de passer des centres aux sommets et inversement : pour initialiser une donnée, pour du post-traitement, etc.

Pour illustrer ces manipulations, nous allons calculer la norme du champ de vitesse créée précédemment, calculer sa moyenne par élément — passage SOMMETs → CENTREs — ; revenir aux SOMMETs ; comparer la norme initiale à la norme obtenue après le passage SOMMET → CENTRE → SOMMET.



DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 19/65

Tutorial CAST3M pour la mécanique des fluides

```

MOT1 = 'MOTS' 'UX' 'UY' ;
NORMO = ('PSCA' VO VO MOT1 MOT1) '**' 0.5 ;
NORMO = 'KCHT' MOD1 'SCAL' 'SOMMET' NORMO ;
NCO = 'NOEL' MOD1 NORMO ;
NSO = 'ELNO' MOD1 NCO ;
DIFF1 = NORMO '-' NSO / NORMO ;
'TRAC' DIFF1 MAILO
      'TITR' 'Erreur relative : r et ELNO(NOEL(r))' ;

```

Si à l'oeil nu on ne constate des différences qu'au voisinage des éléments les plus déformés, le tracé de l'erreur relative montre aussi des différences au bord du domaine. Dans le cas de fonctions moins monotones que celle utilisée ici — par exemple une sinusoïde — un enchaînement de ELNO/NOEL lisse la fonction, rabote les extrema.

Lecture et écriture sur fichier : On peut sauvegarder sur disque les objets créés — en particulier les résultats de calcul — afin de les utiliser plus tard.

Une utilisation possible des fonctionnalités du langage de commande GIBIANE consiste à séparer physiquement un jeu de données en trois fichiers d'extension dgibi. Le premier construira le maillage et le sauvegardera sur disque. Dans le deuxième, on relira un des maillages et on codera la modélisation physique et numérique du problème traité ainsi que sa résolution. Les calculs y seront sauvés sur disque. Dans un troisième fichier enfin, on réalisera le post-traitement.

Cette stratégie permet la mise au point du post-traitement à l'aide des résultats obtenus sur un maillage grossier pendant qu'un gros calcul tourne par ailleurs. De plus on évite de dupliquer les modèles physiques et numériques réalisés.

Il est possible de ne sauvegarder qu'une partie des objets présents en mémoire. Cela permet, soit disant, d'économiser de l'espace disque. A l'usage, on s'aperçoit qu'il vaut mieux sauvegarder l'ensemble des objets. En effet, dans le cas d'une sauvegarde partielle, l'expérience montre qu'il arrive assez souvent de souhaiter se servir après coup d'un des objets que l'on n'a pas sauvé. De plus, la différences de taille de fichiers entre des sauvegardes partielle et totale n'est pas très grande. A cause des dépendances entre objets on sauve bien plus que ce que l'utilisateur spécifie. Nous ne présentons donc ici que la sauvegarde complète.

Les opérateurs SAUV et REST réalisent les opérations d'écriture et de lecture sur disque. Au préalable, les fichiers auront été précisés dans les options, par l'opérateur OPTI. Le nom du fichier doit être l'adresse absolue. Il ne faut pas mettre d'adresse relative. Pour sauver :

```

'OPTI' 'SAUV' '/chemin/monfichierdesauvegarde' ;
...
'SAUV' ;

```

Pour lire :

```

'OPTI' 'REST' '/chemin/monfichierdesauvegarde' ;
...
'REST' ;

```

Le fait de mettre le nom du fichier entre quote est important : c'est le seul moyen de faire respecter la casse — distinction entre majuscule et minuscule — ce qui est vital si on souhaite que le chemin vers le fichier soit reconnu.

Pour les calculs dépendant du temps, la sauvegarde incrémentale permet de ne sauver que les objets dont les pointeurs ont été modifiés depuis la dernière sauvegarde. Consultez la notice pour de plus amples informations.



DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 20/65

Tutorial CAST3M pour la mécanique des fluides

2.4 Complément

Si on augmente le nombre de points sur l'arc de cercle, les éléments situés à la jonction avec l'axe deviennent très distordus. Avez-vous une idée pour corriger ce défaut ?

2.5 Fichier Gibiane

```
OPTI DIME 2 ELEM QUA4 ;
*
* Variables globales
*
* LX0 : Dimension suivant x
* LY0 : Dimension suivant y
* XMIN,YMIN : Coordonnées du coin inférieur
* NX0 : Nombre de points suivant x
* NY0 : Nombre de points suivant y
*
R0 = 1.DO ; LX0 = 6.DO * R0 ; LY0 = 5.DO * R0 ;
XMIN = LX0 / 2.DO * -1.DO ; YMIN = 0.DO ;
XMAX = LX0 / 2.DO ; YMAX = LY0 ;
NX1 = 4 ; NX2 = 6 ; NX3 = NX1 + NX1 + NX2 ; NY0 = 5 ;
*
* Création des points
P1 = XMIN YMIN ;
P2 = (R0 * -1.) YMIN ;
P3 = 0.DO R0 ;
P4 = R0 YMIN ;
P5 = XMAX YMIN ;
P6 = XMAX YMAX ;
P7 = XMIN YMAX ;
*
* Création des droites
P1P2 = P1 DROI NX1 P2 ;
P2P4 = CER3 NX2 P2 P3 P4 ;
P4P5 = P4 DROI NX1 P5 ;
P5P6 = P5 DROI NY0 P6 ;
P6P7 = P6 DROI NX3 P7 ;
P7P1 = P7 DROI NY0 P1 ;
*
* Création du maillage
SURFO = DALL (P1P2 ET P2P4 ET P4P5) P5P6 P6P7 P7P1 ;
CONTO = CONT SURFO ;
*
* Création du modèle NAVIER-STOKES
DISCR = 'LINE' ;
SURF1 = CHAN SURFO QUAF ;
MOD1 = MODE SURF1 'NAVIER_STOKES' DISCR ;
*
```



DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 21/65

Tutorial CAST3M pour la mécanique des fluides


```
* Tracé du maillage CENTRE
RES1 = DOMA MOD1 'CENTRE' ;
TRAC RES1 ;
*
* Manipulation d'un CHAMPOIN à deux composantes
* qui pourrait être un champ de vitesse
MAILO = DOMA MOD1 'MAILLAGE' ;
XO YO = COOR MAILO ;
UXO = NOMC 'UX' (-1. * YO) ;
UYO = NOMC 'UY' XO ;
VO = UXO ET UYO ;
VECT1 = VECT 0.5 VO 'UX' 'UY' ;
TRAC VECT1 MAILO 'TITR' 'Vitesse' ;
*
* Projection CENTRE-SOMMET de la norme de la vitesse
MOT1 = 'MOTS' 'UX' 'UY' ;
NORMO = ('PSCA' VO VO MOT1 MOT1) '**' 0.5 ;
NORMO = 'KCHT' MOD1 'SCAL' 'SOMMET' NORMO ;
NCO = 'NOEL' MOD1 NORMO ;
NSO = 'ELNO' MOD1 NCO ;
DIFF1 = NORMO '-' NSO / NORMO ;
'TRAC' NORMO MAILO CONTO 'TITR' 'Norme de la vitesse' ;
'TRAC' DIFF1 MAILO
'TITR' 'Erreur relative : r et ELNO(NOEL(r))' ;
*
* Sauvegarde sur disk
OPTI SAUV 'tp2.sauv' ;
SAUV ;
*
* ARRET
*FIN ;
```



DEN


DM2S

Tutorial CAST3M pour la mécanique des fluides



Maillage et points CENTRE

FIG. 4 – Maillage et points CENTRE



Vecteur vitesse et sa norme

FIG. 5 – Vitesse et sa norme



DEN

DM2S

Tutorial CAST3M pour la mécanique des fluides




FIG. 6 – Erreur relative entre r et ELNO(NOEL(r))



DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 24/65

Tutorial CAST3M pour la mécanique des fluides

3 Equation de la chaleur

3.1 Objectif et rappels

L'étude de la diffusion d'une variable scalaire permet de présenter les méthodes numériques employées en mécanique des fluides et d'en cerner certaines limites.

Le problème modèle associé aux transferts diffusifs est l'équation de la chaleur. On s'intéressera dans ce TP à la solution stationnaire de cette équation. Deux méthodes permettent de résoudre ce problème :

- ▷ une résolution directe de l'équation de la chaleur écrite en régime permanent,
- ▷ une résolution temporelle où la solution stationnaire est obtenue en tant que limite asymptotique de l'équation de la chaleur.

Nous préciserons les conditions d'utilisation de chacune de ces techniques.

Afin de mener cette étude, deux solutions particulières de l'équation de la chaleur sont exhibées :

- ▷ une solution bilinéaire,
- ▷ une sinusoïde amortie.

La solution bilinéaire permettra d'apprécier l'erreur d'approximation par comparaison avec la solution calculée. Pour cela, différents maillages seront utilisés : un maillage régulier de quadrangles et un maillage régulier de triangles. L'utilisation d'un maillage irrégulier de quadrangles mettra en évidence un autre type d'erreur liée à l'utilisation de méthodes d'intégration numérique.

La solution sinusoïde amortie permettra d'estimer l'ordre de convergence du schéma. Pour cela, plusieurs calculs seront réalisés en raffinant le maillage. L'ordre du schéma sera visualisé en traçant l'évolution de l'erreur en fonction du pas de discrétisation.

Le jeu de données pour CAST3M (en langage gibiane) se trouve à la fin du TP.

3.1.1 Equation de la chaleur

Cette équation modélise les transferts thermiques par conduction et s'écrit

$$\rho c_p \frac{\partial T}{\partial t} = \text{div}(\lambda \vec{\nabla} T)$$

où T désigne la température (en K); ρ la densité (en kg/m^3); c_p la chaleur spécifique à pression constante (en $J/kg/K$); λ la conductivité thermique (en $W/m/K$).

En supposant que les propriétés matérielles sont indépendantes du temps et de l'espace et que le matériau est isotrope, l'équation de la chaleur s'écrit

$$\frac{\partial T}{\partial t} = \alpha \Delta T$$

où $\alpha = \lambda/(\rho c_p)$ désigne la diffusivité thermique (en m^2/s).

Pour pouvoir résoudre ce problème, on doit connaître le champ de température à l'instant initial ainsi que l'évolution au cours du temps de la température (ou de son flux) à la frontière du domaine — le problème est elliptique en espace et parabolique en temps.

Si on s'intéresse au régime permanent, T est solution de

$$\Delta T = 0$$



Tutorial CAST3M pour la mécanique des fluides

Le problème est alors bien posé à condition de connaître les valeurs de la température (ou de son flux) à la frontière du domaine à l'état stationnaire — caractéristique d'un problème elliptique en espace. Si le flux est imposé sur l'ensemble de la frontière la solution est connue à une constante près. En instationnaire par contre, si le flux est imposé sur l'ensemble de la frontière la solution est parfaitement connue, les conditions initiales fixant la valeur de la constante.

3.1.2 Solutions particulières de l'équation de la chaleur

Pour les besoins de ce TP, nous avons exhibé deux solutions stationnaires particulières de l'équation de la chaleur :

- ▷ une solution bilinéaire,
- ▷ une sinusoïde amortie.

Le domaine considéré dans nos calculs est une plaque carrée homogène de côté 1 ($(x, y) \in [0 ; 1]^2$). Les conditions aux limites seront dans chaque cas les restrictions à la frontière de la solution à l'état stationnaire. Pour les méthodes faisant intervenir le temps, à l'instant initial, la température sera supposée uniformément nulle.

Solution bilinéaire

L'équation stationnaire de la chaleur est vérifiée pour :

$$T(x, y) = ax + by + cxy + d$$

On prendra pour les applications numériques $a = 1$, $b = 1$, $c = 2$ et $d = 3$. Cette solution est visualisée (Fig 7). A la frontière du domaine, les conditions aux limites sont donc :

$$T(x, y) = \begin{cases} ax + d & \text{pour } (x, y) \in [0 ; 1] \times 0 \\ (a + c)x + (d + b) & \text{pour } (x, y) \in [0 ; 1] \times 1 \\ by + d & \text{pour } (x, y) \in 0 \times [0 ; 1] \\ (b + c)y + (d + a) & \text{pour } (x, y) \in 1 \times [0 ; 1] \end{cases}$$

Sinusoïde amortie

L'équation stationnaire de la chaleur est vérifiée pour :

$$T(x, y) = a e^{-by} \sin(bx) + d$$

On prendra pour les applications numériques $b = \pi/2$, $a = 1$ et $d = 0.2$. Cette solution est visualisée (Fig 8). A la frontière du domaine, les conditions aux limites sont donc :

$$T(x, y) = \begin{cases} a \sin(bx) + d & \text{pour } (x, y) \in [0 ; 1] \times 0 \\ a e^{-b} \sin(bx) + d & \text{pour } (x, y) \in [0 ; 1] \times 1 \\ d & \text{pour } (x, y) \in 0 \times [0 ; 1] \\ a e^{-by} \sin(b) + d & \text{pour } (x, y) \in 1 \times [0 ; 1] \end{cases}$$

3.1.3 Recherche d'une solution stationnaire

Lorsqu'il existe une solution stationnaire à un problème transitoire et que seule cette solution nous intéresse, on peut chercher à discrétiser directement les équations écrites en régime permanent ou considérer le problème transitoire et itérer en temps jusqu'à l'obtention de l'état stationnaire.

Méthode stationnaire

Tutorial CAST3M pour la mécanique des fluides

L'algorithme le plus naturel pour calculer la solution stationnaire d'un problème transitoire consiste à discrétiser le problème permanent.

La discrétisation des équations conduit à un système matriciel de la forme $Ax = b$. Si le problème traité est *linéaire*, la matrice A est à coefficients constants. Il suffit de l'inverser après avoir imposé les conditions aux limites pour obtenir la solution stationnaire.

Dans le cas où le problème est *non linéaire*, la matrice n'est plus à coefficients constants. Ceux-ci dépendent de la solution :

$$A(x)x = b.$$

Pour l'équation de la chaleur, le problème est non linéaire lorsque l'une des propriétés matérielles dépend de la température. Une méthode de Picard — de Newton, de point fixe, *etc.* — est alors utilisée afin de linéariser la forme matricielle obtenue à l'issue de la discrétisation :

- Initialiser x_0
- Pour $k > 0$
 - Calculer $\delta x = ||x_{k+1} - x_k||$
 - Tant que $\delta x \leq \epsilon$
 - Résoudre $A(x_k)x_{k+1} = b(x_k)$

La convergence de la méthode vers la solution physique est souvent problématique. Elle dépend fortement du choix du premier itéré x_0 et du conditionnement de la matrice. Plus x_0 est proche de la solution cherchée plus on a de chances de converger. Dans de nombreux cas, une initialisation proche de la solution cherchée est impossible ce qui conduit à la divergence de la méthode. C'est pourquoi, dans le cas de problèmes non-linéaires, on recherche l'état stationnaire en tant que limite asymptotique du problème instationnaire. En introduisant le temps, on modifie la nature mathématique du problème ce qui permet dans certains cas d'avoir un problème mieux posé ; en jouant sur la valeur du pas de temps le conditionnement peut être amélioré et la dominance de la diagonale renforcée.

Méthode instationnaire

Supposons la solution connue au temps $t_n = n\Delta t$. On cherche à évaluer la solution de l'équation de la chaleur au temps $t_{n+1} = (n+1)\Delta t$. Suivant le temps auquel on évalue le Laplacien de température, on obtient un schéma *explicite* en temps ou un schéma *implicite* en temps. Nous présentons dans cette partie ces deux schémas en utilisant une discrétisation spatiale et temporelle par différences finies.

Schéma explicite. Dans le cas de l'équation de la chaleur monodimensionnelle, on obtient le schéma explicite en exprimant le Laplacien en début de pas de temps (au temps t_n) :

$$\frac{T_i^{n+1} - T_i^n}{\Delta t} = \alpha \frac{T_{i+1}^n - 2T_i^n + T_{i-1}^n}{\Delta x^2}$$

où Δt et Δx sont respectivement les pas de temps et d'espace et T_i^n désigne la température au point $x_i = i\Delta x$ et au temps $t_n = n\Delta t$.

L'intérêt de la méthode explicite est évident : la solution au temps t_{n+1} est connue explicitement à partir de la solution au temps t_n — sans inversion d'une matrice. Le nombre d'opérations algébriques permettant cette évaluation est donc faible ce qui permet d'effectuer beaucoup de pas de temps à moindre coût.

Malheureusement, pour un maillage de pas Δx , les valeurs de pas de temps sont limitées par le critère de Fourier

$$2\alpha\Delta t < \Delta x^2$$

Pour un maillage fin, le pas de temps peut donc être très petit ce qui peut conduire à un nombre de pas de temps très important pour atteindre l'état stationnaire et à des temps de calcul très long.

**Tutorial CAST3M pour la mécanique des fluides**

Schéma implicite. Dans le cas de l'équation de la chaleur monodimensionnelle, on obtient le schéma implicite en exprimant le Laplacien en fin de pas de temps (au temps t_{n+1}) :

$$\frac{T_i^{n+1} - T_i^n}{\Delta t} = \alpha \frac{T_{i+1}^{n+1} - 2 T_i^{n+1} + T_{i-1}^{n+1}}{\Delta x^2}$$

L'intérêt du schéma implicite par rapport au schéma explicite est qu'il n'y a pas de condition limitant la valeur supérieure du pas de temps. En éléments finis, si on ne diagonalise pas la matrice masse, on peut montrer que le pas de temps doit être assez grand afin d'éviter certains problèmes de stabilité — en particulier perte du principe du maximum :

$$2\alpha\Delta t > \Delta x^2/6.$$

Comme en général on considère de grands pas de temps au regard du temps caractéristique du phénomène étudié, cette contrainte est automatiquement vérifiée. C'est pourquoi, par abus de langage, on dit qu'il n'existe pas de contrainte sur la valeur du pas de temps en implicite.

Dans le cas d'un problème linéaire, en faisant tendre Δt vers l'infini, on montre qu'il suffit d'un pas de temps pour obtenir l'état stationnaire.

Dans le cas où la diffusivité thermique dépend de la température, l'équation de la chaleur devient non-linéaire. La prise en compte de la non-linéarité impose un certain nombre de pas de temps avant d'atteindre la solution stationnaire. On peut linéariser le schéma en affectant à la diffusivité thermique sa valeur au début du pas de temps. Le fait de figer la conductivité thermique au pas de temps précédent, dégrade la précision de la température calculée durant le transitoire. Seule la solution stationnaire vérifie l'équation et est donc à considérer. Si on s'intéresse aussi au transitoire, à chaque pas de temps, on doit améliorer l'évaluation du flux ce qui conduit en général à utiliser une méthode de Picard.

Choix du schéma. Pour un problème linéaire la méthode stationnaire est souvent la plus efficace.

Pour un problème non-linéaire, le choix d'un schéma explicite ou implicite est empirique dans la mesure où c'est l'expérience de l'utilisateur ou, ce qui est moins scientifique, ses habitudes qui prévalent. Un schéma implicite nécessite en général moins de pas de temps pour atteindre l'état stationnaire qu'un schéma explicite — le pas de temps de ce dernier est limité par des contraintes de stabilité. Cependant, au delà d'une certaine taille de maillage, le coût d'un pas de temps implicite devient plus important que le coût d'un pas de temps explicite. La stratégie à adopter est donc à adapter au cas traité.

Dans tous les cas, il ne faut pas perdre de vue que la résolution d'un problème linéaire sert en général d'étape préliminaire au traitement d'un problème non linéaire dans lequel on a figé toutes les non-linéarités — en négligeant la dépendance des paramètres physiques à la température par exemple. Il faut donc faire attention à ne pas trop spécialiser le traitement numérique à un type de problème au risque de devoir tout refaire le jour où on abordera le vrai problème.

3.2 Consignes

Le jeu de données associé à ce TP comporte des menus qui permettent de choisir :

1. le maillage (composé de triangles, de quadrangles réguliers ou non) ;
2. le nombre d'éléments par coté ;
3. le degré des interpolations (linéaire ou quadratique) ;
4. la méthode de résolution (instationnaire explicite, instationnaire implicite, stationnaire) ;
5. dans le cas implicite, le pas de temps et la méthode d'inversion du système matriciel.



DEN

DM2S

SFME/LTMF/RT/03-035/A

Date : 15/09/03

RAPPORT DM2S

Page : 28/65

Tutorial CAST3M pour la mécanique des fluides

La solution codée est la solution sinusoïdale amortie. Après avoir comparé pour les différents paramètres modifiables par menu la solution calculée à la solution analytique, vous devez coder la solution bilinéaire et observer à nouveau le comportement de l'erreur.

Afin de vous aider dans l'analyse des résultats, nous traçons à l'issue du calcul le maillage utilisé, la solution exacte, la solution calculée, l'erreur absolue et l'erreur relative.

Le jeu de données associé à ce TP est l'archétype des jeux de données de mécanique des fluides : on construit le maillage à l'aide des opérateurs classiques de maillage, puis le modèle NAVIER-STOKES avec `MODE`. Viennent ensuite la description des équations à l'aide de `EQEX` et la résolution avec la procédure `EXEC`. On termine de façon classique par le post-traitement des résultats.

N'hésitez donc pas à étudier ce jeu de données afin d'en comprendre tous les rouages.

3.3 Correction

3.3.1 Influence du maillage

Le domaine peut être maillé en utilisant les éléments géométriques suivants :

- ▷ en 2D, des triangles ou des quadrangles ;
- ▷ en 3D, des tétraèdres, des prismes ou des hexaèdres.

Pour une discrétisation spatiale de l'équation de la chaleur par éléments finis — méthode de Galerkin — et en utilisant les éléments finis polynomiaux de Lagrange de plus bas degré afin d'approximer une fonction continue, intégrable et à dérivée intégrable, les valeurs calculées correspondent à l'approximation de la température aux sommets des éléments. A l'intérieur d'un élément, la température est évaluée par interpolation des valeurs de la température aux sommets de l'élément considéré, les fonctions d'interpolation étant les polynômes de Lagrange.

L'élément triangulaire de plus bas degré a trois sommets. La température est donc connue par :

$$T(x, y) = a + bx + cy$$

où a , b et c sont trois constantes déterminées à partir des valeurs de la température aux sommets de l'élément. On ne peut donc calculer de façon exacte que des solutions linéaires à l'aide d'un maillage constitué de triangles linéaires.

Respectivement, dans le cas de quadrangles linéaires, la température est connue par :

$$T(x, y) = a + bx + cy + dxy$$

où a , b , c et d sont quatre constantes déterminées à partir des valeurs de la température aux sommets de l'élément. Seules les solutions bilinéaires sont donc évaluées exactement lorsque le maillage est constitué de quadrangles linéaires.

Les résultats obtenus en utilisant la solution bilinéaire de l'équation de la chaleur permettent de confirmer ces résultats : la solution calculée avec un maillage régulier de quadrangles linéaires permet de retrouver la solution analytique — à la précision machine près — alors qu'une erreur est commise avec un maillage constitué de triangles linéaires (Fig 9).

Lorsque le maillage de quadrangles est quelconque (i.e. les éléments ne sont pas des parallépipèdes), on peut observer que la précision chute puisque l'erreur commise est du même ordre que celle obtenue avec un maillage triangulaire. Cette erreur n'est pas liée au degré des fonctions d'interpolation mais à une erreur commise lors de l'évaluation des coefficients de la matrice et du second membre par des formules de quadrature dans le cas de quadrangles déformés. Afin de réduire cette erreur, il suffit de construire des maillages contenant des éléments peu déformés.

Tutorial CAST3M pour la mécanique des fluides

3.3.2 Estimation de l'ordre de convergence

La solution sinusoïdale amortie va permettre d'évaluer l'ordre de convergence du schéma en espace.

En effet, cette solution n'étant pas bilinéaire, quel que soit le type d'éléments utilisés, on commet obligatoirement une erreur d'interpolation. Cette erreur, notée ϵ , est proportionnelle au pas du maillage Δx à une certaine puissance :

$$\epsilon = a\Delta x^p$$

où a est une constante et p l'ordre de l'approximation.

En prenant le logarithme de cette expression, on obtient

$$\log \epsilon = \log a + p \log \Delta x$$

Le logarithme du pas du maillage et le logarithme de l'erreur sont donc liés par une relation linéaire, la pente de la droite obtenue donnant l'ordre du schéma. L'erreur est évaluée en utilisant la norme L^2 discrète.

L'ordre de convergence obtenu pour un maillage de triangles en résolvant le problème stationnaire est de 2.33 (Fig 11).

Dans le cas où la solution stationnaire est obtenue comme solution asymptotique de l'équation de la chaleur, il est intéressant de tracer l'évolution de la différence entre deux itérés en fonction du nombre de pas de temps. On voit ici une diminution régulière du résidu (Fig 12). On observe également que plus le maillage est fin, plus la convergence est lente. Cela est normal. Le schéma utilisé ici est explicite. Le pas de temps de stabilité pour le maillage le plus fin est donc plus faible que celui du maillage grossier. Il faut donc plus de pas de temps pour atteindre l'état stationnaire avec le maillage fin.

3.3.3 Description du jeu de données

L'expérience a montré qu'il fallait laisser à l'utilisateur le soin de réaliser ses premiers pas seul. A trop vouloir guider le pêcheur, il n'apprend jamais à pêcher.

3.4 Complément

- ▷ Déterminer l'ordre de convergence pour un maillage de quadrangles en utilisant un maillage régulier, puis irrégulier, afin de mesurer l'effet des erreurs d'intégration.
- ▷ Modifier les conditions aux limites à la frontière du domaine afin d'imposer un flux.
- ▷ Que se passe-t-il si on «oublie» d'imposer les conditions aux limites ?
- ▷ A.Ern, J-L.Guermond, *Eléments finis : Théorie, applications, mise en oeuvre*, Mathématiques & Applications, Vol 36, Springer Ed., 2002.
- ▷ C.A.J.Fletcher, *Computational Techniques for Fluid Dynamics, Volume 1*, Springer Verlag Ed., 1988.
- ▷ M.D.Gunzburger, *Finite element methods for viscous incompressible flows : a guide to theory, practice and algorithms*, Academic Press Ed., 1989.
- ▷ C.Hirsch, *Numerical Computation of Internal and External Flows, Volume 1, Fundamentals of numerical discretization*, Wiley Ed., 1988.
- ▷ S.V.Patankar, *Numerical Heat Transfer and Fluid Flow*, Series in Comp. Meth. in Mech. and Thermal Sc., Hemisphere Pub, 1980.
- ▷ O.Pironneau, *Finite element methods for fluids*, John Wiley Ed., 1989.
- ▷ P.J.Roache, *Computational Fluid Dynamics*, Hermosa Publ., Albuquerque, 1972.
- ▷ T.M.Shih, *Numerical Heat Transfer*, Series in Comp. Meth. in Mech. and Thermal Sc., Hemisphere Pub, 1984.
- ▷ O.C. Zienkiewicz, R.L. Taylor, *The finite element method, Vol1 : The basis*, 5th Butterworth-Heinemann Ed., 2000.



Tutorial CAST3M pour la mécanique des fluides

3.5 Fichiers Gibiane

```
*
*****
* Cours MF307 - Tp3
* Diffusion d'un champ scalaire
* Solution stationnaire de l'équation de la chaleur
*****
*
*OPTI 'ECHO' 0 ;
*
-----
*DEBP 'CALCUL' ;
*
* Cette procédure, utilisée comme un opérateur, calcule la différence
* entre deux pas de temps toutes les 5 itérations.
* L'évolution de cette différence (erreur absolue) au cours du temps
* est conservée en vue d'un post-traitement.
-----
*ARGU 'RVX*'TABLE' ;
*
RV = RVX . 'EQEX' ;
TN = RV . 'INCO' . 'TN' ;
*
*SI' ( 'EXIS' RVX 'COMPT' ) ;
RVX . 'COMPT' = RVX . 'COMPT' + 1 ;
'SINO' ;
RVX . 'COMPT' = 1 ;
RV . 'INCO' . 'IT' = 'PROG' 1 ;
RV . 'INCO' . 'ER' = 'PROG' 0. ;
RV . 'INCO' . 'TBIS' = 'COPI' RV . 'INCO' . 'TN' ;
'FINS' ;
*
DD = RVX . 'COMPT' ;
NN = DD / 5 ;
LO = (DD-(5*NN)) 'EGA' 0 ;
'SI' LO ;
ERR = (RV . 'INCO' . 'TN') - (RV . 'INCO' . 'TBIS') ;
RV . 'INCO' . 'TBIS' = 'COPI' (RV . 'INCO' . 'TN') ;
ERRINF = 'MAXI' ERR 'ABS' ;
ELI = ('LOG' (ERRINF + 1.0E-50))/('LOG' 10.) ;
'MESS' 'ITERATION ' (RVX . 'COMPT') ' LOG10 ERREUR ' ELI ;
IT = 'PROG' RVX . 'COMPT' ;
ER = 'PROG' ELI ;
RV . 'INCO' . 'IT' = (RV . 'INCO' . 'IT') 'ET' IT ;
RV . 'INCO' . 'ER' = (RV . 'INCO' . 'ER') 'ET' ER ;
'FINS' ;
-----
*matr chp1 = 'KOPS' 'MATRIK' ;
*FINP' matr chp1 ;
-----
*
*
*
*
*****
*
* ACQUISITION DES OPTIONS FOURNIES PAR L'UTILISATEUR
*
*
*
*
*-----
*
* Choix du type d'éléments
*
'SAUT' 15 'LIGN' ;
'MESS' 'Choix des elements maillant le domaine :' ;
'MESS' ' 1 -> Triangles' ;
'MESS' ' 2 -> Rectangles' ;
'MESS' ' 3 -> Quadrilateres quelconques' ;
'OBTE' CHOIX ;
TEST1 = (CHOIX '>' 3) 'OU' (CHOIX '<' 1) ;
'SI' TEST1 ;
'MESS' 'Erreur lors de la saisie du choix.' ;
'FIN' ;
'FINS' ;
*
* Choix du type d'interpolation
*
'SAUT' 15 'LIGN' ;
'MESS' 'Choix des degrés des éléments finis :' ;
'MESS' ' 1 -> Degré 1' ;
'MESS' ' 2 -> Degré 2' ;
'OBTE' IDEGR ;
TEST1 = (IDEGR '>' 2) 'OU' (IDEGR '<' 1) ;
'SI' TEST1 ;
'MESS' 'Erreur lors de la saisie du choix.' ;
'FIN' ;
'FINS' ;
*
* Choix du nombre d'éléments
*

```

```
'SAUT' 15 'LIGN' ;
'MESS' 'Nombre d'elements par cote :' ;
'OBTE' NX ;
'SI' (NX '<' 1) ;
'MESS' 'Erreur lors de la saisie du choix.' ;
'FIN' ;
'FINS' ;
NY = NX ;
*
*- Choix de la méthode
*
'SAUT' 15 'LIGN' ;
'MESS' 'Type de methode de resolution :' ;
'MESS' ' 1 -> Instationnaire explicite' ;
'MESS' ' 2 -> Instationnaire implicite' ;
'MESS' ' 3 -> Stationnaire' ;
'OBTE' IMETH ;
TEST1 = (IMETH '>' 3) 'OU' (IMETH '<' 1) ;
'SI' TEST1 ;
'MESS' 'Erreur lors de la saisie du choix.' ;
'FIN' ;
'FINS' ;
*
'SI' (('EGA' IMETH 1) 'ET' ('EGA' IDEGR 2)) ;
'MESS' 'Seul degré 1 autorisé avec méthode explicite.' ;
'FIN' ;
'FINS' ;
*
*- Nombre de pas de temps
*
'SI' ('NEG' IMETH 3) ;
'SAUT' 15 'LIGN' ;
'MESS' 'Nombre de pas de temps :' ;
'OBTE' NITER ;
'SI' (NITER '<' 1) ;
'MESS' 'Erreur lors de la saisie du choix.' ;
'FIN' ;
'FINS' ;
*
*- Choix de la méthode d'inversion
*
'SAUT' 15 'LIGN' ;
'MESS' 'Type de resolution matricielle :' ;
'MESS' ' 1 -> Méthode directe (Crout)' ;
'MESS' ' 2 -> Gradient conjugué' ;
'MESS' ' 3 -> Bi gradient conjugué' ;
'OBTE' ININV ;
TEST1 = (ININV '>' 3) 'OU' (ININV '<' 1) ;
'SI' TEST1 ;
'MESS' 'Erreur lors de la saisie du choix.' ;
'FIN' ;
'FINS' ;
*
'SAUT' 15 'LIGN' ;
*
*****
*
* MALLAGE
*
*****
*
*- Initialisation des options
*
'SI' (CHOIX 'EGA' 1) ;
'OPTI' 'DIME' 2 'ELEM' 'TRI3' ;
'SINO' ;
'OPTI' 'DIME' 2 'ELEM' 'QUA4' ;
FINSI ;
*
*- Definition des sommets de la plaque
*
LX = 1.00 ;
LY = 1.00 ;
XMIN = 0.00 ;
YMIN = 0.00 ;
XMAX = XMIN + LX ;
YMAX = YMIN + LY ;
*
A1 = XMIN YMIN ;
A2 = XMAX YMIN ;
A3 = XMAX YMAX ;
A4 = XMIN YMAX ;
*
*- Definition des arêtes
*
'SI' (CHOIX 'EGA' 3) ;
R = 10. ** (1. / (NX - 1)) ;
DA = 1. * ((R - 1.) / (R ** NX - 1)) ;
DB = 10. * DA ;
FBAS = 'DROI' A1 A2 'DINI' DA 'DFIN' DB ;

```



DEN

DM2S

RAPPORT DM2S

Tutorial CAST3M pour la mécanique des fluides

```

FDRO = 'DROI' A2 A3 'DINI' DB 'DFIN' DA ;
FHAU = 'DROI' A3 A4 'DINI' DA 'DFIN' DB ;
FGAU = 'DROI' A4 A1 'DINI' DB 'DFIN' DA ;
'SINGO' ;
FBAS = A1 'DROI' NX A2 ;
FDRO = A2 'DROI' NY A3 ;
FHAU = A3 'DROI' NX A4 ;
FGAU = A4 'DROI' NY A1 ;
'FINS' ;
*
** Maillage de la plaque
*
SI (CHOIX EGA 1);
CNT = FBAS ET FDRO ET FHAU ET FGAU ;
MT = CNT 'SURF' 'PLAN' ;
SINON;
MT = 'DALLER' FBAS FDRO FHAU FGAU 'PLAN' ;
FINSI;
*
** Modèle NAVIER_STOKES
*
MT = 'CHANGER' QUAF MT ;
'SI' ('EGA' IDEGR 1);
$MT = modele MT 'NAVIER_STOKES' 'LINE';
SINON ;
$MT = modele MT 'NAVIER_STOKES' 'QUAF';
'FINSI';
MT = 'DOMA' $mt maillage ;
CNT = 'CONT' mt ;
*
*****
SOLUTION EXACTE
*****
**-> T(x,y) = 0.2 + sin(180*x) * e(-PI*y)
*
XX YY = 'COORD' MT ;
SINUS = 'SIN' ( XX * 180. ) ;
EXPON = 'EXP' ( YY * (0. - PI) ) ;
SOLEX = SINUS * EXPON * 0.2 ;
*
*****
SOLUTION CASTEM
*****
*
* La structure du bloc de chaque méthode est identique :
* -> description de la modélisation physique et numérique (EQEX)
* -> conditions aux limites (EQEX, option CLIM)
* -> initialisations des champs (table INCO)
* -> initialisations des données numériques (table RV)
* -> résolution (EXEC ou EXIC)
*
* Les données numériques sont à ranger dans la table de
* soustype EQEX aux indices suivants :
* NITER : Nombre d'itérations internes (cas non-linéaire)
* OMEGA : Coefficient de relaxation (cas non-linéaire)
* IMPR : Flag d'impression (1 pour oui, 0 pour non)
* ITMA : Nombre de pas de temps
* EPS : Critere de convergence pour les iterations internes
* DT : Pas de temps
*
*-----
* Méthode de résolution instationnaire explicite
* Utilisation de l'opérateur TSCAL
*-----
'SI' (IMETH 'EGA' 1) ;
RV = 'EQEX' $MT 'ITMA' NITER 'ALFA' 0.9
'ZONE' $MT 'OPER' 'CALCUL
'OPTI' 'EFM1' 'CENTREE' 'EXPL'
'ZONE' $MT 'OPER' 'TSCA' 1. 'UN' 0.0 'INCO' 'TN'
'OPTI' 'EFM1' 'CENTREE' 'EXPL'
'ZONE' $MT 'OPER' 'DFDT' 1. 'TN' 'DELTAT' 'INCO' 'TN'
;
RV = 'EQEX' RV 'CLIM' CNT 'TN' 'TIMP' ('REDU' SOLEX CNT) ;
RV . 'INCO' = 'TABLE' 'INCO' ;
RV . 'INCO' . 'TN' = 'KCHT' $MT 'SCAL' 'SOMMET' 0. ;
RV . 'INCO' . 'UN' = 'KCHT' $MT 'VECT' 'SOMMET' (0. 0.) ;
'FINS' ;
*
*-----
* Méthode de résolution instationnaire implicite
* Utilisation des opérateurs DFDT et LAPN
*-----
'SI' (IMETH 'EGA' 2) ;
RV = 'EQEX' $MT 'OPTI' 'EF' 'IMPL' 'CENTREE'

```

```

*
'ZONE' $MT 'OPER' 'DFDT' 1. 'TNM' 'DT' (0. 0.) 0. 'INCO' 'TN'
'ZONE' $MT 'OPER' 'DFDT' 1. 'TNM' 'DT' 'INCO' 'TN'
'ZONE' $MT 'OPER' 'LAPN' 1. 'INCO' 'TN'
'ZONE' $MT 'OPER' 'CALCUL
;
RV = 'EQEX' RV 'CLIM' CNT 'TN' 'TIMP' ('REDU' SOLEX CNT) ;
RV . 'INCO' = 'TABLE' 'INCO' ;
RV . 'INCO' . 'TN' = 'KCHT' $MT 'SCAL' 'SOMMET' 0. ;
RV . 'INCO' . 'TNM' = 'KCHT' $MT 'SCAL' 'SOMMET' 0. ;
RV . 'NITER' = 1 ;
RV . 'OMEGA' = 1.D0 ;
RV . 'IMPR' = 1 ;
RV . 'ITMA' = NITER ;
RV . 'EPS' = 1.D-15 ;
RV . 'DT' = 1.D-1 ;
RV . 'INCO' . 'DT' = RV . 'DT' ;
'FINS' ;
*
*-----
* Méthode de résolution stationnaire (implicite)
* Utilisation de l'opérateur LAPN
*-----
'SI' (IMETH 'EGA' 3) ;
RV = 'EQEX' $MT 'OPTI' 'EF' 'IMPL'
'ZONE' $MT 'OPER' 'LAPN' 1.D0 'INCO' 'TN' ;
RV = 'EQEX' RV 'CLIM' CNT 'TN' 'TIMP' ('REDU' SOLEX CNT) ;
RV . 'INCO' = 'TABLE' 'INCO' ;
RV . 'INCO' . 'TN' = 'KCHT' $MT 'SCAL' 'SOMMET' 0.D0 ;
RV . 'NITER' = 2 ;
RV . 'OMEGA' = 1.D0 ;
RV . 'IMPR' = 1 ;
RV . 'ITMA' = 1 ;
RV . 'EPS' = 1.D-15 ;
RV . 'DT' = 1.D3 ;
RV . 'INCO' . 'DT' = RV . 'DT' ;
'FINS' ;
*
* Options pour le solveur (voir KRES)
rv . 'METHIN' . 'TYPIN' = IMINV ;
rv . 'METHIN' . 'IMPIN' = 2 ;
rv . 'METHIN' . 'NITMAX' = 1500 ;
rv . 'METHIN' . 'PRECOND' = 3 ;
rv . 'METHIN' . 'RESID' = 1.e-14 ;
rv . 'METHIN' . 'FCPRECT' = 1 ;
rv . 'METHIN' . 'FCPRECI' = 1 ;
rv . 'METHIN' . 'XINIT' = RV . 'INCO' . 'TN' ;
rv . 'METHIN' . 'TYRENU' = 'SLOANE' ;
*
EXEC rv;
*
*****
POST TRAITEMENT ET VISUALISATIONS
*****
*
* ERO : Erreur absolue
* ER1 : Erreur relative
* ER2 : Erreur absolue au carrée
* ERR : Erreur L2 discrète
*
ERO = 'ABS' (RV.INCO.'TN' - SOLEX) ;
ER1 = ERO '/' SOLEX ;
ER2 = ERO '*' ERO ;
ERR = ('MAXI' ('RESU' ER2)) / ('NBEL' mt) ** 0.5 ;
*
'TRAC' MT 'TITR' 'Maillage' ;
'TRAC' SOLEX MT CNT 'TITR' 'Solution exacte' ;
'TRAC' (RV . 'INCO' . 'TN') MT CNT 'TITR' 'Solution castem' ;
'TRAC' ERO MT CNT 'TITR' 'Erreur absolue' ;
'TRAC' ER1 MT CNT 'TITR' 'Erreur relative' ;
*
Evolution de l'erreur au cours des itérations
'SI' ('NEG' IMETH 3) ;
EVO1 = 'EVOL' 'MANU' 'n' RV.INCO.'IT' 'Log Epsi' RV.INCO.'ER';
'DESS' EVO1 'TITRE' 'Histoire de l'erreur absolue'
'TITX' 'Iterations' 'TITY' 'Log(Erreur)';
'FINS' ;
*
Tracé de la solution exacte en faux 3D
'OPTI' 'DIME' 3 'ISOV' 'SULI' ;
OEIL = -0.5 1.2 0.4 ;
MONTAGNE SOLEX MT 1.0 'Solution exacte' OEIL 3 ;
'OPTI' 'DIME' 2 ;
'FIN' ;


```



DEN


DM2S

Tutorial CAST3M pour la mécanique des fluides



Solution exacte

FIG. 7 – Solution stationnaire bilinéaire



Solution exacte

FIG. 8 – Solution stationnaire sinusoidale amortie

Tutorial CAST3M pour la mécanique des fluides




FIG. 9 – Maillages et erreurs à l'état stationnaire pour la solution bilinéaire



DEN

DM2S

Tutorial CAST3M pour la mécanique des fluides




FIG. 10 – Maillage et erreur à l'état stationnaire pour la solution sinusoïdale amortie



DEN

DM2S

Tutorial CAST3M pour la mécanique des fluides

Relation Precision / Finesse du maillage

Schema explicite - Elements triangulaires




FIG. 11 – Estimation de l'ordre du schéma explicite pour un maillage constitué de triangles



Tutorial CAST3M pour la mécanique des fluides




FIG. 12 – Convergence vers la solution stationnaire pour différents maillages de triangles (schéma explicite)



Tutorial CAST3M pour la mécanique des fluides

4 Transport par diffusion-convection d'un scalaire passif

4.1 Objectif

L'étude du transport par convection-diffusion d'un scalaire passif va permettre de passer en revue les méthodes numériques développées en éléments finis dans CAST3M afin de résoudre les équations aux dérivées partielles. Ces mêmes méthodes sont utilisées pour les systèmes et permettent de résoudre les équations de Navier-Stokes. Le cas scalaire est donc une étape essentielle.

Les schémas numériques doivent être adaptés suivant le nombre de Peclet — nombre adimensionnel caractérisant le rapport du temps caractéristique de la convection au temps caractéristique de la diffusion. Le cas test dit de Smith et Hutton, proposé à l'origine dans le cadre d'un workshop, est intéressant dans la mesure où différents nombres de Peclet ont été étudiés — de 10 à 10^6 — permettant ainsi de couvrir une large gamme de situations.

4.2 Consignes

Le cas test de Smith et Hutton traite du transport d'un scalaire passif au sein d'un écoulement par convection et diffusion. L'évolution de la concentration de ce traceur est donné par

$$\frac{\partial c}{\partial t} + \vec{u} \cdot \nabla c = \nu \Delta c$$

où c désigne le scalaire passif transporté — c peut représenter la température, une concentration, *etc.* ; \vec{u} la vitesse (m/s) ; ν la diffusivité (m^2/s). Le domaine Ω , siège du transport, est défini par $\Omega = \{(x, y), (x, y) \in [-1, 1] \times [0, 1]\}$. Le champ de vitesse \vec{u} est connu et constant dans le temps :

$$\vec{u}(x, y) = 2y(1 - x^2)\vec{I}_x - 2x(1 - y^2)\vec{I}_y$$

Le fluide pénètre donc dans le domaine par la base (du côté gauche) et sort également par la base (du côté droit). Les autres cotés sont imperméables.

L'objet du workshop était de calculer la solution stationnaire de ce problème pour un profil de concentration imposé en entrée. La comparaison porte sur le profil de concentration à la sortie du domaine.

Initialement, la concentration est identiquement nulle dans Ω . Les conditions aux limites en concentration sont les suivantes :

$$c(x, 0) = 1 + \tanh[10(2x + 1)] \text{ pour } x \in [-1, 0] \text{ (entrée fluide)}$$

$$c(x, y) = 1 - \tanh 10 \text{ sur les cotés imperméables gauche, droit et haut.}$$

Pour la partie de la frontière par laquelle le fluide sort du domaine, on laisse agir les conditions aux limites dites naturelles : on ne fait rien. Pour le flux diffusif cela revient à imposer un flux nul. Pour le flux convectif, on bénéficie du caractère parabolique de l'équation de transport convective : indépendance du flux convectif aux conditions en aval de l'écoulement. Or, le flux diffusif en sortie du domaine de calcul est rarement nul. De ce fait on introduit souvent une erreur en laissant agir les conditions aux limites naturelles. On doit remarquer que plus la convection est dominante, plus l'erreur commise est minime et reste localisée au voisinage de la sortie. C'est pourquoi on réalise parfois des simulations sur un domaine de calcul plus étendu que celui a priori nécessaire. On espère cantonner l'erreur lié à la condition de sortie dans la zone ajoutée, les post-traitements étant réalisés uniquement dans la zone d'intérêt, en amont de la zone ajoutée.

L'imposition des conditions aux limites marque le début de la chronologie.

En choisissant comme vitesse de référence la vitesse moyenne à l'entrée ($u_{ref} = 1$), comme longueur de référence la hauteur du domaine ($L_{ref} = 1$), on obtient $Pe = \nu^{-1}$. Les cas traités correspondent à un Peclet égal à 10, 100, 500, 10^3



DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 38/65

Tutorial CAST3M pour la mécanique des fluides

ou 10^6 . Si pour le Peclet le plus faible convection et diffusion sont du même ordre, au fur et à mesure qu'il augmente, la convection devient dominante.

On cherchera la solution stationnaire soit en tant que limite asymptotique du problème transitoire — en explicite et en implicite — soit directement en résolvant le problème stationnaire.

Suivant les valeurs croissantes du nombre de Peclet, on observera le raidissement du profil en sortie du domaine, ce dernier étant très proche du profil d'entrée pour le plus haut Pe . Dans le cas explicite, on utilisera de la viscosité numérique afin de stabiliser le schéma — options SUPG et SUPGDC.

Enfin, en implicite, on remplacera l'opérateur TSCAL par les opérateurs élémentaires KONV et LAPN pour traiter indépendamment les différents monômes de l'équation de transport.

4.3 Correction

4.3.1 Effet du Pe

A l'état stationnaire, on observe que le front est d'autant plus raide que le Pe est élevé. Dans le cas $Pe = 10^6$, le profil en sortie et le profil en entrée sont symétriques l'un par rapport à l'autre traduisant l'absence de diffusion (transport le long des lignes de courant).

Plus le Pe augmente plus le pas de temps de stabilité du schéma explicite devient pénalisant. Si pour un Pe de 10 et un maillage grossier les temps de calcul sont raisonnables en explicite, la résolution implicite devient rapidement plus efficace.

4.3.2 Décentrement des termes convectifs

En permanent, dès que le Pe de maille — obtenu en considérant comme vitesse caractéristique la vitesse moyenne sur l'élément et comme longueur caractéristique la racine carrée de la surface de la maille — dépasse 2 des instabilités peuvent apparaître et s'amplifier. Pour les contrer, on ajoute de la diffusion numérique en décentrant le terme convectif — option SUPG, SUPG pour Streamline Upwind Petrov-Galerkin. Dans le cas de fronts raides, un terme supplémentaire assimilable à un Laplacien doit parfois être ajouté — option SUPGDC pour SUPG Discontinuity Capturing.

En transitoire, il est possible de stabiliser le schéma en jouant avec le schéma en temps.

Dans le cas présent, nous cherchons la solution stationnaire en tant que limite asymptotique d'un pseudo-transitoire. Nous utilisons, si nécessaire, un décentrement amont lorsque le schéma explicite est choisi. En effet, pour le Pe le plus bas, dès qu'on maille suffisamment le Pe de maille devient inférieur à la valeur critique. Aussi, avec 10 mailles par unité de longueur on peut utiliser un schéma centré. En implicite, quel que soit le Peclet, le schéma est stable. Inutile donc de décentrer.

4.3.3 Description du jeu de données

Les algorithmes de résolution

Algorithme implicite : On doit inverser une matrice pour évaluer la concentration. Lorsqu'on résout le problème stationnaire, la matrice est constituée des contributions provenant des opérateurs KONV et LAPN associés aux termes convectif et diffusif. L'opérateur TSCA regroupe ces deux opérateurs et peut donc aussi être utilisé.



DM2S

Tutorial CAST3M pour la mécanique des fluides

Dans le cas instationnaire, il faut ajouter aux opérateurs précédents l'opérateur DFDT afin d'ajouter à la matrice la contribution de la dérivée en temps. Le pas de temps doit être donné par l'utilisateur et n'est pas évalué automatiquement.

Algorithme explicite : Seul l'opérateur DFDT contribue à la matrice. Pour des questions de performance, cette matrice doit être simple à inverser. L'idéal serait qu'elle soit diagonale. On y arrive en condensant — les anglophones parlent de lumping — la matrice masse. Comme dans le cas explicite le pas de temps est soumis à des contraintes de stabilité, il est possible de laisser le code évaluer le pas de temps. Pour cela, il suffit d'indiquer le mot clef 'DELTAT' en lieu et place du pas de temps dans les arguments de DFDT. En implicite, l'utilisateur doit toujours indiquer la valeur du pas de temps.

Mise en oeuvre dans CAST3M

Pour décrire le système d'équation, on utilise l'opérateur EQEX. Après avoir créé le maillage du domaine de calcul et le modèle Navier-Stokes associé — objets des deux premières sections de ce rapport — on précise dans EQEX un certain nombre de paramètres. Pour cela, on fixe une valeur après un mot clef :

- ITMA : Nombre de pas de temps (type ENTIER) — ce champ est intéressant lorsque le pas de temps est calculé automatiquement afin d'éviter les calculs sans fin ;
- NITER : Nombre d'itérations de point fixe à réaliser à chaque pas de temps (type ENTIER) — à n'utiliser que dans le cas de problème non linéaire (mettre à 1 sinon) ;
- OMEGA : Coefficient de relaxation utilisé dans le cas d'itérations de point fixe (type FLOTTANT) — à n'utiliser que dans le cas où on réalise un point fixe (mettre à 1. dans le cas où NITER=1) ;
- FIDT : Fréquence d'impression des informations sur le listing (type ENTIER) ;
- ALFA : Abusivement appelé CFL, ce paramètre multiplie le pas de temps de stabilité trouvé par le code (type FLOTTANT) — utilisé uniquement lorsque le pas de temps est automatiquement calculé par le code dans le cas explicite (à fixer à 1. par précaution sinon).
- TFINAL : Fixe la valeur maximale de la chronologie au delà de laquelle le calcul s'arrête (type FLOTTANT).

Ainsi dans l'exemple suivant :

```
RV = 'EQEX' DOMTOT 'ITMA' 500 'NITER' 1 'OMEGA' 1. 'FIDT' 5000
```

on va résoudre un problème décrit sur le domaine DOMTOT pendant 500 pas de temps sans réaliser d'itérations de point fixe. On limite le volume des impressions sur le listing en donnant à FIDT une valeur supérieure à ITMA.

Pour décrire les équations, on opère aussi comme dans le cas de la résolution de l'équation de la chaleur. On appelle les opérateurs associés à chacun des monômes de l'équation de transport :

- ▷ On choisit de nommer l'inconnue de concentration 'CN'.
- ▷ On discrétise les termes convectif et diffusif avec l'opérateur 'TSCA' (transport d'un champ scalaire). Les arguments de cet opérateur sont le coefficient de diffusion (coefficient devant le monôme Δc qui correspond à Pe^{-1} dans la forme adimensionnée de l'équation de transport), le vecteur vitesse, un éventuel terme source et le nom de l'inconnue transportée (ici ''CN') :


```
'ZONE' DOMTOT 'OPER' 'TSCA' DIF 'VITESSE' 0. 'INCO' 'CN'
```
- ▷ On peut en lieu et place de l'opérateur TSCAL utiliser les opérateurs LAPN, KONV et FIMP : ils discrétisent le terme diffusif, le terme convectif et le terme source. L'argument de l'opérateur LAPN est le coefficient de diffusion. Les trois arguments de l'opérateur KONV sont un coefficient multiplicateur, le nom du champ de vitesse et le coefficient de diffusion afin de pouvoir si on le souhaite décentrer le schéma numérique (méthode SUPG et SUPGDC). L'argument de l'opérateur FIMP est la densité de source :


```
'ZONE' DOMTOT 'OPER' 'TSCAL' DIF 'VITESSE' Q 'INCO' 'CN'
```

 ou


```
'ZONE' DOMTOT 'OPER' 'LAPN' DIF 'INCO' 'CN'
```

```
'ZONE' DOMTOT 'OPER' 'KONV' COEF 'VITESSE' DIF 'INCO' 'CN'
```

```
'ZONE' DOMTOT 'OPER' 'FIMP' Q 'INCO' 'CN'
```

Tutorial CAST3M pour la mécanique des fluides

avec dans le cas présent $DIF=1./Pe$; $Q=0.$; $COEF=1.$;

- ▷ Comme dans ce cas test il n'y a pas de terme source, il est inutile de faire appel à l'opérateur FIMP. A noter que l'opérateur FIMP permet aussi d'imposer une densité de flux en lieu et place d'une densité de source. Ce sont les éléments du modèle associé à l'opérateur qui permettent de discriminer les deux situations.

Dans le cas instationnaire, on doit aussi considérer le monôme de dérivée temporelle. C'est l'opérateur DFDT qui évalue la contribution de ce terme à la matrice et au second membre de la forme linéaire à inverser. Cet opérateur a trois arguments : un éventuel coefficient multiplicateur, le nom où est stocké l'inconnue évaluée au début du pas de temps et le pas de temps. Dans le cas où aucune itération de point fixe n'est nécessaire, à chaque pas de temps le nouvel itéré écrase l'ancien. Le nom à donner à DFDT est donc le nom générique de l'inconnue. Lorsque en explicite on souhaite calculer automatiquement le pas de temps, il suffit d'indiquer comme troisième argument le mot clef DELTAT.

'ZONE' DOMTOT 'OPER' 'DFDT' 1. 'CN' 'DELTAT' 'INCO' 'CN'

Dans le cas où une option de décentrement est sélectionnée, il faut ajouter aux arguments de DFDT la diffusion et le champ de vitesse.

Les espaces de discrétisation

Le choix de l'espace de discrétisation se fait au moment de la création du modèle NAVIER_STOKES :

LINE : Les espaces d'approximation sont linéaires (éléments dits P1 et Q1) ;

QUAF : Les espaces d'approximation sont quadratiques (éléments dits P2 et Q2).

Les conditions aux limites

Les frontières du domaine sont en partie perméables. Le fluide pénètre dans le domaine par la moitié inférieure droite et en sort par la moitié inférieure gauche.

Le champ scalaire transporté est donc imposé partout sauf en sortie où la dominance du terme convectif prévaut — caractère hyperbolique. La concentration est donc égale à la concentration prescrite sur le contour imperméable LIMP et à l'entrée LIN du domaine. C'est la directive CLIM de l'opérateur EQEX qui permet d'imposer les conditions aux limites. On précise à la directive le nom de l'inconnue à bloquer. Si l'inconnue est scalaire, on donne le mot clef TIMP. Dans le cas d'une inconnue vectorielle, on précise la composante à l'aide des mots clefs UIMP, VIMP et WIMP.

Procédure utilisateur et post-traitement

La procédure utilisateur `residu` est utilisée en explicite. Elle permet de calculer le résidu durant le transitoire. Elle affiche au choix à chaque pas de temps l'évolution du résidu ou les iso-concentrations sur le domaine.

En post-traitement, nous traçons le maillage, le champ de vitesse, les iso-concentrations et une comparaison entre la solution calculée et la solution de référence.

Les procédures utilisateurs et le post-traitement doivent être adaptés à la situation traitée et subissent les évolutions du code. C'est pourquoi il vous appartient de comprendre les exemples fournis afin de pouvoir les adapter à vos futurs travaux.

4.4 Complément

- ▷ R.M. Smith and A.G.Hutton, The numerical treatment of advection : a performance comparison of current methods, Numerical Heat Transfer, Vol 5, pp.439–461, 1982.
- ▷ A.Ern, J-L.Guermond, Eléments finis : Théorie, applications, mise en oeuvre, Mathématiques & Applications, Vol 36, Springer Ed., 2002.
- ▷ C.A.J.Fletcher, Computational Techniques for Fluid Dynamics, Volume 1, Springer Verlag Ed., 1988.
- ▷ C.Hirsch, Numerical Computation of Internal and External Flows, Volume 1, Fundamentals of numerical discretization, Wiley Ed., 1988.



DEN

DM2S

RAPPORT DM2S

Tutorial CAST3M pour la mécanique des fluides

- ▷ O.Pironneau, Finite element methods for fluids, John Wiley Ed., 1989.
- ▷ O.C. Zienkiewicz, R.L. Taylor, The Finite element method, Vol 1 : The basis, 5th Butterworth-Heinemann Ed., 2000.

4.5 Fichier Gibiane

Afin d'alléger ce document, vous trouverez ci-joint uniquement le cas explicite. A partir de cet exemple, il est très aisé de construire le modèle implicite ... à condition de supprimer les bonnes cartes.

```

*
-----
* Convection/Diffusion : CAS SMITH ET HUTTON
* REFERENCE : NUMERICAL HEAT TRANSFER, VOL.5, p.439, 1982
-----
* Les faces latérales et supérieure d'une boîte rectangulaire sont
* imperméables. Le fluide rentre et sort de la boîte par la face
* inférieure. Il rentre par la moitié gauche de la face inférieure
* et sort par la moitié droite de cette même face.
*
* Le champ de vitesse est connu et donné par :
* v(x,y) = 2y(1-x2) i - 2x(1-y2) j
* la taille de la boîte étant [-1,1]x[0,1]
*
* L'écoulement transporte un scalaire passif qui est injecté en entrée
* suivant le profil :
* c(x,0) = 1 + tanh(10(2x+1))
* avec x variant de -1 à 0 (entrée du domaine).
* Sur les frontières imperméables du domaine, la concentration est
* constante et égale à 1-tanh(10).
*
* On cherche la solution stationnaire du transport par diffusion et
* convection du champ scalaire passif. On compare la solution en sortie
* avec la solution de référence. Plusieurs solutions sont calculées
* suivant le Peclet (rapport entre la convection et la diffusion).
*
-----
* OPTI 'DIME' 2 'ISOV' 'SULI' ;
*
* Pe : Peclet (convection sur diffusion)
* Les cas traités dans la référence sont 10, 100, 500, 1000 et 1000000
Pe = 10 ;
*
* KSUPG : Option de décentrement (CENTREE/SUPG/SUPGDC)
* GRAPH : Boolean pour l'affichage des tracés à l'issue du calcul
* POST1 : Boolean pour affichage Résidu à chaque pas
* POST2 : Boolean pour affichage C(x,t) à chaque pas
* COMPLET : Boolean modifiant la finesse du maillage et la précision des calculs
* NX : Nombre de maille suivant x
* NY : Nombre de maille suivant y
* nbiter : Nombre de pas de temps
* tolera : Tolérance pour la comparaison avec la solution de référence
KSUPG = 'SUPG' ;
GRAPH = VRAI ;
POST1 = FAUX ;
POST2 = 'NON' POST1 ;
COMPLET = FAUX ;
'SI' ( COMPLET ) ;
  'OPTI' 'ELEM' 'TRIS' ;
  NY = 20 ;
  NX = 2 * NY ;
  nbiter = 2000 ;
  tolera = 0.006 ;
'SINON' ;
  'OPTI' 'ELEM' 'QUA4' ;
  NY = 10 ;
  NX = 2 * NY ;
  nbiter = 100 ;
  tolera = 0.012 ;
'FINSI' ;
*
-----
* Calcul du résidu en température et arrêt suivant un critère
*-----
* E/ : RVX : TABLE : TABLE des données créées par EQEX
* : ARG1 : Fréquence d'impression
* : ARG2 : Critère d'arrêt
* /S : MAT1 : MATRIK : Objet vide
* /S : CHP1 : CHPO : Objet vide
*-----

```

```

*
'DEBPROC' residu rvx*table ;
RV = rvx . 'EQEX' ;
FREQ = RVX . 'ARG1' ;
EPSO = RVX . 'ARG2' ;
NITER = RV . 'NITER' ;
DD = RV . 'PASDETPS' . 'NUPASDT' ;
NN = DD '/' FREQ ;
LO = 'EGA' (DD '-' (FREQ*NN)) 0 ;
'SI' LO ;
  RANGO = RV . 'PASDETPS' . 'NUPASDT' ;
  TIMEO = RV . 'PASDETPS' . 'TPS' ;
  CNO = RV . 'INCO' . 'CN' ;
  CNMO = RV . 'INCO' . 'CN2' ;
  ERRO = ('MAXIMUM' ('ABS' (CNO '-' CNMO))) '+' 1.D-20 ;
  ERR10 = ('LOG' ERRO ) '/' ('LOG' 10.) ;
  'MESSAGE' 'Résidu en Température au pas'
  RANGO (' TIMEO ' ) : ' ERRO ' : ' ERR10 ;
  RV . 'INCO' . 'IT' = RV . 'INCO' . 'IT' 'ET' ('PROG' RANGO) ;
  RV . 'INCO' . 'TI' = RV . 'INCO' . 'TI' 'ET' ('PROG' TIMEO) ;
  RV . 'INCO' . 'ER' = RV . 'INCO' . 'ER' 'ET' ('PROG' ERR10) ;
  EV1 = 'EVOL' 'MANUEL' (RV . 'INCO' . 'IT') (RV . 'INCO' . 'ER') ;
  Y1 = ('LOG' EPSO) '/' ('LOG' 10) ;
  'SI' POST1 ;
  X1 = 0. ; X2 = RV . 'ITMA' ;
  'DESSIN' EV1 'YBOR' Y1 0. 'NCLK'
  'TITR' 'Evolution du résidu' ;
'FINSI' ;
'SI' POST2 ;
  L1 = (PROG 0. PAS 100. 2000.) * 1.D-3 ;
  trace L1 cno domtot cnt1 'TITR' 'Concentration' 'NCLK' ;
'FINSI' ;
'SI' ((ERR10 < Y1) 'ET' (DD > 10)) ;
  RV . 'FINAL' = RV . 'PASDETPS' . 'TPS' ;
'FINSI' ;
'FINSI' ;
RV . 'INCO' . 'CN2' = 'COPIER' RV . 'INCO' . 'CN' ;
mat1 chp1 = 'KOPS' 'MATRIK' ;
'FIMP' mat1 chp1 ;
*-----
*
*-- MAILLAGE
*
*
* Points
A1 = -1.0 0.0 ;
A2 = 1.0 0.0 ;
A3 = 1.0 1.0 ;
A4 = -1.0 1.0 ;
A0 = 0.0 0.0 ;
*
* Lignes
LIN = A1 'DROI' NY A0 ;
LOUT = A0 'DROI' NY A2 ;
FBAS = LIN 'ET' LOUT ;
FDRO = A2 'DROI' NY A3 ;
FHAU = A3 'DROI' NX A4 ;
FGAU = A4 'DROI' NY A1 ;
LIMP = FDRO 'ET' FHAU 'ET' FGAU ;
*
* Maillage
DOMTOT = 'DALL' FBAS FDRO FHAU FGAU 'PLAN' ;
CNT1 = 'CONT' DOMTOT ;
*
* Modèles et sous-modèles
DOM2 = 'CHAN' 'QUAF' DOMTOT ;
LIN2 = 'CHAN' 'QUAF' LIN ;
LIMP2 = 'CHAN' 'QUAF' LIMP ;
$DOMTOT = 'MODE' DOM2 'NAVIER_STOKES' 'LINE' ;
$LIN = 'MODE' LIN2 'NAVIER_STOKES' 'LINE' ;

```



DEN

DM2S

RAPPORT DM2S

Tutorial CAST3M pour la mécanique des fluides

```

$LIMP = 'MODE' LIMP2 'NAVIER_STOKES' 'LINE' ;
*
* Récupération des maillages et fusion des supports
DOMTOT = 'DOMA' $DOMTOT 'MAILLAGE' ;
LIN = 'DOMA' $LIN 'MAILLAGE' ;
LIMP = 'DOMA' $LIMP 'MAILLAGE' ;
FDMTOT = 'DOMA' $DMTOT 'FACE' ;
CLIN = 'DOMA' $LIN 'CENTRE' ;
CLIMP = 'DOMA' $LIMP 'CENTRE' ;
'ELIM' FDMTOT (CLIN 'ET' CLIMP) 1.D-3 ;
*
* Champ de vitesse
XX YY = 'COORD' DOMTOT ;
VXSH = (2.*YY)*(1.0-(XX*XX)) ;
VYSH = (-2.*XX)*(1.0-(YY*YY)) ;
VXO = 'NOMC' 'UX' VXSH ;
VYO = 'NOMC' 'UY' VYSH ;
VX = 'KCHT' $DMTOT 'SCAL' 'SOMMET' 'COMP' 'UX' VXO ;
VY = 'KCHT' $DMTOT 'SCAL' 'SOMMET' 'COMP' 'UY' VYO ;
CHVIT = 'KCHT' $DMTOT 'VECT' 'SOMMET' 'COMP' 'UX' 'UY' (VX 'ET' VY) ;
*
* Diffusion (1/Pe)
DIF = 1.0 / ('FLOT' Pe) ;
*
* Profil de concentration à l'entrée
XBAS = 'COORD' 1 LIN ;
TOTO = 2.0*XBAS + 1.0 * 10. ;
SOLUTION = 'TANH' TOTO ;
SOLUTION = 1.0 + SOLUTION ;
CHP1 = 'KCHT' $LIN 'SCAL' 'SOMMET' 0. SOLUTION ;
CHP1 = 'NOMC' 'CN' CHP1 ;
*
* Conditions aux limites en concentration sur les frontières imperméables
C1 = (1.0 - (TANH 10.0)) ;
*
* Description du problème de transport
RV1 = 'EGEX' $DMTOT 'ITMA' nbiter 'ALFA' 0.7
'ZONE' $DMTOT 'OPER' residu 1 1.D-7
'OPTI' KSUPG 'EFM1' 'EXPL'
'ZONE' $DMTOT 'OPER' 'TSCAL' DIF 'VITESSE' 0. 'INCO' 'CN'
'OPTI' 'CENTREE'
'ZONE' $DMTOT 'OPER' 'DFDT' 1. 'CN' 'DELTAT' 'INCO' 'CN' ;
*
* Description des conditions aux limites
RV1 = 'EGEX' RV1
'CLIM' 'CN' 'TIMP' LIN CHP1
'CLIM' 'CN' 'TIMP' LIMP C1 ;
*
* Description des conditions initiales
RV1 'INCO' = TABLE 'INCO' ;
RV1 'INCO' . 'CN' = 'KCHT' $DMTOT 'SCAL' 'SOMMET' 0. ;
*
* Autres data
RV1 'INCO' . 'VITESSE' = CHVIT ;
RV1 'INCO' . 'CN2' = 'KCHT' $DMTOT 'SCAL' 'SOMMET' 0. ;
RV1 'INCO' . 'IT' = 'PROG' ;
RV1 'INCO' . 'TI' = 'PROG' ;
RV1 'INCO' . 'ER' = 'PROG' ;
*
*- CALCUL
*
EXEC RV1 ;
*
*- ANALYSE DES RESULTATS
*
EVOL1 = 'EVOL' 'CHPO' (RV1 . 'INCO' . 'CN') 'SCAL' FBAS ;
EVOL2 = 'EVOL' 'CHPO' XX 'SCAL' FBAS ;
LIX = 'EXTR' EVOL2 'ORDO' ;
LIU = 'EXTR' EVOL1 'ORDO' ;
EVOL3 = 'EVOL' 'MANU' 'X' LIX 'C(X,0)' LIU ;
*
* Solutions de référence
LIXT = PROG 0. 0.1 0.2 0.3 0.4 0.5 0.6 0.7 0.8 0.9 1.0 ;
SI ( Pe 'EGA' 10 ) ;

```

```

LIUT = PROG 1.989 1.402 1.146 0.946 0.775 0.621 0.480 0.349
0.227 0.111 0.000 ;
FINSI ;
SI ( Pe 'EGA' 100 ) ;
LIUT = PROG 2.000 1.940 1.836 1.627 1.288 0.869 0.480 0.209
0.070 0.017 0.000 ;
FINSI ;
SI ( Pe 'EGA' 500 ) ;
LIUT = PROG 2.000 2.000 1.998 1.965 1.702 0.947 0.242 0.023
0.001 0.000 0.000 ;
FINSI ;
SI ( Pe 'EGA' 1000 ) ;
LIUT = PROG 2.000 2.000 2.000 1.985 1.841 0.951 0.154 0.001
0.000 0.000 0.000 ;
FINSI ;
SI ( Pe 'EGA' 1000000 ) ;
LIUT = PROG 2.000 2.000 2.000 1.999 1.964 1.000 0.036 0.001
0.000 0.000 0.000 ;
FINSI ;
EVOL4 = EVOL 'MANU' 'X' LIXT 'Uref(X,0)' LIUT ;
*
LIUC = IPOL LIXT LIX LIU ;
NP = DIME LIXT ;
ERRO = 0. ;
REPETER BLOC1 NP ;
UCAL = EXTRAIRE LIUC &BLOC1 ;
UREF = EXTRAIRE LIUT &BLOC1 ;
ERRO = ERRO + ((UCAL-UREF)*(UCAL-UREF)) ;
FIN BLOC1 ;
ERRO = ERRO/NP ;
ERRO = ERRO '**' 0.5 ;
MESSAGE 'ERREUR ' ERRO ;
SI ( ERRO > tolera ) ;
* ERREUR 5 ;
FINSI ;
*
* *- POST-TRAITEMENT
*
'SI' GRAPH ;
*
* Maillage
'TRAC' DOMTOT 'TITR' 'Maillage' ;
*
* Vitesse
UNCH = 'VECT' CHVIT 0.1 'UX' 'UY' 'ROUGE' ;
'TRAC' UNCH DOMTOT CNT1 'TITR' 'Vitesse transportante' ;
*
* Concentration
'TRAC' DOMTOT (RV1 . 'INCO' . 'CN') CNT1 'TITR' 'Concentration' ;
*
* Convergence vers la solution stationnaire
EV1 = 'EVOL' 'MANU' 'ITERATIONS' (RV1 . 'INCO' . 'IT')
'LOG|E|inf' (RV1 . 'INCO' . 'ER') ;
'DESS' EV1 'XBOR' 0. ('FLOT' NBITER) 'YBOR' -20.0 0.0
'TITR' 'Evolution du résidu' ;
*
* Comparaison avec la solution analytique
TAB1 = TABLE ;
TAB1 . TITRE = TABLE ;
TAB1 . TITRE . 1 = 'MOT' 'CAST3M' ;
TAB1 . TITRE . 2 = 'MOT' 'REFERENCE' ;
TAB1 . 2 = 'MARQ LOSA NOLI' ;
'TITR' 'Comparaison CAST3M/Référence' ;
'DESS' (EVOL3 ET EVOL4) 'LEGE' TAB1
'MIMA' 'XBOR' -1.0 1.0 'YBOR' -1.0 3.0 ;
*
EVOL1 ;
*FIN ;

```



DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 43/65

Tutorial CAST3M pour la mécanique des fluides




FIG. 13 – Maillage 40×20 constitué de triangles et champ de vitesse pour le cas-test de Smith et Hutton



DEN

DM2S

Tutorial CAST3M pour la mécanique des fluides





FIG. 14 - Cas $\nu = 10^{-1}$, TSCAL option SUPG





FIG. 15 - Cas $\nu = 10^{-2}$, TSCAL option SUPG





FIG. 16 - Cas $\nu = 10^{-6}$, TSCAL option SUPG





FIG. 17 - Cas $\nu = 10^{-6}$, TSCAL option SUPG-CC



DEN

DM2S

SFME/LTMF/RT/03-035/A

Date : 15/09/03

RAPPORT DM2S

Page : 45/65

Tutorial CAST3M pour la mécanique des fluides

5 Marche descendante

5.1 Objectif

L'étude de l'écoulement laminaire dans un canal plan avec une marche descendante permet de présenter les méthodes numériques employées dans CAST3M afin de résoudre les équations de Navier-Stokes pour un fluide Newtonien et un écoulement incompressible. Les différents solveurs et éléments finis en vitesse/pression utilisés sont présentés à raison d'un jeu de données par couple solveur/élément. Dans les jeux de données de non-régression livrés avec CAST3M, treize jeux de données couvrent une grande partie des possibilités. Le nom de ces jeux de données commence par back. Nous nous contenterons ici d'étudier un élément pour chaque méthode de résolution.

Rappelons que pour un fluide Newtonien et un écoulement incompressible, nous devons résoudre

$$\begin{cases} \operatorname{div} \vec{u} = 0 \\ \frac{\partial \vec{u}}{\partial t} + \vec{u} \nabla \vec{u} = \frac{1}{\rho} \{-\nabla P + \mu \Delta \vec{u}\} \end{cases}$$

où \vec{u} désigne la vitesse (m/s) ; ρ la densité de référence du fluide (kg/m^3) ; P la pression (Pa) ; μ la viscosité dynamique (kg/ms).

5.2 Consignes

Le cas étudié est issu d'un benchmark ayant fait l'objet d'une synthèse éditée par K.Morgan, J.Périaux and F.Thomasset : Analysis of laminar flow over a backward facing step, Vol 9 of Notes on Num. Fluid Mech., Vieweg, 1984.

On modélise l'écoulement laminaire incompressible stationnaire dans un canal autour d'une marche descendante. On note h (resp. H) la hauteur du canal en amont (resp. en aval) de la marche, l (resp. D) la longueur du canal en amont (resp. en aval) de la marche. Un profil de vitesse parabolique est imposé à l'entrée, la vitesse maximale étant notée U_{max} . L'écoulement se sépare au niveau de la marche et se ré-attache en aval de la marche à une abscisse à déterminer.




FIG. 18 – Ecoulement autour d'une marche descendante

Le cas test considéré ici correspond aux valeurs $D = 18m$, $l = 3m$, $H = 1.5m$ et $h = 1m$. Les équations de Navier-Stokes sont adimensionnées par les échelles de longueur et de vitesse $(H - h)$ et U_{max} — donc $(H - h)/U_{max}$ pour le temps — et on a

$$\begin{cases} \operatorname{div} \vec{u} = 0 \\ \frac{\partial \vec{u}}{\partial t} + \vec{u} \nabla \vec{u} = -\nabla P + \frac{1}{Re} \Delta \vec{u} \end{cases}$$

Tutorial CAST3M pour la mécanique des fluides

où la pression est adimensionnée par ρU_{max}^2 , le nombre de Reynolds $Re = U_{max}(H - h)/\nu$ étant fixé par la suite à 150.

Les conditions aux limites sont classiques pour ce type d'écoulement :

- ▷ Profil parabolique de la vitesse à l'entrée (Poiseuille) ;
- ▷ Contraintes normales nulles en sortie ;
- ▷ Adhérence sur les parois.

Le coefficient de frottement à la paroi C_f défini par

$$\frac{1}{2} Re C_f(x) = \left(\frac{\partial u}{\partial y} \right)_{paroi}$$

permet de localiser le point de ré-attachement puisqu'il s'annule en ce lieu.

5.3 Correction

5.3.1 Structure de l'écoulement

La présence de la marche provoque le décollement du fluide et la formation d'un tourbillon en aval de la marche. Le point de ré-attachement correspond à la position d'équilibre entre le tourbillon et l'écoulement principal. La longueur expérimentale adimensionnée est de 6. Suivant la méthode utilisée on obtient une différence de plus ou moins 0.5, différences comparables à celles observées dans la littérature.

Il faut veiller à mailler suffisamment la partie aval du canal afin que la condition associée au régime établi soit effectivement vérifiée.

Compte tenu de la formulation variationnelle utilisée, la structure de l'écoulement a un effet sur la condition aux limites en sortie. Cet effet s'atténue lorsque les effets visqueux s'estompent. La condition à la limite dégénère alors en une condition de pression nulle sur la surface.

5.3.2 Description du jeu de données

Les algorithmes de résolution

Algorithme implicite : Les équations en vitesse et pression sont discrétisées et résolues simultanément.

Méthode de projection semi-explicite : Les équations en vitesse et pression sont découplées : elles sont au sein d'un pas de temps résolues les unes après les autres. L'équation de quantité de mouvement est résolue explicitement si bien que le schéma est stable sous conditions — le pas de temps est limité par des contraintes de stabilités linéaires (conditions sur le nombre CFL et le nombre de Fourier). La pression est solution d'un pseudo-Laplacien. *L'algorithme semi-explicite est la variante la plus explicite des méthodes de projection.*

Méthode de projection semi-implicite : Les équations en vitesse et pression sont, comme précédemment, découplées. Cependant, elles sont résolues par une approche implicite. De ce fait, le pas de temps n'est pas soumis à des contraintes de stabilités linéaires. Pour obtenir une solution physique aux équations découplées, il doit cependant rester dans des limites raisonnables.

Mise en oeuvre dans CAST3M

Pour décrire le système d'équation, on utilise comme pour la description d'une équation scalaire l'opérateur **EQEX**. Après avoir créé le maillage du domaine de calcul et le modèle Navier-Stokes associé — cf les deux premiers TP — on précise dans **EQEX** un certain nombre de paramètres. Pour cela, on fixe une valeur après un mot clef :



DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 47/65

Tutorial CAST3M pour la mécanique des fluides

ITMA : Nombre de pas de temps (type ENTIER) — ce champ est intéressant lorsque le pas de temps est calculé automatiquement afin d'éviter les calculs sans fin ;

NITER : Nombre d'itérations de point fixe à réaliser à chaque pas de temps (type ENTIER) — à n'utiliser que dans le cas de problème non linéaire (mettre à 1 sinon) ;

OMEGA : Coefficient de relaxation utilisé dans le cas d'itérations de point fixe (type FLOTTANT) — à n'utiliser que dans le cas où on réalise un point fixe (mettre à 1. dans le cas où NITER=1) ;

FIDT : Fréquence d'impression des informations sur le listing (type ENTIER) ;

ALFA : Abusivement appelé CFL, ce paramètre multiplie le pas de temps de stabilité trouvé par le code (type FLOTTANT) — utilisé uniquement lorsque le pas de temps n'est pas imposé par l'utilisateur (voir la notice de l'opérateur DFDT).

TFINAL : Fixe la valeur maximale de la chronologie au delà de laquelle le calcul s'arrête (type FLOTTANT).

Ainsi dans l'exemple suivant :

```
RV = 'EQEX' DOMTOT 'ITMA' 500 'NITER' 1 'OMEGA' 1. 'FIDT' 5000
```

on va résoudre un problème décrit sur le domaine DOMTOT pendant 500 pas de temps sans réaliser d'itérations de point fixe. On limite le volume des impressions sur le listing du calcul en donnant à FIDT une valeur supérieure à ITMA.

Pour décrire les équations, on opère aussi comme dans le cas de la résolution du problème de transport d'un champ scalaire passif par diffusion et convection. On appelle les opérateurs associés à chacun des monômes des équations :

- ▷ On choisit de nommer l'inconnue de vitesse 'UN' et l'inconnue de pression 'PRES'.
- ▷ On discrétise l'équation de quantité de mouvement via l'opérateur 'NS'. Dans le cas présent, l'argument de cet opérateur est le coefficient de viscosité cinématique ν (coefficient devant le monôme $\Delta \vec{u}$ lorsque celui devant le monôme de dérivée temporelle (opérateur DFDT) est égal à 1.) :

```
'ZONE' DOMTOT 'OPER' 'NS' NU 'INCO' 'UN'
```

avec $NU=1./Re$ en adimensionné. L'opérateur NS ne discrétise ni le monôme de dérivée en temps de la vitesse ni le monôme en pression. Ce sont les opérateurs DFDT et KBBT qui s'en chargent.

- ▷ On discrétise les termes en pression dans un troisième temps. L'opérateur KBBT discrétise le terme en gradient de pression de l'équation de quantité de mouvement (terme qui n'est pas discrétisé par l'opérateur NS) ainsi que le terme en divergence \vec{u} . Dans un contexte de résolution implicite, c'est cet opérateur qui couple les inconnues de vitesse et de pression. En méthode de projection, les matrices associées aux opérateurs gradient et divergence permettent d'évaluer le pseudo-Laplacien en pression. L'argument de cet opérateur est un éventuel coefficient multiplicateur. Ce coefficient vaut 1. si l'opérateur est appelé en implicite et -1. sinon.

```
'ZONE' DOMTOT 'OPER' 'KBBT' C1 'INCO' 'UN' 'PRES'
```

avec $C1=1.$ en implicite et $-1.$ en explicite.

- ▷ Dans le cas semi-explicite, le pseudo-Laplacien en pression est calculé directement par l'opérateur PRESSION et les éléments relatifs à la pression sont construits à l'aide de l'opérateur EQPR (pour équation en pression). La table ainsi créée est rangée dans la table associée à l'opérateur EQEX à l'indice PRESSION.

```
RV = 'EQEX' ... ;
```

```
RVP = 'EQPR' DOMTOT 'KTYPI' 1 'ZONE' DOMTOT 'OPER' 'PRESSION' 0. ;
```

```
RV . 'PRESSION' = RVP ;
```

Les espaces de discrétisation

Le choix des espaces de discrétisation se fait en deux étapes. On indique dans l'objet modèle Navier-Stokes l'espace de discrétisation choisi pour les inconnues autres que la pression. On dispose des choix suivants :

LINE : Les espaces d'approximation sont linéaires (éléments dits P1 et Q1) ;

MACRO : Les éléments linéaires sont stabilisés par la technique des macro-éléments, les espaces d'approximation étant linéaires sur chaque micro-élément ;

QUAF : Les espaces d'approximation sont quadratiques (éléments dits P2 et Q2).



DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 48/65

Tutorial CAST3M pour la mécanique des fluides

On indique l'espace de discrétisation de la pression après la directive OPTI de l'opérateur EQEX afin que les opérateurs de discrétisation puissent travailler :

CENTRE : La pression est constante sur chaque élément ;

CENTREP1 : La pression est linéaire sur chaque élément et discontinue d'un élément à ses voisins ;

MSOMMET : La pression est linéaire sur chaque élément et continue d'un élément à ses voisins.

Comme nous l'avons vu en cours, tous les couples vitesse/pression ne sont pas judicieux. Six possibilités vous sont offertes via la variable ICHOI :

1. **LINE** et **CENTRE** : Cet élément est le plus simple mais sujet aux modes de pression parasites.
2. **MACRO** et **CENTRE** : Cet élément est la version stabilisée de l'élément précédant.
3. **MACRO** et **CENTREP1** : Cet élément est stable dans le cas de quadrangles linéaires. Pour les triangles, il faudrait ajouter la bulle.
4. **QUAF** et **CENTREP1** : En augmentant le degré d'approximation en vitesse on gagne en précision mais au prix d'un temps de calcul plus important. En effet, le couplage entre les degrés de liberté en vitesse augmente d'où des matrices moins creuses qu'en linéaire.
5. **LINE** et **MSOMMET** : Bien que sujet aux modes de pression parasites, cet élément permet d'approximer linéairement et continûment vitesses et pressions pour un coût raisonnable.
6. **QUAF** et **MSOMMET** : Pour ceux qui souhaitent de la stabilité, de la précision et des pressions continues. A noter que par rapport au choix **CENTREP1**, le choix **MSOMMET** diminue la taille du problème en pression (continuité des pressions).

Les couples **LINE/CENTREP1** et **QUAF/CENTRE** sont sans objets.

Le mot clef associé à l'espace de discrétisation en vitesse est un des arguments du modèle Navier-Stokes. Il sert aussi à indiquer l'espace de discrétisation des inconnues autres que la pression. Celui associé à la pression intervient dans les options de l'opérateur EQEX afin que les opérateurs de discrétisation puissent travailler.

ATTENTION : *Tous les éléments vitesse/pression ne sont pas disponibles avec tous les algorithmes. Ainsi, on se contentera d'éléments linéaires en semi-explicite (i.e. pas de QUAF en semi-explicite) ; de pressions discontinues en implicite (i.e. pas de MSOMMET en implicite). Un jour viendra ... peut-être.*

Les conditions aux limites

Comme dans le cas scalaire la directive CLIM de l'opérateur EQEX permet d'imposer les conditions aux limites.

Dans le cas où la pression est continue, il faut l'imposer en sortie afin de ne pas tomber sur une indétermination. C'est pourquoi dans ce cas on fixe la pression à 0 en un point du support de l'inconnue de pression.

Dans le cas où le domaine est fermée, il faut aussi imposer la pression en un point afin de lever l'indétermination occasionnée par la contrainte d'incompressibilité sur la frontière du domaine.

Les utilitaires

Deux procédures utilisateurs sont fournies en tête du jeu de données : **attac** et **residu**. Elles permettent respectivement de calculer le point de ré-attachement et l'évolution du résidu en vitesse durant le transitoire. Cette dernière procédure est appelée à chaque pas de temps et l'évolution du résidu en vitesse est tracé si on l'a demandé.

La procédures **attac** est appelée par la procédure **residu** ainsi qu'au moment du post-traitement à l'issue du calcul. En plus du point de ré-attachement, nous post-traitons la vitesse et la pression.

Les procédures utilisateurs et le post-traitement doivent souvent être adaptés à la situation traitée, aux évolutions du code. C'est pourquoi il vous appartient de comprendre les exemples fournis afin de pouvoir les adapter à vos futurs travaux.



DEN

DM2S

SFME/LTMF/RT/03-035/A
Date : 15/09/03

RAPPORT DM2S

Page : 49/65

Tutorial CAST3M pour la mécanique des fluides

5.4 Complément

- ▷ A.Ern, J-L.Guermond, Elements finis : Théorie, applications, mise en oeuvre, Mathématiques & Applications, Vol 36, Springer Ed., 2002.
- ▷ C.A.J.Fletcher, Computational Techniques for Fluid Dynamics, Volume 2, Springer Verlag Ed., 1988.
- ▷ P.M.Gresho and R.L.Sani, Incompressible flow and the finite element method, Wiley Ed, 1998.
- ▷ C.Hirsch, Numerical Computation of Internal and External Flows, Volume 2, Fundamentals of numerical discretization, Wiley Ed., 1988.
- ▷ L.Quartapelle, Numerical Solution of the Incompressible Navier-Stokes Equations, Birkhauser, 1993.
- ▷ K.Morgan, J.Périaux and F.Thomasset, Analysis of laminar flow over a backward facing step, Vol9 of Notes on Num. Fluid Mech., Vieweg, 1984.
- ▷ O.Pironneau, Finite element methods for fluids, John Wiley Ed., 1989.
- ▷ J.C.Tannehill, D.A.Anderson and R.H.Pletcher, Computational Fluid Mechanics and Heat Transfer, Series in Comp. and Phys. Processes in Mech. and Thermal Sc., Taylor & Francis, 2nd ed, 1997.
- ▷ F.M. White, Viscous Fluid Flow, Mc Graw-Hill, 2nd ed, 1991.
- ▷ O.C. Zienkiewicz, R.L. Taylor, The Finite element method, Vol 3 : Fluid dynamics, 5th Butterworth-Heinemann Ed., 2000.

5.5 Fichiers Gibiane

5.5.1 Schéma semi explicite

```
***** CAS TEST : back_pression_1.dgibi *****
*
* Ce test permet de vérifier le bon fonctionnement des opérateurs
* utilisés pour la résolution des équations de NAVIER-STOKES en EF
* par un algorithme semi-explicite
*
* Le cas étudié est celui d'un écoulement laminaire dans un canal
* en présence d'une marche descendante. On teste la position du
* point de réattachement par rapport au pied de la marche.
*
* Référence : K. Morgan, J. Périaux and F. Thomasset, editors,
* Analysis of laminar flow over a backward facing step, Vol9 of
* Notes on numerical Fluid Mechanics, Vieweg, 1984.
*
*-----*
* Auteur : F.Dabbene (LTMF)                                03/03
*-----*
'SAUT' 'PAGE' ;
'OPTION' 'DIME' 2 'ELEM' 'QUAS' 'ECHO' 0 ;
*
*- Pilotage du calcul
*
* DISCR : Element en vitesse (LINE/MACRO, pas QUAF)
* KPRES : Element en pression (CENTRE, pas CENTREP1/MSONMET)
* KSUPG : Méthode de décentrement pour NS (CENTREE/SUPG/SUPGDC)
* (DFDT en CENTREE en semi-explicite EFM1)
*
* EPSO : Test d'arrêt
* RAF : Taille de maille définie par 0.1*raf
*
* FREQO : Fréquence d'évaluation du résidu
* GRAPH : Booléen pour les tracés
* COMPLET : Booléen volume de calcul
*
DISCR = 'MACRO' ;
KPRES = 'CENTRE' ;
'SI' ('EGA' discr 'QUAF') ;
'ERRE' 'VITESSE QUAF interdite en semi-explicite' 5 ;
'SINON' ;
'SI' ('NEG' kpres 'CENTRE') ;
```

```
'ERRE' 'PRESSION CENTRE obligatoire en semi-explicite' 5 ;
'FINSI' ;
'FINSI' ;
KSUPG = 'SUPG' ;
GRAPH = FAUX ;
COMPLET = FAUX ;
'SI' COMPLET ;
FREQO = 500 ;
EPSO = 1.D-6 ;
RAF = 1. ;
'SINON' ;
FREQO = 20 ;
EPSO = 1.D-2 ;
RAF = 3. ;
'FINSI' ;
*
*-----*
* Recherche du point de réattachement (point où dUx/dy=0)
*-----*
* 1/ Après avoir calculé le gradient de Ux, on ne conserve que les valeurs
* sur la frontière qui nous intéresse sous la forme d'une évolution.
* 2/ On borne l'évolution au voisinage du point de réattachement afin
* d'avoir une variation monotone sur l'intervale de dUx/dy.
* 3/ On recherche par interpolation le zero de la fonction (dUx/dy(s),s)
* La valeur obtenu est l'abscisse curviligne cherchée
*
* Remarques :
* (i) 1/ permettrait de calculer le coeff de frottement à la paroi :
* il suffirait de diviser EV1 par le bon coefficient (Re/2 ici)
* (ii) Il est impératif que dUx/dy soit monotone sur l'intervale 2/ afin
* qu'il y ait unicité du zero (principe des valeurs intermédiaires)
*-----*
'DEBPROC' attac ;
* 1/
Ux = 'KCHT' $DOMTOT 'SCAL' 'SOMMET' 'COMP' 'UX' 0. RV.'INCO'. 'UN' ;
DUxDY = 'EXCO' 'UY' ('KOPS' Ux 'GRADS' $DOMTOT) 'SCAL' ;
EV1 = 'EVOL' 'CHPO' DUxDY BOTTOM ;
* 2/
EV2 = 'EXTR' EV1 'APRE' 5. ;
EV3 = 'EXTR' EV2 'AVAN' 10. ;
* 3/
```



DEN

DM2S

RAPPORT DM2S

Tutorial CAST3M pour la mécanique des fluides

```

LX1 = 'EXTR' EV3 'ABSC' ;
LY1 = 'EXTR' EV3 'ORDD' ;
'ORDO' LY1 ;
Ymin = 'MINI' LY1 ;
Ymax = 'MAXI' LY1 ;
Delta = Ymin * Ymax ;
'SI' (Delta < 0.) ;
YO = 0. ;
'SINON' ;
'MESS' 'Fonction non monotone' ;
YO = Ymin '+' Ymax '/' 2. ;
* 'ERRE' 'Fonction non monotone' 5 ;
'FINSI' ;
V1 = 'IPOL' YO LY1 LX1 ;
'FINP' V1 ;
*
*****
* Calcul du résidu basé sur la composante horizontale de la vitesse
* et arrêt suivant un critère transmis
*****
* E/ : RVX      : TABLE      : TABLE des données créées par EQEX
*      ARG1      : Fréquence d'impression
*      ARG2      : Critère d'arrêt
*      ARG3      : Boolean de tracé
* /S : MAT1      : MATRIK      : Objet vide
* /S : CHP1      : CHPO        : Objet vide
*-----
* MAT1 et CHP1 permettent d'assurer la compatibilité des opérateurs
* de discrétisation avec les procédures personnelles
*-----
'DEBPROC' residu rvx*table ;
RV = RVX . 'EQEX' ;
FREQ = RVX . 'ARG1' ;
EPSO = RVX . 'ARG2' ;
GRAPH = RVX . 'ARG3' ;
NITER = RV . 'NITER' ;
DD = RV . 'PASDETPS' . 'NUPASDT' ;
NN = DD '/' FREQ ;
LO = 'EGA' (DD '-' (FREQ*NN)) 0 ;
'SI' LO ;
RANGO = RV . 'PASDETPS' . 'NUPASDT' ;
TIMEO = RV . 'PASDETPS' . 'TPS' ;
UNO = 'EXCO' 'UX' RV . 'INCO' . 'UN' 'SCAL' ;
UNMO = 'EXCO' 'UX' RV . 'INCO' . 'UN2' 'SCAL' ;
ERR0 = ('MAXIMUM' ('ABS' (UNO '-' UNMO))) '+' 1.D-20 ;
ERR10 = ('LOG' ERRO) '/' ('LOG' 10.) ;
'MESSAGE' 'Résidu en vitesse suivant X au pas'
RANGO ('t=' TIMEO ') : ERRO : ERR10 ;
RV . 'INCO' . 'IT' = RV . 'INCO' . 'IT' 'ET' ('PROG' RANGO) ;
RV . 'INCO' . 'TI' = RV . 'INCO' . 'TI' 'ET' ('PROG' TIMEO) ;
RV . 'INCO' . 'ER' = RV . 'INCO' . 'ER' 'ET' ('PROG' ERR10) ;
V1 = attac ;
RV . 'INCO' . 'POSI' = RV . 'INCO' . 'POSI' 'ET' ('PROG' V1) ;
Y1 = ('LOG' EPSO) '/' ('LOG' 10) ; Y2 = 0. ;
'SI' GRAPH ;
EV1 = 'EVOL' 'MANU' (RV . 'INCO' . 'IT') (RV . 'INCO' . 'ER') ;
'DESSIN' EV1 'YBOR' Y1 Y2 'NCLK' ;
'FINSI' ;
'SI' (ERR10 < Y1) 'ET' (DD > ('MAXI' ('LECT' 10 FREQ))) ;
RV . 'TFINAL' = RV . 'PASDETPS' . 'TPS' ;
'FINSI' ;
'FINSI' ;
RV . 'INCO' . 'UN2' = 'COPIER' RV . 'INCO' . 'UN' ;
mat1 chp1 = 'KOPS' 'MATRIK' ;
'FINP' mat1 chp1 ;

*
*****
* Maillage
*****
*-----
* < L1      X      L2 - L1      >
*
* 7-----6-----5
* |             |             |
* | INLET  H1    |             | OUTLET
* |         v    |             |
* | 1-----2    |             |
* |         |    |             |
* |         3-----4-----4
* |             |             |
* |             |             |
* |             |             |
*-----
* L1 : Longueur de la section d'entrée (avant la marche)
* L2 : Longueur de la totalité du dispositif
* H1 : Hauteur de la section d'entrée
* H2 : Hauteur de la section de sortie
* d1 : Dimension caractéristique d'une maille
L1 = 3.0 ;
L2 = 22.0 ;

```

```

H1 = 1.0 ;
H2 = 1.5 ;
d1 = 0.1 * raf ;
*
* H2-H1 : Hauteur de la marche servant à l'adimensionnalisation
HDIM = H2 - H1 ;
L1 = L1 / HDIM ;
L2 = L2 / HDIM ;
H1 = H1 / HDIM ;
H2 = H2 / HDIM ;
d1 = d1 / HDIM ;
*
* Points du maillage
p1 = 0. (H2-H1) ;
p2 = L1 (H2-H1) ;
p3 = L1 0. ;
p4 = L2 0. ;
p5 = L2 H2 ;
p6 = L1 H2 ;
p7 = 0. H2 ;
p8 = L2 (H2-H1) ;
*
* Section d'entrée
pip2 = p1 'DROIT' 'DINI' d1 'DFIN' d1 p2 ;
p2p6 = p2 'DROIT' 'DINI' d1 'DFIN' d1 p6 ;
p6p7 = p6 'DROIT' 'DINI' d1 'DFIN' d1 p7 ;
p7p1 = p7 'DROIT' 'DINI' d1 'DFIN' d1 p1 ;
mesh1 = 'DALL' pip2 p2p6 p6p7 p7p1 'PLAN' ;
*
* Section de sortie
p6p2 = 'INVE' p2p6 ;
p2p3 = p2 'DROIT' 'DINI' d1 'DFIN' d1 p3 ;
p3p4 = p3 'DROIT' 'DINI' d1 'DFIN' d1 p4 ;
p4p8 = p4 'DROIT' 'DINI' d1 'DFIN' d1 p8 ;
p8p5 = p8 'DROIT' 'DINI' d1 'DFIN' d1 p5 ;
p5p6 = p5 'DROIT' 'DINI' d1 'DFIN' d1 p6 ;
mesh2 = 'DALL' (p6p2 'ET' p2p3) p3p4 (p4p8 'ET' p8p5) p5p6 'PLAN' ;
*
* Modèles NAVIER-STOKES
*****
* Définition des équations
* $DOMTOT : Modèle volumique défini sur le maillage complet
*
* Conditions aux limites en vitesse :
* $INLET : Modèle surfacique défini à l'entrée fluide (Poiseuille)
* $WALL : Modèle surfacique défini sur les murs (adhérence en paroi)
*
* Conditions aux limites en pression :
* $OUTLET : Modèle surfacique défini à la sortie fluide (sert à imposer
* la pression de sortie en pression continue MSONMET)
*
* Post-traitement
* $BOTTOM : Modèle surfacique défini sur le plancher après la marche
* (sert à évaluer la position du point de réattachement)
*
DOMTOT = 'CHAN' 'QUAF' (mesh1 ET mesh2) ;
$DOMTOT = 'MODE' $DOMTOT 'NAVIER-STOKES' DISCR ;
$INLET = 'MODE' p7p1 'NAVIER-STOKES' DISCR ;
$OUTLET = 'MODE' (p4p8 'ET' p8p5) 'NAVIER-STOKES' DISCR ;
$BOTTOM = 'MODE' p3p4 'NAVIER-STOKES' DISCR ;
$WALL = 'MODE' (pip2 ET p2p3 ET p3p4 ET p5p6 ET p6p7)
'NAVIER-STOKES' DISCR ;
*
* Elimination ad hoc
* (En 2D, il faut éliminer les points centres des modèles surfaciques
* avec les points faces des modèles volumiques à cause des MACROs)
FDMTOT = 'DOMA' $DOMTOT 'FACE' ;
CINLET = 'DOMA' $INLET 'CENTRE' ;
COUTLET = 'DOMA' $OUTLET 'CENTRE' ;
CBOTTOM = 'DOMA' $BOTTOM 'CENTRE' ;
CWALL = 'DOMA' $WALL 'CENTRE' ;
'ELM' (FDMTOT 'ET' CINLET 'ET' COUTLET 'ET' CWALL 'ET' CBOTTOM) EPSO ;
*
* On écrase les anciens maillages afin d'éviter toute ambiguïté
DOMTOT = 'DOMA' $DOMTOT 'MAILLAGE' ;
INLET = 'DOMA' $INLET 'MAILLAGE' ;
OUTLET = 'DOMA' $OUTLET 'MAILLAGE' ;
BOTTOM = 'DOMA' $BOTTOM 'MAILLAGE' ;
WALL = 'DOMA' $WALL 'MAILLAGE' ;
*
* Maillage pour d'éventuelles conditions aux limites en pression
'SI' ('EGA' kpres 'MSONMET') ;
OUTLETP = 'DOMA' $outlet KPRES ;
'SINON' ;
OUTLETP = 'DOMA' $domtot KPRES ;
'FINSI' ;
*
* Description des équations
*****
* Grandeurs adimensionnées
Umax = 1.0 ;

```



DEN

DM2S

Tutorial CAST3M pour la mécanique des fluides

```

Re = 150. ;
*
* Profil de vitesse parabolique à l'entrée
VINLET = 'COOR' 2 INLET ;
YMAX = 'MAXI' VINLET ;
YMIN = 'MINI' VINLET ;
UIN = (VINLET '-' YMAX) '*' (VINLET '-' YMIN) ;
UIN = UIN '*' (-4.0*Umax/((YMAX-YMIN)*(YMAX-YMIN))) ;
UIN = 'NOMC' 'UX' UIN 'NATU' 'DISCRET' ;
VIN = 'KCHT' $INLET 'SCAL' 'SOMMET' 'COMP' 'UY' 0. ;
*
* Description du système en vitesse-pression
RV = 'EQEX' $DOMTOT 'ITMA' 50000 'ALFA' 0.75 'FIDT' 100
      'OPTI' 'EFM1' 'EXPL' KSUPG KPRES
      'ZONE' $DOMTOT 'OPER' residu FREQO EPSO GRAPH
      'ZONE' $DOMTOT 'OPER' 'NS' (1./Re) 'INCO' 'UN'
*
* Conditions aux limites par défaut
*
* 'ZONE' $OUTLET 'OPER' 'TOIM' (0. 0.) 'INCO' 'UN'
'OPTI' 'EFM1' 'CENTREE'
'ZONE' $DOMTOT 'OPER' 'DFDT' 1. 'UN' 'DELTAT' 'INCO' 'UN'
;
RVP = 'EQPR' $DOMTOT 'KTYPI' 1
      'ZONE' $DOMTOT 'OPER' 'PRESSION' 0.
*
* En milieu ouvert, pression discontinue il est inutile
* d'imposer la pression en un point
*
* 'CLIM' 'PRES' 'TIMP' (OUTLETP 'ELEM' 1) 0.
;
*
* Description des conditions aux limites
RV = 'EQEX' RV
      'CLIM' 'UN' 'UIMP' WALL 0. 'UN' 'VIMP' WALL 0.
      'UN' 'UIMP' INLET UIN 'UN' 'VIMP' INLET VIN
;
*
* Déclaration des inconnues et initialisations (table INCO)
RV . 'INCO' = 'TABLE' 'INCO' ;
RV . 'INCO' . 'UN' = 'KCHT' $DOMTOT 'VECT' 'SOMMET' (0. 0.) ;
*RV . 'INCO' . 'PRES' = 'KCHT' $DOMTOT 'SCAL' KPRES 0. ;
*
* Champs supplémentaires pour la procédure residu
RV . 'INCO' . 'IT' = 'PROG' ;
RV . 'INCO' . 'TI' = 'PROG' ;
RV . 'INCO' . 'ER' = 'PROG' ;
RV . 'INCO' . 'POSI' = 'PROG' ;
RV . 'INCO' . 'UN2' = 'KCHT' $DOMTOT 'VECT' 'SOMMET' (1.D-3 1.D-3) ;
*
* Algorithme de résolution : méthode semi-explicite
RV . 'PRESSION' = RVP ;
*
EXEC RV ;
*
*****
* Post-traitement

```

```

*****
*
* Localisation du point de réattachement
* Tracés de la vitesse et de la pression
*
CNT1 = WALL ;
NLIO = 14 ;
*
* Point de réattachement (point où dUx/dy=0)
V1 = attac ;
'SAUT' 10 'LIGNE' ;
'MESS' 'ABSCISSE DU POINT DE REATTACHEMENT :' V1 ;
'SI' GRAPH ;
EV1 = 'EVOL' 'MANU' RV . 'INCO' . 'TI' RV . 'INCO' . 'POSI' ;
'DESS' EV1 'MIMA' 'GRIL' 'TITR' 'Localisation reattachement = f(t)' ;
*
* Vitesse
un = RV . 'INCO' . 'UN' ;
vun = 'VECT' UN 0.5 'UX' 'UY' 'JAUNE' ;
trace vun DOMTOT CNT1 'TITR' 'Vitesse' ;
*
* Pression
pn = 'ELNO' $domtot RV . 'INCO' . 'PRESSION' ;
*
* Champ créé si tout se passe bien
*
* pn = RV . 'INCO' . 'PN' ;
trace pn domtot CNT1 NLIO 'TITR' 'Pression' ;
'FINSI' ;
*
*****
* Test de non régression
*****
*
'SI' COMPLET ;
TEST = ('ABS' (V1 '-' 6.158) < 0.0005 ;
'SIMON' ;
TEST = ('ABS' (V1 '-' 9.114) < 0.0005 ;
'FINSI' ;
'MESSAGE' 'Element Vitesse : ' DISCR ;
'MESSAGE' 'Element Pression : ' KPRES ;
'MESSAGE' 'Décentrement : ' KSUPG ;
'SI' TEST ;
'ERREUR' 0 ;
'SIMON' ;
'ERREUR' 5 ;
'FINSI' ;
*
*
'FIN' ;

```

5.5.2 Méthode de projection

```

***** CAS TEST : back_proj_1.dgibi *****
*
* Ce test permet de vérifier le bon fonctionnement des opérateurs
* utilisés pour la résolution des équations de NAVIER-STOKES en EF
* par un algorithme de projection
*
* Le cas étudié est celui d'un écoulement laminaire dans un canal
* en présence d'une marche descendante. On teste la position du
* point de réattachement par rapport au pied de la marche.
*
* Référence : K. Morgan, J. Périaux and F. Thomasset, editors,
* Analysis of laminar flow over a backward facing step, Vol9 of
* Notes on numerical Fluid Mechanics, Vieweg, 1984.
*
-----
* Auteur : F.Dabbene (LTMF) 03/03
-----
*
'SAUT' 'PAGE' ;
'OPTION' 'DIME' 2 'ELEM' 'QUAS' 'ECHO' 0 ;
*
* Pilotage du calcul
*
* DISCR : Element en vitesse (LINE/MACRO/QUAF)
* KPRES : Element en pression (CENTRE/CENTREP2/MSONMET)

```

```

* KSUPG : Méthode de décentrement (CENTREE/SUPG/SUPGDC)
* KMASS : Matrice masse lumpée ou non (EPM1/EF)
*
* DT : Valeur du pas de temps
* (En cas de problème de convergence, augmenter DT et/ou
* prendre KMASS en EPM1)
* EPSO : Test d'arrêt
* RAF : Taille de maille définie par 0.1*raf
*
* FREQO : Fréquence d'évaluation du résidu
* GRAPH : Boolean pour les tracés
* COMPLET : Boolean volume de calcul
*
DISCR = 'LINE' ;
KPRES = 'CENTRE' ;
KSUPG = 'SUPG' ;
KMASS = 'EF' ;
GRAPH = 'FAUX' ;
COMPLET = 'FAUX' ;
'SI' COMPLET ;
DT = 2. ;
FREQO = 50 ;
EPSO = 1.D-6 ;
RAF = 1. ;
'SIMO' ;
DT = 3. ;

```



DEN

DM2S

RAPPORT DM2S

Tutorial CAST3M pour la mécanique des fluides

```

FREQO = 20 ;
EPSO = 1.D-2 ;
RAF = 2. ;
'FINSI' ;
*
-----
* Recherche du point de réattachement (point où dUx/dy=0)
-----
* 1/ Après avoir calculé le gradient de Ux, on ne conserve que les valeurs
* sur la frontière qui nous intéresse sous la forme d'une évolution.
* 2/ On borne l'évolution au voisinage du point de réattachement afin
* d'avoir une variation monotone sur l'intervalle de dUx/dy.
* 3/ On recherche par interpolation le zero de la fonction (dUx/dy(s),s)
* La valeur obtenu est l'abscisse curviligne cherchée
*
Remarques :
* (i) 1/ permettrait de calculer le coeff de frottement à la paroi :
* il suffirait de diviser EV1 par le bon coefficient (Re/2 ici)
* (ii) Il est impératif que dUx/dy soit monotone sur l'intervalle 2/ afin
* qu'il y ait unicité du zero (principe des valeurs intermédiaires)
-----
*'DEBPROC' attac ;
* 1/
Ux = 'KCHT' $DOMTOT 'SCAL' 'SOMMET' 'COMP' 'UX' 0. RV.'INCO'.'UN' ;
dUxDY = 'EXCO' 'UY' ('KOPS' Ux 'GRADS' $DOMTOT) 'SCAL' ;
EV1 = 'EVOL' 'CHPO' dUxDY BOTTOM ;
* 2/
EV2 = 'EXTR' EV1 'APRE' 5. ;
EV3 = 'EXTR' EV2 'AVAN' 10. ;
* 3/
LX1 = 'EXTR' EV3 'ABSC' ;
LY1 = 'EXTR' EV3 'ORDO' ;
'ORDO' LY1 ;
Ymin = 'MINI' LY1 ;
Ymax = 'MAXI' LY1 ;
Delta = Ymin * Ymax ;
'SI' (Delta < 0.) ;
YO = 0. ;
'SINON' ;
'MESS' 'Fonction 'NON' monotone' ;
YO = Ymin ;
* 'ERRE' 'Fonction 'NON' monotone' ;
'FINSI' ;
V1 = 'IPOL' YO LY1 LX1 ;
'FINP' V1 ;
*
*****
* Calcul du résidu basé sur la composante horizontale de la vitesse
* et arrêt suivant un critère transmis
*****
* E/ : RVX : TABLE : TABLE des données créées par EQEX
* ARG1 : Fréquence d'impression
* ARG2 : Critère d'arrêt
* ARG3 : Boolean de tracé
* /S : MAT1 : MATRIK : Objet vide
* /S : CHP1 : CHPO : Objet vide
-----
* MAT1 et CHP1 permettent d'assurer la compatibilité des opérateurs
* de discrétisation avec les procédures personnelles
-----
*'DEBPROC' residu rvx*table ;
RV = RVX . 'EQEX' ;
FREQ = RVX . 'ARG1' ;
EPSO = RVX . 'ARG2' ;
GRAPH = RVX . 'ARG3' ;
NITER = RV . 'NITER' ;
DD = RV . 'PASDETPS' . 'NUPASDT' ;
NN = DD '/' FREQ ;
LO = 'EGA' (DD '-' (FREQ*NN)) 0 ;
'SI' LO ;
RANGO = RV . 'PASDETPS' . 'NUPASDT' ;
TIMEO = RV . 'PASDETPS' . 'TPS' ;
UNO = 'EXCO' 'UX' RV . 'INCO' . 'UN' 'SCAL' ;
UNMO = 'EXCO' 'UX' RV . 'INCO' . 'UN2' 'SCAL' ;
ERRO = ('MAXIMUM' ('ABS' (UNO '-' UNMO))) '+' 1.D-20 ;
ERR10 = ('LOG' ERRO) '/' ('LOG' 10.) ;
'MESSAGE' 'Résidu en vitesse suivant X au pas'
RANGO ('t=' TIMEO) ':' ERRO ':' ERR10 ;
RV . 'INCO' . 'IT' = RV . 'INCO' . 'IT' 'ET' ('PROG' RANGO) ;
RV . 'INCO' . 'TI' = RV . 'INCO' . 'TI' 'ET' ('PROG' TIMEO) ;
RV . 'INCO' . 'ER' = RV . 'INCO' . 'ER' 'ET' ('PROG' ERR10) ;
V1 = attac ;
RV . 'INCO' . 'POSI' = RV . 'INCO' . 'POSI' 'ET' ('PROG' V1) ;
Y1 = ('LOG' EPSO) '/' ('LOG' 10) ; Y2 = 0. ;
'SI' GRAPH ;
EV1 = 'EVOL' 'MANU' (RV . 'INCO' . 'IT') (RV . 'INCO' . 'ER') ;
'DESSIN' EV1 'YBOR' Y1 Y2 'NCKL' ;
'FINSI' ;
'SI' ((ERR10 < Y1) 'ET' (DD > ('MAXI' ('LECT' 10 FREQ)))) ;
RV . 'FINAL' = RV . 'PASDETPS' . 'TPS' ;
'FINSI' ;
'FINSI' ;
RV . 'INCO' . 'UN2' = 'COPIER' RV . 'INCO' . 'UN' ;
mat1 chp1 = 'KOPS' 'MATRIK' ;
'FINP' mat1 chp1 ;

```

```

*
*****
* Maillage
*****
*
-----
* < L1 X L2 - L1 >
*
* 7-----6-----5
*
* INLET H1 H2 OUTLET
*
* 1-----2
* | v
* 3-----4
*
* BOTTOM
*
-----
*
*
* L1 : Longueur de la section d'entrée (avant la marche)
* L2 : Longueur de la totalité du dispositif
* H1 : Hauteur de la section d'entrée
* H2 : Hauteur de la section de sortie
* d1 : Dimension caractéristique d'une maille
L1 = 3.0 ;
L2 = 22.0 ;
H1 = 1.0 ;
H2 = 1.5 ;
d1 = 0.1 * raf ;
*
* H2-H1 : Hauteur de la marche servant à l'adimensionnalisation
HDIM = H2 - H1 ;
L1 = L1 / HDIM ;
L2 = L2 / HDIM ;
H1 = H1 / HDIM ;
H2 = H2 / HDIM ;
d1 = d1 / HDIM ;
*
* Points du maillage
p1 = 0. (H2-H1) ;
p2 = L1 (H2-H1) ;
p3 = L1 0. ;
p4 = L2 0. ;
p5 = L2 H2 ;
p6 = L1 H2 ;
p7 = 0. H2 ;
p8 = L2 (H2-H1) ;
*
* Section d'entrée
pip2 = p1 'DROIT' 'DINI' d1 'DFIN' d1 p2 ;
p2p6 = p2 'DROIT' 'DINI' d1 'DFIN' d1 p6 ;
p6p7 = p6 'DROIT' 'DINI' d1 'DFIN' d1 p7 ;
p7p1 = p7 'DROIT' 'DINI' d1 'DFIN' d1 p1 ;
mesh1 = 'DALL' pip2 p2p6 p6p7 p7p1 'PLAN' ;
*
* Section de sortie
p6p2 = 'INVE' p2p6 ;
p2p3 = p2 'DROIT' 'DINI' d1 'DFIN' d1 p3 ;
p3p4 = p3 'DROIT' 'DINI' d1 'DFIN' d1 p4 ;
p4p8 = p4 'DROIT' 'DINI' d1 'DFIN' d1 p8 ;
p8p5 = p8 'DROIT' 'DINI' d1 'DFIN' d1 p5 ;
p5p6 = p5 'DROIT' 'DINI' d1 'DFIN' d1 p6 ;
mesh2 = 'DALL' (p6p2 'ET' p2p3) p3p4 (p4p8 'ET' p8p5) p5p6 'PLAN' ;
*
*****
* Modèles NAVIER_STOKES
*****
*
* Définition des équations
* $DOMTOT : Modèle volumique défini sur le maillage complet
*
* Conditions aux limites en vitesse :
* $INLET : Modèle surfacique défini à l'entrée fluide (Poiseuille)
* $WALL : Modèle surfacique défini sur les murs (adhérence en paroi)
*
* Conditions aux limites en pression :
* $OUTLET : Modèle surfacique défini à la sortie fluide (sert à imposer
la pression de sortie en pression continue MSONMET)
*
* Post-traitement
* $BOTTOM : Modèle surfacique défini sur le plancher après la marche
(sert à évaluer la position du point de réattachement)
*
DOMTOT = 'CHAN' 'QUAF' (mesh1 ET mesh2) ;
$DOMTOT = 'MODE' DOMTOT 'NAVIER_STOKES' DISCR ;
$INLET = 'MODE' p7p1 'NAVIER_STOKES' DISCR ;
$OUTLET = 'MODE' (p4p8 'ET' p8p5) 'NAVIER_STOKES' DISCR ;
$BOTTOM = 'MODE' p3p4 'NAVIER_STOKES' DISCR ;
$WALL = 'MODE' (pip2 ET p2p3 ET p3p4 ET p5p6 ET p6p7)
'NAVIER_STOKES' DISCR ;

```



DEN

DM2S

RAPPORT DM2S

Tutorial CAST3M pour la mécanique des fluides

```

*
* Elimination ad hoc
* (En 2D, il faut éliminer les points centres des modèles surfaciques
* avec les points faces des modèles volumiques à cause des MACROS)
FDMTOT = 'DOMA' $DMTOT 'FACE' ;
CINLET = 'DOMA' $INLET 'CENTRE' ;
COUTLET = 'DOMA' $OUTLET 'CENTRE' ;
CBOTTOM = 'DOMA' $BOTTOM 'CENTRE' ;
CWALL = 'DOMA' $WALL 'CENTRE' ;
'ELIM' (FDMTOT 'ET' CINLET 'ET' COUTLET 'ET' CWALL 'ET' CBOTTOM) EPSO ;
*
* On écrase les anciens maillages afin d'éviter toute ambiguïté
DMTOT = 'DOMA' $DMTOT 'MALLAGE' ;
INLET = 'DOMA' $INLET 'MALLAGE' ;
OUTLET = 'DOMA' $OUTLET 'MALLAGE' ;
BOTTOM = 'DOMA' $BOTTOM 'MALLAGE' ;
WALL = 'DOMA' $WALL 'MALLAGE' ;
*
* Maillage pour d'éventuelles conditions aux limites en pression
'SI' ('EGA' kpres 'MSOMMET') ;
  OUTLETP = 'DOMA' $outlet KPRES ;
'SINON' ;
  OUTLETP = 'DOMA' $domtot KPRES ;
'FINSI' ;
*
*****
* Description des équations
*****
*
* Grandeurs adimensionnées
Umax = 1.0 ;
Re = 150. ;
*
* Profil de vitesse parabolique à l'entrée
YINLET = 'COOR' 2 INLET ;
YMAX = 'MAXI' YINLET ;
YMIN = 'MINI' YINLET ;
UIN = (YINLET '-' YMAX) '*' (YINLET '-' YMIN) ;
UIN = UIN '*' (-4.0*Umax/((YMAX-YMIN)*(YMAX-YMIN))) ;
UIN = 'NOMC' 'UX' UIN 'NATU' 'DISCRET' ;
VIN = 'KCHT' $INLET 'SCAL' 'SOMMET' 'COMP' 'UY' 0. ;
*
* Description du système en vitesse-pression
RV = 'EQEX' $DMTOT 'ITMA' 5000 'ALFA' 1. 'FIDT' 20000
  'OPTI' 'EF' 'IMPL' KSUPG KPRES
  'ZONE' $DMTOT 'OPER' residu FREQO EPSO GRAPH
  'ZONE' $DMTOT 'OPER' 'NS' (1./Re) 'INCO' 'UN'
  Conditions aux limites par défaut
*
  'ZONE' $OUTLET 'OPER' 'TOIM' (0. 0.) 'INCO' 'UN'
*
  'OPTI' KMASS KSUPG
  'ZONE' $DMTOT 'OPER' 'DFDT' 1. 'UN' DT 'UN' (1./Re) 'INCO' 'UN'
;
'SI' ('EGA' kpres 'MSOMMET') ;
RVP = 'EQEX' 'OPTI' 'EF' KPRES 'IMPL'
  'ZONE' $DMTOT 'OPER' 'KBST' -1. 'INCO' 'UN' 'PRES'
  'CLIM' 'PRES' 'TIMP' OUTLETP 0.
;
'SINON' ;
RVP = 'EQEX' 'OPTI' 'EF' KPRES 'IMPL'
  'ZONE' $DMTOT 'OPER' 'KBST' -1. 'INCO' 'UN' 'PRES'
*
  En milieu ouvert, pression discontinue il est inutile
  d'imposer la pression en un point
*
  'CLIM' 'PRES' 'TIMP' (OUTLETP 'ELEM' 1) 0.
;
'FINSI' ;
*
* Description des conditions aux limites
RV = 'EQEX' RV
  'CLIM' 'UN' 'UIMP' WALL 0. 'UN' 'VIMP' WALL 0.
  'UN' 'UIMP' INLET VIN 'UN' 'VIMP' INLET VIN
;
*
* Déclaration des inconnues et initialisations (table INCO)
RV . 'INCO' = 'TABLE' 'INCO' ;
RV . 'INCO' . 'UN' = 'KCHT' $DMTOT 'VECT' 'SOMMET' (0. 0.) ;

```

```

RV . 'INCO' . 'PRES' = 'KCHT' $DMTOT 'SCAL' KPRES 0. ;
*
* Champs supplémentaires pour la procédure residu
RV . 'INCO' . 'IT' = 'PROG' ;
RV . 'INCO' . 'TI' = 'PROG' ;
RV . 'INCO' . 'ER' = 'PROG' ;
RV . 'INCO' . 'POSI' = 'PROG' ;
RV . 'INCO' . 'UN2' = 'KCHT' $DMTOT 'VECT' 'SOMMET' (1.D-3 1.D-3) ;
*
* Algorithme de résolution : méthode de projection
RV . 'PROJ' = RVP ;
*
EXEC RV ;
*****
* Post-traitement
*****
*
* Localisation du point de réattachement
* Tracés de la vitesse et de la pression
*
CNT1 = WALL ;
NLIO = 14 ;
*
* Point de réattachement (point où dUx/dy=0)
V1 = attac ;
'SAUT' 10 'LIGNE' ;
'MESS' 'ABSCISSE DU POINT DE REATTACHEMENT :' V1 ;
'SI' GRAPH ;
  EV1 = 'EVOL' 'MANU' RV . 'INCO' . 'TI' RV . 'INCO' . 'POSI' ;
  'DESS' EV1 'MIMA' 'GRIL' 'TITR' 'Localisation reattachement = f(t)' ;
*
* Vitesse
un = RV . 'INCO' . 'UN' ;
vun = 'VECT' UN 0.1 'UX' 'UY' 'JAUNE' ;
trace vun DDMTOT CNT1 'TITR' 'Vitesse' ;
*
* Pression
pe = 'EXCO' (RV . 'INCO' . 'PRESSION') 'PRES' 'SCAL' ;
'SI' ('EGA' kpres 'MSOMMET') ;
  mp = 'DOMA' $DMTOT 'MMAIL' ;
  trace pe mp ('CONTOUR' mp) ;
'SINO' ;
  pn = 'ELNO' $DMTOT ('KCHT' $DMTOT 'SCAL' kpres 0. pe) kpres ;
  trace pn domtot CNT1 NLIO 'TITR' 'Pression' ;
'FINSI' ;
*
*****
* Test de non régression
*****
*
'SI' COMPLET ;
  TEST = ('ABS' (V1 '-' 5.8608)) < 0.0005 ;
'SINON' ;
  TEST = ('ABS' (V1 '-' 5.2004)) < 0.0005 ;
'FINSI' ;
'MESSAGE' 'Element Vitesse : ' DISCR ;
'MESSAGE' 'Element Pression : ' KPRES ;
'MESSAGE' 'Decentrement : ' KSUPG ;
'MESSAGE' 'Matrice masse : ' KMASS ;
'SI' TEST ;
  'ERREUR' 0 ;
'SINON' ;
  'ERREUR' 5 ;
'FINSI' ;
*
'FIN' ;

```

5.5.3 Schéma implicite

```

***** CAS TEST : back_impl1.dgibi *****
*
* Ce test permet de vérifier le bon fonctionnement des opérateurs
* utilisés pour la résolution des équations de NAVIER-STOKES en EF

```

```

* par un algorithme implicite
*
* Le cas étudié est celui d'un écoulement laminaire dans un canal
* en présence d'une marche descendante. On teste la position du
* point de réattachement par rapport au pied de la marche.

```



DEN

DM2S

RAPPORT DM2S

Tutorial CAST3M pour la mécanique des fluides

```

*
* Référence : K. Morgan, J. Périaux and F. Thomasset, editors,
* Analysis of laminar flow over a backward facing step, Vol9 of
* Notes on numerical Fluid Mechanics, Vieweg, 1984.
*
-----
* Auteur : F.Dabbene (LTMF)                                03/03
-----
*
'SAUT' 'PAGE' ;
'OPTION' 'DIME' 2 'ELEM' 'QUAS' 'ECHO' 0 ;
*
*- Pilotage du calcul
*
* DISCR : Element en vitesse (LINE/MACRO/QUAF)
* KPRES : Element en pression (CENTRE/CENTREP1, pas MSOMMET)
* KSUPG : Méthode de décentrement (CENTRE/SUPG/SUPGDC)
* KMASS : Matrice masse lumpée ou non (EFM1/EF)
*
* DT      : Valeur du pas de temps
* EPSO    : Test d'arrêt
* RAF     : Taille de maille définie par 0.1*raf
*
* FREQ0   : Fréquence d'évaluation du résidu
* GRAPH   : Boolean pour les tracés
* COMPLET : Boolean volume de calcul
*
DISCR = 'LINE' ;
KPRES = 'CENTRE' ;
'SI' ('EGA' kpres 'MSOMMET') ;
  'ERRE' 'PRESSION MSOMMET interdite en implicite' 5 ;
'FINSI' ;
KSUPG = 'CENTRE' ;
KMASS = 'EF' ;
GRAPH = 'FAUX' ;
COMPLET = 'FAUX' ;
'SI' COMPLET ;
  DT = 0.5 ;
  FREQ0 = 50 ;
  EPSO = 1.D-6 ;
  RAF = 1. ;
'SINO' ;
  DT = 2. ;
  FREQ0 = 20 ;
  EPSO = 1.D-2 ;
  RAF = 2. ;
'FINSI' ;
*
-----
* Recherche du point de réattachement (point où dUx/dy=0)
-----
* 1/ Après avoir calculé le gradient de Ux, on ne conserve que les valeurs
* sur la frontière qui nous intéresse sous la forme d'une évolution.
* 2/ On borne l'évolution au voisinage du point de réattachement afin
* d'avoir une variation monotone sur l'intervale de dUx/dy.
* 3/ On recherche par interpolation le zero de la fonction (dUx/dy(s),s)
* La valeur obtenue est l'abscisse curviligne cherchée
*
* Remarques :
* (i) 1/ permettrait de calculer le coeff de frottement à la paroi :
*      il suffirait de diviser EV1 par le bon coefficient (Re/2 ici)
* (ii) Il est impératif que dUx/dy soit monotone sur l'intervale 2/ afin
*      qu'il y ait unicité du zero (principe des valeurs intermédiaires)
*
-----
*
'DEBPROC' attac ;
* 1/
Ux = 'KCHT' $DOMTOT 'SCAL' 'SOMMET' 'COMP' 'UX' 0. RV.'INCO'.'UN' ;
DuxDY = 'EXCO' 'UY' ('KOPS' Ux 'GRADS' $DOMTOT) 'SCAL' ;
EV1 = 'EVOL' 'CHPO' DuxDY BOTTOM ;
* 2/
EV2 = 'EXTR' EV1 'APRE' 5. ;
EV3 = 'EXTR' EV2 'AVAN' 10. ;
* 3/
LX1 = 'EXTR' EV3 'ABSC' ;
LY1 = 'EXTR' EV3 'ORDO' ;
'ORDO' LY1 ;
Ymin = 'MINI' LY1 ;
Ymax = 'MAXI' LY1 ;
Delta = Ymin * Ymax ;
'SI' (Delta < 0.) ;
  YO = 0. ;
'SINON' ;
  'MESS' 'Fonction non monotone' ;
  YO = Ymin '+' Ymax '/' 2. ;
* 'ERRE' 'Fonction non monotone' 5 ;
'FINSI' ;
V1 = 'IPOL' YO LY1 LX1 ;
'FINP' V1 ;
*
=====
* Calcul du résidu basé sur la composante horizontale de la vitesse
* et arrêt suivant un critère transmis
=====
* E/ : RVX      : TABLE      : TABLE des données créées par EQEX
*      ARG1     :              : Fréquence d'impression
*      ARG2     :              : Critère d'arrêt

```

```

*
* ARG3      : Boolean de tracé
* /S : MAT1  : MATRIK      : Objet vide
* /S : CHP1  : CHPO       : Objet vide
-----
* MAT1 et CHP1 permettent d'assurer la compatibilité des opérateurs
* de discrétisation avec les procédures personnelles
-----
'DEBPROC' residu rvx*table ;
RV = RVX . 'EQEX' ;
FREQ = RVX . 'ARG1' ;
EPSO = RVX . 'ARG2' ;
GRAPH = RVX . 'ARG3' ;
NITER = RV . 'NITER' ;
DD = RV . 'PASDETPS' . 'NUPASDT' ;
NN = DD '/' FREQ ;
LO = 'EGA' (DD '-' (FREQ*NN)) 0 ;
'SI' LO ;
  RANGO = RV . 'PASDETPS' . 'NUPASDT' ;
  TIMEO = RV . 'PASDETPS' . 'TPS' ;
  UNO = 'EXCO' 'UX' RV . 'INCO' . 'UN' 'SCAL' ;
  UNMO = 'EXCO' 'UX' RV . 'INCO' . 'UN2' 'SCAL' ;
  ERRO = ('MAXIMUM' ('ABS' (UNO '-' UNMO))) '+' 1.D-20 ;
  ERR10 = ('LOG' ERRO ) '/' ('LOG' 10.) ;
  'MESSAGE' 'Résidu en vitesse suivant X au pas'
  RANGO ('t= TIMEO ') : 'ERRO' : 'ERR10' ;
  RV . 'INCO' . 'IT' = RV . 'INCO' . 'IT' 'ET' ('PROG' RANGO) ;
  RV . 'INCO' . 'TI' = RV . 'INCO' . 'TI' 'ET' ('PROG' TIMEO) ;
  RV . 'INCO' . 'ER' = RV . 'INCO' . 'ER' 'ET' ('PROG' ERR10) ;
  V1 = attac ;
  RV . 'INCO' . 'POSI' = RV . 'INCO' . 'POSI' 'ET' ('PROG' V1) ;
  Y1 = ('LOG' EPSO) '/' ('LOG' 10) ; Y2 = 0. ;
'SI' GRAPH ;
  EV1 = 'EVOL' 'MANU' (RV . 'INCO' . 'IT')(RV . 'INCO' . 'ER') ;
  'DESSIN' EV1 'YBOR' Y1 Y2 'NCLK' ;
'FINSI' ;
'SI' ((ERR10 < Y1) 'ET' (DD > ('MAXI' ('LECT' 10 FREQ)))) ;
  RV . 'TFINAL' = RV . 'PASDETPS' . 'TPS' ;
'FINSI' ;
'FINSI' ;
RV . 'INCO' . 'UN2' = 'COPIER' RV . 'INCO' . 'UN' ;
mat1 chp1 = 'KOPS' 'MATRIK' ;
'FINP' mat1 chp1 ;
*
*
=====
* Maillage
=====
*
*
-----
*
* < L1 X L2 - L1 >
* 7-----6-----5
* INLET H1 H2 OUTLET
* 1-----2
* | v
* 3-----4
* BOTTOM
*
-----
*
*
* L1 : Longueur de la section d'entrée (avant la marche)
* L2 : Longueur de la totalité du dispositif
* H1 : Hauteur de la section d'entrée
* H2 : Hauteur de la section de sortie
* d1 : Dimension caractéristique d'une maille
L1 = 3.0 ;
L2 = 22.0 ;
H1 = 1.0 ;
H2 = 1.5 ;
d1 = 0.1 * raf ;
*
* H2-H1 : Hauteur de la marche servant à l'adimensionnalisation
HDIM = H2 - H1 ;
L1 = L1 / HDIM ;
L2 = L2 / HDIM ;
H1 = H1 / HDIM ;
H2 = H2 / HDIM ;
d1 = d1 / HDIM ;
*
* Points du maillage
p1 = 0. (H2-H1) ;
p2 = L1 (H2-H1) ;
p3 = L1 0. ;
p4 = L2 0. ;
p5 = L2 H2 ;
p6 = L1 H2 ;
p7 = 0. H2 ;
p8 = L2 (H2-H1) ;
*
* Section d'entrée

```



DEN

DM2S

RAPPORT DM2S

Tutorial CAST3M pour la mécanique des fluides

```

p1p2 = p1 'DROIT' 'DINI' d1 'DFIN' d1 p2 ;
p2p6 = p2 'DROIT' 'DINI' d1 'DFIN' d1 p6 ;
p6p7 = p6 'DROIT' 'DINI' d1 'DFIN' d1 p7 ;
p7p1 = p7 'DROIT' 'DINI' d1 'DFIN' d1 p1 ;
mesh1 = 'DALL' p1p2 p2p6 p6p7 p7p1 'PLAN' ;
*
* Section de sortie
p6p2 = 'INVE' p2p6 ;
p2p3 = p2 'DROIT' 'DINI' d1 'DFIN' d1 p3 ;
p3p4 = p3 'DROIT' 'DINI' d1 'DFIN' d1 p4 ;
p4p8 = p4 'DROIT' 'DINI' d1 'DFIN' d1 p8 ;
p8p5 = p8 'DROIT' 'DINI' d1 'DFIN' d1 p5 ;
p5p6 = p5 'DROIT' 'DINI' d1 'DFIN' d1 p6 ;
mesh2 = 'DALL' (p6p2 'ET' p2p3) p3p4 (p4p8 'ET' p8p5) p5p6 'PLAN' ;
*
*
* Modèles NAVIER_STOKES
*
* Définition des équations
$DOMTOT : Modèle volumique défini sur le maillage complet
*
* Conditions aux limites en vitesse :
*$INLET : Modèle surfacique défini à l'entrée fluide (Poiseuille)
*$WALL : Modèle surfacique défini sur les murs (adhérence en paroi)
*
* Conditions aux limites en pression :
*$OUTLET : Modèle surfacique défini à la sortie fluide (sert à imposer
la pression de sortie en pression continue MSOMMET)
*
* Post-traitement
*$BOTTOM : Modèle surfacique défini sur le plancher après la marche
(sert à évaluer la position du point de réattachement)
*
DOMTOT = 'CHAN' 'QUAF' (mesh1 ET mesh2) ;
$DOMTOT = 'MODE' DOMTOT 'NAVIER_STOKES' DISCR ;
$INLET = 'MODE' p7p1 'NAVIER_STOKES' DISCR ;
$OUTLET = 'MODE' (p4p8 'ET' p8p5) 'NAVIER_STOKES' DISCR ;
$BOTTOM = 'MODE' p3p4 'NAVIER_STOKES' DISCR ;
$WALL = 'MODE' (p1p2 ET p2p3 ET p3p4 ET p5p6 ET p6p7)
'NAVIER_STOKES' DISCR ;
*
* Elimination ad hoc
* (En 2D, il faut éliminer les points centres des modèles surfaciques
* avec les points faces des modèles volumiques à cause des MACROS)
FDMTOT = 'DOMA' $DOMTOT 'FACE' ;
CINLET = 'DOMA' $INLET 'CENTRE' ;
COUTLET = 'DOMA' $OUTLET 'CENTRE' ;
CBOTTOM = 'DOMA' $BOTTOM 'CENTRE' ;
CWALL = 'DOMA' $WALL 'CENTRE' ;
'ELIM' (FDMTOT 'ET' CINLET 'ET' COUTLET 'ET' CWALL 'ET' CBOTTOM) EPSO ;
*
* On écrase les anciens maillages afin d'éviter toute ambiguïté
DOMTOT = 'DOMA' $DOMTOT 'MALLAGE' ;
INLET = 'DOMA' $INLET 'MALLAGE' ;
OUTLET = 'DOMA' $OUTLET 'MALLAGE' ;
BOTTOM = 'DOMA' $BOTTOM 'MALLAGE' ;
WALL = 'DOMA' $WALL 'MALLAGE' ;
*
* Maillage pour d'éventuelles conditions aux limites en pression
'SI' ('EGA' kpres 'MSOMMET') ;
OUTLETP = 'DOMA' $outlet KPRES ;
'SINON' ;
OUTLETP = 'DOMA' $domtot KPRES ;
'FINSI' ;
*
* Description des équations
*
* Grandeurs adimensionnées
Umax = 1.0 ;
Re = 150. ;
*
* Profil de vitesse parabolique à l'entrée
YINLET = 'COOR' 2 INLET ;
YMAX = 'MAXI' YINLET ;
YMIN = 'MINI' YINLET ;
UIN = (YINLET '-' YMAX) '*' (YINLET '-' YMIN) ;
UIN = UIN '*' (-4.0*Umax/((YMAX-YMIN)*(YMAX-YMIN))) ;
UIN = 'NOMC' 'UX' UIN 'NATU' 'DISCRET' ;
VIN = 'KCHT' $INLET 'SCAL' 'SOMMET' 'COMP' 'UY' 0. ;
*
* Description du système en vitesse-pression
RV = 'EQEX' $DOMTOT 'ITMA' 50000 'ALFA' 1. 'FIDT' 100
'OPTI' 'EF' 'IMPL' KSUPG KPRES
'ZONE' $DOMTOT 'OPER' residu FREQO EPSO GRAPH

```

```

'ZONE' $DOMTOT 'OPER' 'NS' (1./Re) 'INCO' 'UN'
'ZONE' $DOMTOT 'OPER' 'KBBT' 1. 'INCO' 'UN' 'PRES'
*
* Conditions aux limites par défaut
* 'ZONE' $OUTLET 'OPER' 'TOIM' (0. 0.) 'INCO' 'UN'
'OPTI' KMASS KSUPG
'ZONE' $DOMTOT 'OPER' 'DFDT' 1. 'UN' DT 'UN' (1./Re) 'INCO' 'UN'
*
*
* Description des conditions aux limites
RV = 'EQEX' RV
'CLIM' 'UN' 'UIMP' WALL 0. 'UN' 'VIMP' WALL 0.
'UN' 'UIMP' INLET UIN 'UN' 'VIMP' INLET VIN
*
*
* Déclaration des inconnues et initialisations (table INCO)
RV 'INCO' = 'TABLE' 'INCO' ;
RV 'INCO' 'UN' = 'KCHT' $DOMTOT 'VECT' 'SOMMET' (0. 0.) ;
* BUG QUAF/CENTRE rv 'INCO' 'PRES' = 'KCHT' $DOMTOT 'SCAL' 'CENTREPO' 0. ;
RV 'INCO' 'PRES' = 'KCHT' $DOMTOT 'SCAL' 'KPRES' 0. ;
*
* Champs supplémentaires pour la procédure residu
RV 'INCO' 'IT' = 'PROG' ;
RV 'INCO' 'TI' = 'PROG' ;
RV 'INCO' 'ER' = 'PROG' ;
RV 'INCO' 'POSI' = 'PROG' ;
RV 'INCO' 'UN2' = 'KCHT' $DOMTOT 'VECT' 'SOMMET' (1.D-3 1.D-3) ;
*
EXEC RV ;
*
* Post-traitement
*
* Localisation du point de réattachement
* Tracés de la vitesse et de la pression
*
CNT1 = WALL ;
NLI0 = 14 ;
*
* Point de réattachement (point où dUx/dy=0)
V1 = attac ;
'SAUT' 10 'LIGNE' ;
'MESS' 'ABSCISSE DU POINT DE REATTACHEMENT :' V1 ;
'SI' GRAPH ;
EV1 = 'EVOL' 'MANU' RV 'INCO' 'TI' RV 'INCO' 'POSI' ;
'DESS' EV1 'MIMA' 'GRIL' 'TITR' 'Localisation reattachement = f(t)' ;
*
* Vitesse
un = RV 'INCO' 'UN' ;
vun = 'VECT' UN 0.5 'UX' 'UY' 'JAUNE' ;
trace vun DOMTOT CNT1 'TITR' 'Vitesse' ;
*
* Pression
pe = RV 'INCO' 'PRES' ;
'SI' ('EGA' kpres 'MSOMMET') ;
mp = 'DOMA' $DOMTOT 'MALL' ;
trace pe mp ('CONTOUR' mp) ;
'SINON' ;
pn = 'ELNO' $DOMTOT ('KCHT' $DOMTOT 'SCAL' kpres 0. pe) kpres ;
'FINSI' ;
*
* Test de non régression
*
'SI' COMPLET ;
TEST = ('ABS' (V1 '-' 5.750)) < 0.0005 ;
'SINON' ;
TEST = ('ABS' (V1 '-' 6.939)) < 0.0005 ;
'FINSI' ;
'MESSAGE' 'Element Vitesse : ' DISCR ;
'MESSAGE' 'Element Pression : ' KPRES ;
'MESSAGE' 'Décentrement : ' KSUPG ;
'MESSAGE' 'Matrice masse : ' KMASS ;
'SI' TEST ;
'ERREUR' 0 ;
'SINON' ;
'ERREUR' 5 ;
'FINSI' ;
*
'FIN' ;

```



DEN

DM2S

SFME/LTMF/RT/03-035/A

Date : 15/09/03

RAPPORT DM2S

Page : 56/65

Tutorial CAST3M pour la mécanique des fluides

6 Approximation de Boussinesq des équations de Navier-Stokes

6.1 Objectif

L'étude de la convection naturelle dans un cylindre contenant de l'oxyde d'uranium permet de présenter les méthodes numériques employées dans CAST3M afin de résoudre les équations de Navier-Stokes pour l'écoulement incompressible d'un fluide dilatable modélisé par l'approximation de Boussinesq. Plutôt que de multiplier les exemples, nous indiquons à l'aide de ce cas test les différents solveurs et éléments finis en vitesse/pression utilisés.

Rappelons que dans le cas où les propriétés thermodynamiques du fluide sont indépendantes de la température et où il est licite de représenter les effets de densité par l'approximation de Boussinesq — prise en compte des variations de densité uniquement dans le terme de flottabilité —, nous devons résoudre

$$\begin{cases} \operatorname{div} \vec{u} = 0 \\ \frac{\partial \vec{u}}{\partial t} + \vec{u} \nabla \vec{u} = \frac{1}{\rho_0} \{-\nabla P + \mu \Delta \vec{u} + \rho \vec{g}\} \\ \frac{\partial T}{\partial t} + \vec{u} \nabla T = \alpha_0 \Delta T + \frac{Q}{\rho c_p} \end{cases}$$

où \vec{u} désigne la vitesse (m/s); ρ la densité du fluide (kg/m^3); P la pression (Pa); T la température (K); μ la viscosité dynamique (kg/ms); α la diffusivité thermique (m^2/s) — $\alpha = \lambda/\rho c_p$ où λ désigne la conductivité thermique et c_p la chaleur spécifique à pression constante et Q une densité de source de chaleur (W/m^3). Les quantités indicées par 0 sont prises à la température de référence T_0 .

6.2 Consignes

Le cas étudié est issu d'un article de D.K. Gartling intitulé "A finite element analysis of volumetrically heated fluids in an axisymmetric enclosure", article paru dans *Finite Elements in Fluids*, vol4 en 1982.

L'enceinte contenant l'oxyde d'uranium est un cylindre fermé de hauteur H et de rayon R . Les parois de l'enceinte sont maintenues à la température ambiante.

L'oxyde d'uranium contenu dans l'enceinte cylindrique est en solution et dégage de la chaleur compte tenu de sa radioactivité. La présence de cette source d'énergie, supposée uniforme et constante dans le temps, génère des mouvements de convection naturelle au sein du cylindre. Si à faible nombre de Grashof on obtient une solution stationnaire, on observe des structures d'écoulement beaucoup plus complexes à partir d'un certain seuil.

Après avoir adimensionné l'approximation de Boussinesq des équations de Navier-Stokes, nous vous demandons de rendre compte des structures d'écoulement pour les nombres de Grashof suivants : 4.39×10^4 , 2.19×10^5 et 4.39×10^5 . Pour cela, il vous faut modifier dans le jeu de données fourni le nombre de Grashof, choisir un solveur et un couple d'éléments finis en vitesse/pression (variables **Gr**, **ICH01** et **IRES0** du jeu de données). En première approximation, le point fixe est inutile (**ICH01=1**). Utilisez **ICH01=2** pour l'algorithme de projection, **ICH01=3** pour la résolution implicite et **ICH01=4** pour l'algorithme semi-explicite. Le cas **ICH01=5** est une autre mise en oeuvre informatique du semi-explicite donnant les mêmes résultats que **ICH01=4**.

Pour l'adimensionnalisation des équations, la longueur de référence est le rayon R de l'enceinte, le temps de référence est R^2/α et la température de référence est QR^2/λ . Peut-être serait-il bon de se poser la question suivante : En quoi ces choix sont-ils non classiques et pourquoi procède-t-on ainsi dans le cas présent ? Durant l'adimensionnalisation, vous



Tutorial CAST3M pour la mécanique des fluides

devez normalement voir apparaître les nombres sans dimension suivants (nombres de Prandtl et de Grashof modifié) :

$$Pr = \frac{\nu}{\alpha}$$

$$Gr^* = \frac{g\beta R^5 Q}{\nu^2 \lambda}$$

où β désigne le coefficient de dilatabilité thermique : $\beta = -\frac{1}{\rho_0} \left. \frac{\partial \rho}{\partial T} \right|_P$.

6.3 Correction

L'adimensionnalisation des équations à l'aide des échelles présentées ci-dessus conduit au système suivant (par abus de notation nous reprenons ici pour les grandeurs adimensionnées les mêmes notations que pour les quantités non adimensionnées, sauf pour la température pour laquelle nous introduisons la variable θ) :

$$\begin{cases} \operatorname{div} \vec{u} = 0 \\ \vec{u}_t + \vec{u} \nabla \vec{u} = -\nabla P + Pr \Delta \vec{u} + Gr^* Pr^2 \theta \\ \theta_t + \vec{u} \nabla \theta = \Delta \theta + 1 \end{cases}$$

Pour un fluide et une température de référence donnés, le Pr est constant. En faisant varier le Gr^* , nous allons donc modifier dans l'équation de quantité de mouvement l'équilibre entre les forces inertielles, les forces visqueuses et la flottabilité.

6.3.1 Effet de Gr^* sur l'écoulement

A partir d'un état initial où le fluide est au repos, la température uniforme et égale à la température extérieure, on applique à $t = 0$ une densité uniforme de chaleur. On regarde alors pour chacun des Gr^* l'évolution au cours du temps des structures d'écoulement.

$Gr^* = 4.39 \times 10^4$: Au bout d'environ une minute, la solution obtenue est stationnaire. Les lignes de courant montrent un tourbillon principal, le fluide descendant le long des parois froides du cylindre. L'évolution de la fonction de courant au cours du temps illustre la mise en place du tourbillon — croissance régulière du maximum de la fonction de courant. Les isothermes sont plus tassées dans la partie haute du cylindre ce qui montre que malgré un nombre de Gr^* modéré les forces de flottabilité ne sont pas négligeables (en conduction pure, les isothermes seraient régulièrement espacées).

$Gr^* = 2.19 \times 10^5$: On obtient à nouveau une solution stationnaire présentant comme pour le précédent Gr^* un tourbillon principal. Cependant, si on regarde l'évolution durant le transitoire du point chaud, on observe qu'il se déplace le long de l'axe avant de se décaler vers l'extérieur lorsqu'il atteint les 2/3 de la hauteur du cylindre. A l'état stationnaire, le fluide plus froid situé à la même cote que le point chaud aurait tendance à descendre. Or on n'observe qu'un seul tourbillon. C'est donc que les forces inertielles sont encore suffisantes pour s'opposer à la flottabilité en ces points. On peut s'attendre pour des Gr^* plus importants à ce que cet équilibre soit rompu et donc à l'apparition d'un tourbillon secondaire.

$Gr^* = 4.39 \times 10^5$: Pour ce Gr^* , les forces de flottabilité prennent le pas sur les forces inertielles pour les points situés entre l'axe de révolution et le point chaud. Il apparaît donc un tourbillon secondaire. Le tourbillon principale et le tourbillon secondaire interagissent. Au cours du temps, pour un pas de temps de $\Delta t = 0.05$ ils croissent et décroissent périodiquement semble-t-il, sans accrocher un mode stable. De ce fait, le point chaud se déplace entre une position proche de l'axe et une position plus excentrée au grès de l'évolution de la taille des deux tourbillons. On n'obtient donc pas pour ce Gr^* de solution stationnaire. Pour un pas de temps de $\Delta t = 0.01$ on obtient une solution stationnaire à nouveau. La solution pseudo-périodique semble donc être un artefact numérique.



Tutorial CAST3M pour la mécanique des fluides

Il convient de signaler que l'auteur de l'article indique que contrairement à d'autres références, il obtient pour ce Grashof une solution stable à deux tourbillons, des solutions périodiques n'apparaissant qu'aux nombres de Grashof plus élevés. Nous avons donc réalisé quelques simulations à plus haut Grashof. A nouveau, selon le pas de temps choisi, on obtient une solution stationnaire pour un petit pas de temps ; une solution pseudo-périodique au delà d'un pas de temps critique. Physiquement, tout est en place pour une bifurcation puisqu'il n'y a aucune raison pour que quelque soit le Grashof les deux vortex s'équilibrent. Il serait sans doute amusant de réaliser une étude de stabilité.

6.3.2 Description du jeu de données

Les principales méthodes de discrétisation des équations de Navier-Stokes sont dans le jeu de données associé au TP d'aujourd'hui. Nous allons passer en revue la manière de coder dans CAST3M ces différentes méthodes ainsi que les éléments finis en vitesse/pression associés. On consultera le cours associé à ce TP pour plus de détails sur ces algorithmes. La variable IRESO (resp. ICHOI) pilote le choix de l'algorithme (resp. de l'élément fini en vitesse/pression). L'inconnue en température est discrétisée dans le même espace que la vitesse.

Les algorithmes de résolution

Algorithme implicite : Les équations en vitesse, pression et température sont discrétisées et résolues simultanément.

Méthode de projection semi-explicite : Les équations en vitesse, pression et température sont découplées : elles sont au sein d'un pas de temps résolues les unes après les autres. L'équation de quantité de mouvement est résolue explicitement si bien que le schéma est stable sous conditions — le pas de temps est limité par des contraintes de stabilités linéaires (conditions sur le nombre CFL et le nombre de Fourier). La pression est solution d'un pseudo-Laplacien. La température est aussi résolue explicitement. *L'algorithme semi-explicite est la variante la plus explicite des méthodes de projection.*

Méthode de projection semi-implicite : Les équations en vitesse, pression et température sont, comme précédemment, découplées. Cependant, elles sont résolues par une approche implicite. De ce fait, le pas de temps n'est pas soumis à des contraintes de stabilités linéaires. Pour obtenir une solution physique aux équations découplées, il doit cependant rester dans des limites raisonnables.

Mise en oeuvre dans CAST3M

Pour décrire le système d'équation, on utilise comme pour la description d'une équation scalaire l'opérateur EQEX. Après avoir créé le maillage du domaine de calcul et le modèle Navier-Stokes associé — cf les deux premiers TP — on précise dans EQEX un certain nombre de paramètres. Pour cela, on fixe une valeur après un mot clef :

ITMA : Nombre de pas de temps (type ENTIER) — ce champ est intéressant lorsque le pas de temps est calculé automatiquement afin d'éviter les calculs sans fin ;

NITER : Nombre d'itérations de point fixe à réaliser à chaque pas de temps (type ENTIER) — à n'utiliser que dans le cas de problème non linéaire (mettre à 1 sinon) ;

OMEGA : Coefficient de relaxation utilisé dans le cas d'itérations de point fixe (type FLOTTANT) — à n'utiliser que dans le cas où on réalise un point fixe (mettre à 1. dans le cas où NITER=1) ;

FIDT : Fréquence d'impression des informations sur le listing (type ENTIER) ;

ALFA : Abusivement appelé CFL, ce paramètre multiplie le pas de temps de stabilité trouvé par le code (type FLOTTANT) — utilisé uniquement lorsque le pas de temps n'est pas imposé par l'utilisateur (voir la notice de l'opérateur DFDT).

TFINAL : Fixe la valeur maximale de la chronologie au delà de laquelle le calcul s'arrête (type FLOTTANT).

Ainsi dans l'exemple suivant :

```
RV = 'EQEX' DOMTOT 'ITMA' 500 'NITER' 1 'OMEGA' 1. 'FIDT' 5000
```

on va résoudre un problème décrit sur le domaine DOMTOT pendant 500 pas de temps sans réaliser d'itérations de point fixe. On limite le volume des impressions sur le listing du calcul en donnant à FIDT une valeur supérieure à ITMA.

Tutorial CAST3M pour la mécanique des fluides

Pour décrire les équations, on opère aussi comme dans le cas de la résolution du problème de transport d'un champ scalaire passif par diffusion et convection. On appelle les opérateurs associés à chacun des monomes des équations :

- ▷ On choisit de nommer l'inconnue de vitesse 'UN', l'inconnue de pression 'PRES' et l'inconnue de température 'TN'.
- ▷ On discrétise l'équation de quantité de mouvement via l'opérateur 'NS'. Les arguments de cet opérateur sont le coefficient de viscosité cinématique ν (coefficient devant le monome $\Delta \vec{u}$ qui correspond à Pr dans la forme adimensionnée des équations), le "coefficient" de flottabilité $\vec{g}\beta$ (correspond au vecteur $(0, -Gr*Pr^2)$), le nom de l'inconnue de température (ici 'TN') et enfin la valeur de la température de référence en adimensionné (ici 0.) soit :
'ZONE' DOMTOT 'OPER' 'NS' NU GB 'TN' 0. 'INCO' 'UN'
avec NU=Pr ; GB=0. (-1.*Gr*Pr*Pr).
Les trois derniers arguments de l'opérateur NS modélisent le terme de flottabilité. Dans le cas où on ne s'intéresse qu'à la résolution des équations de Navier-Stokes, sans effet thermique, donc sans terme de flottabilité, l'opérateur NS a un ou deux arguments (la viscosité cinématique et un éventuel terme source).
- ▷ On discrétise l'équation de transport du champ de température en utilisant soit l'opérateur TSCAL (transport d'un champ scalaire), soit les opérateurs LAPN, KONV et FIMP : ils discrétisent le terme diffusif, le terme convectif et le terme source. Les arguments de l'opérateur TSCAL sont le coefficient de diffusion, le champ de vitesse et le terme source. L'argument de l'opérateur LAPN est le coefficient de diffusion (ici 1.). Les trois arguments de l'opérateur KONV sont un coefficient multiplicateur (ici 1.), le nom du champ de vitesse (ici 'UN') et le coefficient de diffusion afin de pouvoir si on le souhaite décentrer le schéma numérique (méthode 'SUPG' par exemple). L'argument de l'opérateur FIMP est la densité de source de chaleur (ici 1.) :
'ZONE' DOMTOT 'OPER' 'TSCAL' ALFA 'UN' Q 'INCO' 'TN'
ou
'ZONE' DOMTOT 'OPER' 'LAPN' ALFA 'INCO' 'TN'
'ZONE' DOMTOT 'OPER' 'KONV' COEF 'UN' ALFA 'INCO' 'TN'
'ZONE' DOMTOT 'OPER' 'FIMP' Q 'INCO' 'TN'
avec ALFA=1. ; Q=1. ; COEF=1. ;
- ▷ On discrétise les termes en pression dans un troisième temps. L'opérateur KBBT discrétise le terme en gradient de pression de l'équation de quantité de mouvement (terme qui n'est pas discrétisé par l'opérateur NS) ainsi que le terme en divergence \vec{u} . Dans un contexte de résolution implicite, c'est cet opérateur qui couple les inconnues de vitesse et de pression. En méthode de projection, les matrices associées aux opérateurs gradient et divergence — l'une est la transposée de l'autre d'où le nom de l'opérateur — permettent d'évaluer le pseudo-laplacien en pression. L'argument de cet opérateur est un éventuel coefficient multiplicateur. Ce coefficient vaut 1. si l'opérateur est appelé en implicite et -1. sinon.
'ZONE' DOMTOT 'OPER' 'KBBT' C1 'INCO' 'UN' 'PRES'
avec C1=1. en implicite et -1. en explicite.

Les espaces de discrétisation

Le choix des espaces de discrétisation se fait en deux étapes. On indique dans l'objet modèle Navier-Stokes l'espace de discrétisation choisi pour les inconnues autres que la vitesse. On dispose des choix suivants :

LINE : Les espaces d'approximation sont linéaires (éléments dits P1 et Q1) ;

MACRO : Les éléments linéaires sont stabilisés par la technique des macro-éléments, les espaces d'approximation étant linéaires sur chaque micro-élément ;

QUAF : Les espaces d'approximation sont quadratiques (éléments dits P2 et Q2).

On indique l'espace de discrétisation de la pression après la directive OPTI de l'opérateur EQEX afin que les opérateurs de discrétisation puissent travailler :

CENTRE : La pression est constante sur chaque élément ;

CENTREP1 : La pression est linéaire sur chaque élément et discontinue d'un élément à ses voisins ;

MSOMMET : La pression est linéaire sur chaque élément et continue d'un élément à ses voisins.



Tutorial CAST3M pour la mécanique des fluides

Comme nous l'avons vu en cours, tous les couples vitesse/pression ne sont pas judicieux. Cinq possibilités vous sont offertes via la variable ICHOI :

1. **LINE** et **CENTRE** : Cet élément est le plus simple mais sujet aux modes de pression parasites.
2. **MACRO** et **CENTRE** : Cet élément est la version stabilisée de l'élément précédant.
3. **MACRO** et **CENTREP1** : Cet élément est stable dans le cas de quadrangles linéaires. Pour les triangles, il faudrait ajouter la bulle.
4. **QUAF** et **CENTREP1** : En augmentant le degré d'approximation en vitesse on gagne en précision mais au prix d'un temps de calcul plus important. En effet, le couplage entre les degrés de liberté en vitesse augmente d'où des matrices moins creuses qu'en linéaire.
5. **LINE** et **MSOMMET** : Bien que sujet aux modes de pression parasites, cet élément permet d'approximer linéairement et continuellement vitesses et pressions pour un coût raisonnable.
6. **QUAF** et **MSOMMET** : Pour ceux qui souhaitent de la stabilité, de la précision et des pressions continues. A noter que par rapport au choix **CENTREP1**, le choix **MSOMMET** diminue la taille du problème en pression (continuité des pressions).

Les couples **LINE/CENTREP1** et **QUAF/CENTRE** sont sans objets.

Le mot clef associé à l'espace de discrétisation en vitesse est un des arguments du modèle Navier-Stokes. Il sert aussi à indiquer l'espace de discrétisation des inconnues autres que la pression, ici la température. Celui associé à la pression intervient dans les options de l'opérateur **KBBT**.

ATTENTION : *Tous les éléments vitesse/pression ne sont pas disponibles avec tous les algorithmes. Ainsi, on se contentera d'éléments linéaires en semi-explicite (i.e. pas de QUAF en semi-explicite) ; de pressions discontinues en implicite (i.e. pas de MSOMMET en implicite). Un jour viendra ... peut-être.*

Les conditions aux limites

Le domaine de calcul est fermé et axisymétrique. La composante horizontale de la vitesse est donc nulle sur le contour du domaine (axe de symétrie et contour extérieur) ; la composante verticale est nulle sur le contour extérieur ; la température est égale à la température de référence sur le contour extérieur. C'est — comme dans le cas scalaire — la directive **CLIM** de l'opérateur **EQEX** qui permet d'imposer les conditions aux limites.

Comme le domaine est fermée, il faut supprimer une équation du système puisque la contrainte d'incompressibilité lie les inconnues en vitesse. C'est pourquoi on fixe la pression à 0 en un point du support de l'inconnue de pression.

Les utilitaires

Trois procédures utilisateurs sont fournies en tête du jeu de données : **courant**, **calcnuss** et **residu**. Elles permettent respectivement de calculer les lignes de courant, le nombre de Nusselt et l'évolution du résidu en température durant le transitoire. Cette dernière procédure est appelé à chaque pas de temps et l'évolution du résidu en température est tracé à chaque pas de temps.

Les autres procédures sont appelés au moment du post-traitement à l'issue du calcul. En plus des lignes de courant et du Nusselt, nous y post-traitons la température, la vitesse et la pression.

Les procédures utilisateurs et le post-traitement doivent souvent être adaptés à la situation traitée, aux évolutions du code. Ainsi, la procédure **calcnuss** peut être simplifié en calculant directement le gradient de température au sommet (nouvelle option **GRADS** de **KOPS**). C'est pourquoi il vous appartient de comprendre les exemples fournis afin de pouvoir les adapter à vos futurs travaux.

6.4 Complément

- ▷ A.Ern, J-L.Guermond, *Éléments finis : Théorie, applications, mise en oeuvre, Mathématiques & Applications*, Vol 36, Springer Ed., 2002.



DEN

DM2S

RAPPORT DM2S

Tutorial CAST3M pour la mécanique des fluides

- ▷ D.K.Gartling, A finite element analysis of volumetrically heated fluids in an axisymmetric enclosure, in Finite Elements in Fluids, vol4, 1982.
- ▷ P.-L.Viollet, Mécanique des Fluides à Masse Volumique Variable, Presses de l'école des Ponts et Chaussées, 1997.
- ▷ O.C. Zienkiewicz, R.L. Taylor, The Finite element method, Vol 3 : Fluid dynamics, 5th Butterworth-Heinemann Ed., 2000.

6.5 Fichier Gibiane

```
***** CAS TEST : TP4.DGIBI *****
*
* Convection naturelle dans un cylindre uniformement chauffé
* (Navier-Stokes incompressible et approximation de Boussinesq)
*
* Ref biblio : Gartling D.K., A finite element analysis of
* volumetrically heated fluids in an axisymmetric enclosure,
* in Finite Elements in Fluids, vol4, pp233-250, 1982.
*
*-----*
*
* Algorithme de projection et élément (v,T)/p MACRO/CENTREP1
*
*-----*
*
* Auteur : Frédéric DABBENE
*         ttmf3@emt2.smts.cea.fr
*
*-----*
* Procédures spécifiques à évoluer et à généraliser
*-----*
* COURANT : Calcul de la fonction de courant d'un domaine fermé
* CALCNUSS : Calcul du Nusselt (en adimensionné = grad THETA)
* RESIDU : Calcul du résidu en température
*-----*
*
*
*
* COMPLET : Boolean mis à FAUX pour les tests de non régression
* GRAPH : Boolean réalisation ou non des post-traitements
* POST1 : Boolean indiquant si on affiche le résidu en NCLK
* POST2 : Boolean indiquant si on affiche la température en NCLK
* IRESU : Critère d'arrêt sur l'incrément de température
*
*
* COMPLET = FAUX ;
* GRAPH = FAUX ;
* POST1 = FAUX ;
* POST2 = FAUX ;
* 'SI' COMPLET ;
* IRESU = 1.D-6 ;
* 'SING' ;
* IRESU = 1.D-3 ;
* 'FINSI' ;
*
*-----*
* Calcul de la fonction de courant d'un domaine fermé
*-----*
* E/ : UN : CHPO : Champ de vitesse
* E/ : $DOMAINE : MMODEL 'NAVIER_STOKES' volumique
* E/ : $CONTOUR : MMODEL 'NAVIER_STOKES' surfacique
* /S : PSI : CHPO : Fonction de courant
*-----*
* On vérifie qu'on est en dimension 2 mais pas que div(UN)=0
* Si le domaine est ouvert, modifier les conditions aux limites
*-----*
'DEBPROC' courant un*chpoint $domaine*mmodel $contour*mmodel ;
VALO = 'VALE' 'DIME' ;
TEST = 'EGA' VALO 2 ;
'SI' TEST ;
'MESS' 'Remember that Velocity have to be at zero divergence' ;
'SIMON' ;
'ERRE' 832 ;
'QUIT' courant ;
'FINSI' ;
VAL1 = 'VALE' 'MODE' ;
TEST = 'EGA' VAL1 'AXIS' ;
'SI' TEST ;
ROTUO = 'KOPS' un 'ROT' $domaine ;
```

```
XC1 YC1 = 'COORD' ('DOMA' $domaine 'CENTRE') ;
VZ1 = 'NOEL' $domaine ('EXCO' un 'UY' 'SCAL') ;
ROTUO = 2. * VZ1 '-' (ROTUO * XC1) ;
'SING' ;
ROTUO = 'KOPS' un 'ROT' $domaine ;
'FINSI' ;
CONTO = 'DOMA' $contour 'MAILLAGE' ;
RK = 'EQEX' $domaine 'OPTI' 'EP' 'IMPL'
'ZONE' $domaine 'OPER' 'LAPH' -1. 'INCO' 'PSI'
'ZONE' $domaine 'OPER' 'FIMP' ROTUO 'INCO' 'PSI'
'CLIM' 'PSI' 'TIMP' CONTO 0. ;
RK . 'INCO' = 'TABLE' 'INCO' ;
RK . 'INCO' . 'PSI' = 'KCHT' $domaine 'SCAL' 'SOMMET' 0. ;
EXEC RK ;
psi = 'COPI' RK . 'INCO' . 'PSI' ;
'FIMP' psi ;
*
*-----*
* Calcul du Nusselt (en adimensionné = grad THETA)
*-----*
* E/ : TN : CHPO : Température
* E/ : $DOMAINE : MMODEL 'NAVIER_STOKES' volumique
* E/ : $COTE : MMODEL 'NAVIER_STOKES' surfacique
* /S : NUSSI : CHPO : Nusselt local aux sommets
* /S : RES1 : FLOTTANT : Nusselt moyen
*-----*
*
'DEBPROC' calcnuss tn*chpoint $domaine*mmodel $cote*mmodel ;
GRADCO = 'KOPS' tn 'GRAD' $domaine ;
GRADSO = 'ELMO' $domaine GRADCO ;
GSO = 'KCHT' $cote 'VECT' 'SOMMET' GRADSO ;
GCO = 'NOEL' $cote GSO ;
NORM1 = 'DOMA' $domaine 'NORMALE' ;
NCOTE = 'KCHT' $cote 'VECT' 'CENTRE' NORM1 ;
MOT1 = 'MOTS' 'UX' 'UY' ;
NUSSI = 'PSCA' GCO NCOTE MOT1 MOT1 ;
S1 = 'DOMA' $cote 'VOLUME' ;
NUSS2 = NUSSI * S1 ;
RES1 = 'MAXI' ('RESU' NUSS2) ;
'FIMP' nussi res1 ;
*
*-----*
* Calcul du résidu en température et arrêt suivant un critère
*-----*
* E/ : RVX : TABLE : TABLE des données créées par EQEX
* ARG1 : Fréquence d'impression
* ARG2 : Critère d'arrêt
* /S : MAT1 : MATRIK : Objet vide
* /S : CHP1 : CHPO : Objet vide
*-----*
*
'DEBPROC' residu rvx*table ;
RV = rvx . 'EQEX' ;
FREQ = RVX . 'ARG1' ;
EPSO = RVX . 'ARG2' ;
NITER = RV . 'NITER' ;
DD = RV . 'PASDETPS' . 'NUPASDT' ;
MN = DD /' FREQ ;
LO = 'EGA' (DD '-' (FREQ*MN)) 0 ;
'SI' LO ;
RANGO = RV . 'PASDETPS' . 'NUPASDT' ;
TIMEO = RV . 'PASDETPS' . 'TPS' ;
TNO = RV . 'INCO' . 'TN' ;
TNMO = RV . 'INCO' . 'TN2' ;
ERRO = ('MAXIMUM' ('ABS' (TNO '-' TNMO))) '+' 1.D-20 ;
ERR10 = ('LOG' ERRO) /' ('LOG' 10.) ;
'MESSAGE' 'Résidu en Température au pas'
RANGO (' TIMEO ') : ' ERRO ' : ' ERR10 ;
RV . 'INCO' . 'IT' = RV . 'INCO' . 'IT' 'ET' ('PROG' RANGO) ;
```



DEN

DM2S

RAPPORT DM2S

Tutorial CAST3M pour la mécanique des fluides

```

RV . 'INCO' . 'TI' = RV . 'INCO' . 'TI' 'ET' ('PROG' TIMEO) ;
RV . 'INCO' . 'ER' = RV . 'INCO' . 'ER' 'ET' ('PROG' ERR10) ;
EV1 = 'EVOL' 'MANUEL' (RV . 'INCO' . 'IT') (RV . 'INCO' . 'ER') ;
Y1 = ('LOG' EPSO) '/' ('LOG' 10) ;
'SI' POST1 ;
  X1 = 0. ; X2 = RV . 'ITMA' ;
  'DESSIN' EV1 'YBOR' Y1 0. 'NCLK' ;
'FINSI' ;
'SI' POST2 ;
  L1 = (PROG 10. PAS 5. 90.) * 1.D-3 ;
  trace L1 tNO DOM1 CNT1 'TITR' 'Température' 'NCLK' ;
'FINSI' ;
'SI' (ERR10 < Y1) 'ET' (DD > 10) ;
  RV . 'TFINAL' = RV . 'PASDETPS' . 'TPS' ;
'FINSI' ;
'FINSI' ;
RV . 'INCO' . 'TN2' = 'COPIER' RV . 'INCO' . 'TN' ;
mati chpi = 'KOPS' 'MATRIK' ;
'FINP' mati chpi ;
-----
*
*
* OPTION 'DIME' 2 'ELEM' 'QUAS' 'MODE' 'AXIS' 'ISOV' 'SULI' ;
*
* NLIO : Nombre d'isovaleurs
*
NLIO = 14 ;
*
-----
* Choix de l'algorithme et des discrétisations en vitesse/pression
* Les couples vitesse/pression (IHOI/)
* : ( 1/ LINE + CENTRE (déconseillé, -couteux, -stable -précis)
*   (P1/P0 et Q1/P0)
* : ( 2/ MACRO + CENTRE (déconseillé, -couteux, =stable -précis)
*   (P1/P0 stabilisé et Q1/P0 stabilisé)
* : ( 3/ MACRO + CENTREP1 (conseillé, -couteux, +stable, =précis)
*   (isoP2/isoPinc à éviter (PAS DE TRIANGLE SVP))
* : ( isoQ2/isoPinc à consommer sans modération)
* : ( 4/ QUAF + CENTREP1 (conseillé, -couteux, +stable, +précis)
* : ( 5/ QUAF + MSOMMET (conseillé, +couteux, +stable, +précis)
* Les autres combinaisons ne sont pas dans les notices de
* NAVI, EXEC, EQEX, EQPR et DOMA. Donc ...
-----
* Les algorithmes (IRESO/)
* : ( 1/ avec itérations internes et méthode de projection
*   (à utiliser avec précaution : TESTS INSUFFISANTS)
* : ( 2/ sans itérations internes et méthode de projection
* : ( 3/ full implicite
* : ( 4/ semi-explicite et RV . 'PRESSION' = RVP
*   (déconseillé, ancienne syntaxe : nouvelle ci-dessous)
* : ( 5/ semi-explicite et RV . 'POISSON' = RVP
* Les autres choix ne sont pas dans les notices de
* NAVI, EXEC, EQEX, EQPR et DOMA. Donc ...
-----
* Le décentrement (IDCEN/)
* 0/ Pas de décentrement (Galerkin)
* 1/ SUPG (Petrov-Galerkin)
* / SUPGDC (En dernier recours)
-----
IRESO = 2 ;
IHOI = 3 ;
IDCEN = 0 ;
*
* On force MACRO CENTRE dans le cas semi-implicite
'SI' ('EGA' IRESO 4) ;
  'MESS' 'On force MACRO CENTRE dans le cas semi-explicite' ;
  IHOI = 2 ;
  IDCEN = 0 ;
'FINSI' ;
'SI' ('EGA' IHOI 1) ;
  DISCR = 'LINE' ;
  KPRES = 'CENTRE' ;
'FINSI' ;
'SI' ('EGA' IHOI 2) ;
  DISCR = 'MACRO' ;
  KPRES = 'CENTRE' ;
'FINSI' ;
'SI' ('EGA' IHOI 3) ;
  DISCR = 'MACRO' ;
  KPRES = 'CENTREP1' ;
'FINSI' ;
'SI' ('EGA' IHOI 4) ;
  DISCR = 'QUAF' ;
  KPRES = 'CENTREP1' ;
'FINSI' ;
'SI' ('EGA' IHOI 5) ;
  DISCR = 'QUAF' ;
  KPRES = 'MSOMMET' ;
'FINSI' ;
'SI' ('EGA' IDCEN 0) ;
  KSUPG = 'CENTREE' ;
'SINON' ;
  KSUPG = 'SUPG' ;
* KSUPG = 'SUPGDC' ;
'FINSI' ;
*

```

```

=====
* Maillage
=====
*
* Dimensions caractéristiques
L = 1.25 ;
H = 2.5 ;
A = H '/' L ;
AS2 = A '/' 2. ;
FLAG1 = ICHOI < 4 ;
*
* Décalage par rapport à l'axe pour les éléments quadratiques
'SI' FLAG1 ;
  RMIN = 0.0 ;
'SINO' ;
  RMIN = 0.01 ;
'FINSI' ;
*
* Points du maillage
PO = 0.5 AS2 ;
P1 = RMIN 0.0 ;
P2 = 0.5 0.0 ;
P3 = 1.0 0.0 ;
P4 = 1.0 AS2 ;
P5 = 1.0 A ;
P6 = 0.5 A ;
P7 = RMIN A ;
P8 = RMIN AS2 ;
*
* Données pour le maillage
NXY = -4 ;
RAF = 2 ;
NXO = RAF * NXY ;
NX = RAF * NXY ;
NY = RAF * NXY ;
DO = 0.10 ;
D1 = 0.10 ;
D2 = 0.10 ;
*
* Droites du maillage filaire
P1P2 = P1 'DROI' NXO P2 'DINI' DO 'DFIN' D2 ;
P2P3 = P2 'DROI' NX P3 'DINI' D2 'DFIN' D1 ;
P1P3 = P1P2 'ET' P2P3 ;
P3P4 = P3 'DROI' NY P4 'DINI' D1 'DFIN' D2 ;
P4P5 = P4 'DROI' NY P5 'DINI' D2 'DFIN' D1 ;
P3P5 = P3P4 'ET' P4P5 ;
P5P6 = P5 'DROI' NX P6 'DINI' D1 'DFIN' D2 ;
P6P7 = P6 'DROI' NXO P7 'DINI' D2 'DFIN' D0 ;
P5P7 = P5P6 'ET' P6P7 ;
P7P8 = P7 'DROI' NY P8 'DINI' D1 'DFIN' D2 ;
P8P1 = P8 'DROI' NY P1 'DINI' D2 'DFIN' D1 ;
P7P1 = P7P8 'ET' P8P1 ;
*
* Maillages, sous-maillages et modèles associés
CNT1 = P1P3 'ET' P3P5 'ET' P5P7 'ET' P7P1 ;
CNT2 = P1P3 'ET' P3P5 'ET' P5P7 ;
DOM1 = 'DALL' P1P3 P3P5 P5P7 P7P1 ;
DOM1 = 'CHAN' DOM1 'QUAF' ;
*
$DOM1 = 'MODE' DOM1 'NAVIER_STOKES' DISCR ;
$CNT1 = 'MODE' CNT1 'NAVIER_STOKES' DISCR ;
$CNT2 = 'MODE' CNT2 'NAVIER_STOKES' DISCR ;
*
CNT1 = 'DOMA' $CNT1 'CENTRE' ;
CNT2 = 'DOMA' $CNT2 'CENTRE' ;
DOMF = 'DOMA' $DOM1 'FACE' ;
'ELIM' DOMF (CNT1 'ET' CNT2) 1.D-4 ;
DOM1 = 'DOMA' $DOM1 'MALLAGE' ;
CNT1 = 'DOMA' $CNT1 'MALLAGE' ;
CNT2 = 'DOMA' $CNT2 'MALLAGE' ;
MP1 = ('DOMA' $DOM1 KPRES) 'ELEM' 1 ;
***** ?????????????????? Menvlp = doma $vc 'ENVELOPPE' ;
***** ?????????????????? nentr = doma $entree 'NORMALEV' ;
'DOMA' $DOM1 'IMPR' ;
*
=====
* Définition des équations vitesse, pression et température
=====
*
* Paramètres physiques et pas de temps
Pr = 0.65 ;
Gr = 4.4E4 ;
**Gr = 4.4E5 ;
GB1 = Gr * Pr * Pr ;
NU = Pr ;
ALFA = 1. ;
GB = 0. (-1. * GB1) ;
DT = 0.05 ;
*
* Equations en vitesse et température
* ITMA : Nombre de pas de temps
* NITER : Nombre d'itérations internes
* OMEGA : Facteur de relaxation des itérations internes
* FIDT : Nombre maximum de pas de temps
*RV = 'EQEX' $DOM1 'ITMA' 100 'NITER' 10 'OMEGA' 0.9 'FIDT' 10000
'SI' ('EGA' IRESO 1) ;

```



DEN

DM2S

RAPPORT DM2S

Tutorial CAST3M pour la mécanique des fluides

```

RV = 'EQEX' $DOM1 'ITMA' 500 'NITER' 5 'OMEGA' 0.2 'FIDT' 10000
'OPTI' 'EF' KSUPG 'IMPL' KPRES
'ZONE' $DOM1 'OPER' 'residu' 1 IRESU
'ZONE' $DOM1 'OPER' 'NS' NU GB 'TN' 0. 'INCO' 'UN'
'ZONE' $DOM1 'OPER' 'LAPH' ALFA 'INCO' 'TN'
'ZONE' $DOM1 'OPER' 'KONV' 1. 'UN' ALFA 'INCO' 'TN'
'ZONE' $DOM1 'OPER' 'FIMP' 1. 'INCO' 'TN'
'OPTI' 'EF' 'CENTREE'
'ZONE' $DOM1 'OPER' 'DFDT' 1. 'UNM' DT 'INCO' 'UN'
'ZONE' $DOM1 'OPER' 'DFDT' 1. 'TNM' DT 'INCO' 'TN' ;
'FINSI' ;
'SI' ('EGA' IRESO 2) ;
RV = 'EQEX' $DOM1 'ITMA' 1000 'NITER' 1 'OMEGA' 1. 'FIDT' 10000
'OPTI' 'EF' KSUPG 'IMPL' KPRES
'ZONE' $DOM1 'OPER' 'residu' 1 IRESU
'ZONE' $DOM1 'OPER' 'NS' NU GB 'TN' 0. 'INCO' 'UN'
'ZONE' $DOM1 'OPER' 'LAPH' ALFA 'INCO' 'TN'
'ZONE' $DOM1 'OPER' 'KONV' 1. 'UN' ALFA 'INCO' 'TN'
'ZONE' $DOM1 'OPER' 'FIMP' 1. 'INCO' 'TN'
'OPTI' 'EF' 'CENTREE'
'ZONE' $DOM1 'OPER' 'DFDT' 1. 'UN' DT 'INCO' 'UN'
'ZONE' $DOM1 'OPER' 'DFDT' 1. 'TN' DT 'INCO' 'TN' ;
'FINSI' ;
'SI' ('EGA' IRESO 3) ;
RV = 'EQEX' $DOM1 'ITMA' 1000 'NITER' 1 'OMEGA' 1. 'FIDT' 10000
'OPTI' 'EF' KSUPG 'IMPL' KPRES
'ZONE' $DOM1 'OPER' 'residu' 1 IRESU
'ZONE' $DOM1 'OPER' 'NS' NU GB 'TN' 0. 'INCO' 'UN'
'ZONE' $DOM1 'OPER' 'LAPH' ALFA 'INCO' 'TN'
'ZONE' $DOM1 'OPER' 'KONV' 1. 'UN' ALFA 'INCO' 'TN'
'ZONE' $DOM1 'OPER' 'FIMP' 1. 'INCO' 'TN'
** 'ZONE' $DOM1 'OPER' 'TSCAL' ALFA 'UN' 1. 'INCO' 'TN'
'OPTI' 'EFM1' 'CENTREE'
'ZONE' $DOM1 'OPER' 'DFDT' 1. 'UN' DT 'INCO' 'UN'
'ZONE' $DOM1 'OPER' 'DFDT' 1. 'TN' DT 'INCO' 'TN' ;
'FINSI' ;
'SI' ('EGA' IRESO 4) ;
RV = 'EQEX' $DOM1 'ITMA' 1000 'ALFA' 1.
'ZONE' $DOM1 'OPER' 'residu' 1 IRESU
'OPTI' 'SUPG'
'ZONE' $DOM1 'OPER' 'NS' NU GB 'TN' 0. 'INCO' 'UN'
'OPTI' 'SUPG'
'ZONE' $DOM1 'OPER' 'TSCAL' ALFA 'UN' 1. 'INCO' 'TN'
'OPTI' 'CENTREE'
'ZONE' $DOM1 'OPER' 'DFDT' 1. 'UN' 'DELTAT' 'INCO' 'UN'
'ZONE' $DOM1 'OPER' 'DFDT' 1. 'TN' 'DELTAT' 'INCO' 'TN' ;
'FINSI' ;
'SI' ('EGA' IRESO 5) ;
RV = 'EQEX' $DOM1 'ITMA' 1000 'ALFA' 1.
'ZONE' $DOM1 'OPER' 'residu' 1 IRESU
'OPTI' 'SUPG'
'ZONE' $DOM1 'OPER' 'NS' NU GB 'TN' 0. 'INCO' 'UN'
'OPTI' 'SUPG'
'ZONE' $DOM1 'OPER' 'TSCAL' ALFA 'UN' 1. 'INCO' 'TN'
'OPTI' 'CENTREE'
'ZONE' $DOM1 'OPER' 'DFDT' 1. 'UN' 'DELTAT' 'INCO' 'UN'
'ZONE' $DOM1 'OPER' 'DFDT' 1. 'TN' 'DELTAT' 'INCO' 'TN' ;
'FINSI' ;
*
* Conditions aux limites en vitesse et température
RV = 'EQEX' RV 'CLIM'
'TN' 'TIMP' CNT2 0.
'UN' 'UIMP' CNT1 0. 'UN' 'VIMP' CNT2 0. ;
*
* Equation en pression avec condition de Dirichlet en un point
'SI' ('<EG' IRESO 2) ;
RVP = 'EQEX' 'OPTI' 'EF' KPRES
'ZONE' $DOM1 'OPER' 'KBBT' -1. 'INCO' 'UN' 'PRES'
'CLIM' 'PRES' 'TIMP' MPI 0. ;
'FINSI' ;
'SI' ('EGA' IRESO 3) ;
RV = 'EQEX' RV 'OPTI' 'EF' KSUPG 'IMPL' KPRES
'ZONE' $DOM1 'OPER' 'KBBT' 1. 'INCO' 'UN' 'PRES'
'CLIM' 'PRES' 'TIMP' MPI 0. ;
'FINSI' ;
'SI' ('EGA' IRESO 4) ;
RVP = 'EQPR' $DOM1 KTYPI 1
'ZONE' $DOM1 'OPER' 'PRESSION' 0.
PIMP 0. ;
'FINSI' ;
'SI' ('EGA' IRESO 5) ;
* beta1 = 100. ;
RVP = 'EQEX' 'OPTI' 'EF' KPRES
'ZONE' $DOM1 'OPER' 'KBBT' -1. beta1 'INCO' 'UN' 'PRES'
'ZONE' $DOM1 'OPER' 'KBBT' -1. 'INCO' 'UN' 'PRES'
'CLIM' 'PRES' 'TIMP' MPI 0. ;
'FINSI' ;
*
* Initialisation des champs (table INCO)
rv . 'INCO' = TABLE 'INCO' ;
rv . 'INCO' . 'TN' = 'KCHT' $DOM1 'SCAL' 'SOMMET' 0. ;
rv . 'INCO' . 'UN' = 'KCHT' $DOM1 'VECT' 'SOMMET' (0. 0.) ;
rv . 'INCO' . 'PRES' = 'KCHT' $DOM1 'SCAL' KPRES 0. ;
rv . 'INCO' . 'UNM' = 'COPI' rv . 'INCO' . 'UN' ;
rv . 'INCO' . 'TNM' = 'COPI' rv . 'INCO' . 'TN' ;
*
* Champs supplémentaires pour la procédure residu

```

```

rv . 'INCO' . 'UN2' = 'COPI' rv . 'INCO' . 'UN' ;
rv . 'INCO' . 'TN2' = 'COPI' rv . 'INCO' . 'TN' ;
rv . 'INCO' . 'IT' = 'PROG' ;
rv . 'INCO' . 'TI' = 'PROG' ;
rv . 'INCO' . 'ER' = 'PROG' ;
*
* Méthode d'inversion du problème en vitesse
rv . 'METHIN' . TYPINV = 1 ;
rv . 'METHIN' . IMPINV = 0 ;
rv . 'METHIN' . NITMAX = 100 ;
rv . 'METHIN' . PRECOND = 3 ;
rv . 'METHIN' . RESID = 1.e-6 ;
rv . 'METHIN' . 'FCPRECT'=1 ;
rv . 'METHIN' . 'FCPRECI'=1 ;
*
* Méthode d'inversion du problème en pression
'SI' ('<EG' IRESO 2) ;
rvp . 'METHIN' . TYPINV = 1 ;
rvp . 'METHIN' . IMPINV = 0 ;
rvp . 'METHIN' . NITMAX = 100 ;
rvp . 'METHIN' . PRECOND = 3 ;
rvp . 'METHIN' . RESID = 1.e-12 ;
rvp . 'METHIN' . 'FCPRECT'=100 ;
rvp . 'METHIN' . 'FCPRECI'=100 ;
'FINSI' ;
*
* Couplage vitesse/pression : Méthode de projection
'SI' ('<EG' IRESO 2) ;
RV . 'PROJ' = RVP ;
'FINSI' ;
'SI' ('EGA' IRESO 3) ;
* En implicite complet, EXEC sans infos en +
'FINSI' ;
'SI' ('EGA' IRESO 4) ;
RV . 'PRESSION' = RVP ;
'FINSI' ;
'SI' ('EGA' IRESO 5) ;
RV . 'POISSON' = RVP ;
'FINSI' ;
*
*****
* Résolution
*****
*
'OPTION' echo 0 ;
EXEC RV ;
'OPTION' echo 0 ;
*
*****
* Post traitement
*****
*
* Nusselt
nug resg = calcnuss (RV . 'INCO' . 'TN') $DOM1 $CNT2 ;
*
* Fonction de courant
psii = courant (RV . 'INCO' . 'UN') $DOM1 $CNT1 ;
*-----> Début des tracés
'SI' GRAPH ;
trace psii DOM1 CNT1 NLI0 'TITR' 'Fonction de courant' ;
*
* Température
tn = RV . 'INCO' . 'TN' ;
trace tn DOM1 CNT1 NLI0 'TITR' 'Température' ;
*
* Vitesse
un = RV . 'INCO' . 'UN' ;
vun = 'VECT' UN 0.005 'UX' 'UY' 'JAUNE' ;
trace VUN DOM1 CNT1 'TITR' 'Vitesse' ;
*
* Pression
'SI' ('<EG' IRESO 2) ;
pe = 'EXCO' rv . 'INCO' . 'PRESSION' 'PRES' ;
'FINSI' ;
'SI' ('EGA' IRESO 3) ;
pe = RV . 'INCO' . 'PRES' ;
'FINSI' ;
'SI' ('>EG' IRESO 4) ;
pe = RV . 'INCO' . 'PRESSION' ;
'FINSI' ;
pn = 'ELNO' $DOM1 ('KCHT' $DOM1 'SCAL' kpres pe) kpres ;
trace pn dom1 CNT1 NLI0 'TITR' 'Pression' ;
*
'FINSI' ;
*-----> Fin des tracés
*
*****
* Test de non régression
*****
*
'SAUT' 5 'LIGN' ;
VTMAX = maxi (RV . 'INCO' . 'TN') ;
mininu = 'MINI' nug ; maxinu = 'MAXI' nug ;
VTIME = RV . 'PASDETPS' . 'TPS' ;
'SI' COMPLET ;
*
* TMAX = 0.12727 ;

```



DEN

DM2S

RAPPORT DM2S

Tutorial CAST3M pour la mécanique des fluides

```


TNMAX =-3.52366E-2 ;
TNMIN =-0.99271 ;
TNAVE =-5.9121 ;
TIMEM = 3.25 ;
'SINOM' ;
TTMAX = 0.12813 ;
TNMAX =-3.50596E-2 ;
TNMIN =-0.96950 ;
TNAVE =-5.8951 ;
TIMEM = 0.6 ;
'FINSI' ;
DTMAX = 1,D-5 ;
DNMAX = 1,D-4 ;
DTIME = 1,D-4 ;
ER1 = VTMAX - TTMAX ABS ; TEST1 = ER1 < DTMAX ;
ER2 = MININU - TNMIN ABS ; TEST2 = ER2 < DNMAX ;
ER3 = RESG - TNAVE ABS ; TEST3 = ER3 < DNMAX ;
ER4 = VTIME - TIMEM ABS ; TEST4 = ER4 < DTIME ;
'MESS' 'CHAMP/VALEUR/CIBLE/ERREUR/TOLERANCE/TEST' ;
'MESS' '-----' ;

```


```

'MESS' 'Temperature' VTMAX TTMAX ER1 DTMAX ;
list TEST1 ;
'MESS' 'Nusselt min' mininu TNMIN ER2 DNMAX ;
list TEST2 ;
'MESS' 'Nusselt moy' resg TNMIN ER3 DNMAX ;
list TEST3 ;
'MESS' 'Temps final' VTIME TIMEM ER4 DTIME ;
list TEST4 ;
TEST5 = TEST1 ET TEST2 ET TEST3 ET TEST4 ;
SI TEST5 ;
ERRE 0 ;
SINO ;
ERRE 5 ;
FINSI ;
'FIN' ;

```



VAL - ISO
> 0.00E+00
< 1.27E-01



Température

FIG. 19 – $Gr^* = 4.39 \times 10^4$, Température



Tutorial CAST3M pour la mécanique des fluides




FIG. 20 - $Gr^* = 4.39 \times 10^4$, Vitesse et ligne de courant




FIG. 21 - $Gr^* = 4.39 \times 10^4$, Pression

Tutorial CAST3M pour la mécanique des fluides

Conclusion

Nous espérons que ces quelques exemples seront suffisants pour vous permettre de construire des modèles numériques plus complexes. Il est évident que les cas traités ici sont simples même si ils permettent d'identifier la plupart des difficultés.

Nous nous cantonnons en effet aux équations de transport d'un champ scalaire et aux équations de Navier-Stokes pour un fluide incompressible ; à la résolution de ces équations par éléments finis.

Il nous faudrait aussi traiter sur le plan physique du rayonnement, des mélanges de gaz réactifs ou non, du bas-Mach, de la turbulence, de la condensation, de la combustion et de la détonation ; sur le plan numérique des volumes finis et des formulations mixtes.

Ces modèles sont également disponibles dans CAST3M. Nous espérons pouvoir les présenter dans de futurs documents.

Sur le plan pratique, signalons que les jeux de données présentés ici font presque tous partie de la bibliothèque de cas tests de non-régression de CAST3M. Inutile donc de les taper : ils sont disponibles sur le site web CAST3M — compatible avec la version développement de CAST3M — et dans le répertoire dgibi de CAST3M si la version que vous utilisez est suffisamment récente.

Cas traité	Nom du cas test
Equation de la chaleur	tp3.dgibi
Smith & Hutton	smithhutton.dgibi (méthode explicite) smithhutton_impl.dgibi (méthode implicite)
Marche descendante laminaire	back_impl_*.dgibi (méthode implicite, *=1,4) back_pression_*.dgibi (méthode semi-explicite, *=1,2) back_proj_*.dgibi (méthode de projection, *=1,7)
Cylindre chauffant	tp4.dgibi

Remerciements

Pour la réalisation de ce document l'auteur a bénéficié de l'aide précieuse bien que parfois non intentionnelle d'étudiants et de collègues. Qu'ils soient ici chaleureusement remerciés de leur concours est bien la moindre des choses.